OSBORNE/McGraw-Hill

Practical BASIC Programs
Apple II® Edition

Financial • Management Decision • Statistics • Math • Science • Financia



Edited by Lon Poole

Practical BASIC Programs

Apple II® Edition

Disclaimer of Warranties and Limitation of Liabilities

The authors have taken due care in preparing this book and the programs in it, including research, development, and testing to ascertain their effectiveness. The authors and the publishers make no expressed or implied warranty of any kind with regard to these programs nor the supplementary documentation in this book. In no event shall the authors or the publishers be liable for incidental or consequential damages in connection with or arising out of the furnishing, performance, or use of any of these programs.

Published by
OSBORNE/McGraw-Hill
630 Bancroft Way
Berkeley, California 94710
U. S. A.

For information on translations and book distributors outside of the U. S. A., please write OSBORNE/McGraw-Hill at the above address.

PRACTICAL BASIC PROGRAMS — APPLE II® EDITION

2 3 4 5 6 7 8 9 0 MLML 8 7 6 5 4 3 2

ISBN 0-931988-66-7

Cover design by Joseph Mauro.

Acknowledgments

Steven Cook, Martin McNiff, and Robert Thomson conceived, designed, wrote, and tested many of these programs, and prepared the final write-ups and program listings for publication in the original Osborne book, *Practical BASIC Programs.*

Dr. Samuel H. Westerman provided the concepts, designs, and initial program listings for 18 of these programs: Income Averaging, Continuous Interest Compounding, Depreciation Switch, Apportionment by Ratios, Profit Sharing Contributions, Statistical Estimation Theory, Statistics, Unbiased Estimator of Standard Deviation, Chi-Square, Data Forecasting Divergence, Newtonian Interpolation, Lagrangian Interpolation, Sums of Powers, Factorials, Temperature Conversion, and Musical Transposition. He also provided source material for the write-ups for the 18 programs.

Richard E. Beckwith, Ph.D., provided the concept, design, program code, and write-up for the program Swedish Machine (Queuing Theory).

George M. Blake suggested programs Accrued Interest on Bonds and Current Value of a Treasury Bill.

The programs were converted to Apple II by Cynthia Greever.

Contents

Preface

We collected the programs in this book to address the continuing need for readily available and easy-to-use computer programs that do something useful. The supply of such programs has not kept pace with the demand. The the number of computer users is growing at an astounding rate, thanks chiefly to the availability of inexpensive small computers. An increasing number of these people, many of them first-time users, are interested only in the practical aspects of computing. Today, those who view the computer solely as a means of entertainment are few and far between. While more practical programs are now available, many contributed by new users, there just aren't enough. And those that do exist are hard to find. So we brought together in this book forty relatively short programs covering a wide range of practical applications.

Introduction

Purpose

Considering all the small computers people have bought in recent years, it should be easy to find practical computer programs. This is especially true since few users still consider their computer just a diversion. But practical programs are not readily available. The purpose of this book is to help fill that void. All forty programs in this book are useful computer applications. The Applesoft BASIC program listings are included. Type them into your computer and they are ready to run. Both the programmer and the nonprogrammer benefit from this; neither has any programming to do. All of which saves everyone time; the nonprogrammer needn't learn programming and the programmer has more time to write programs no one else has written.

While you don't have to be a programmer to use this book, you must understand the subject matter of the programs you wish to use. It is beyond the scope of this book to explain how, when, where, or why you would use any of them. This does not mean you must be a tax accountant in order to use the Income Averaging program, or a management science professional to use the Transportation Algorithm program. There are sample runs and practice problems for each program. Chances are you can figure out the program's applications from them. And if you understand the applications to some extent, but would like more information, you will find further reading suggested in the References section of many programs.

This book has a secondary purpose as well, and that is to show by example the wide range of subjects that lend themselves to computerization. All too often, computer users who have cut their teeth on entertainment computing have trouble coming up with ideas for practical computing. So even if you don't see a program in this book that is exactly what you need, you may find it easier to invent your own practical applications after studying some of these.

As you look through the programs in this book, you may discover that you can use pieces of the programs or some of the programming techniques in your own work. For example, embodied in these programs is a function for rounding arithmetic calculations to the nearest cent and a subroutine for pausing at the end of each full display screen. For that matter you may be able to use an entire program as a component part of your own larger, more complex program. Some of these programs themselves make use of programs from the book *Some Common BASIC Programs, Apple II Edition,* also published by OSBORNE/McGraw-Hill.

Organization

These programs find their primary applications in four general areas: financial, management decision, statistics, and mathematics and science. This arbitrary classification has no bearing on the utility of the programs per se. Clearly, the question is not what label we have applied to a program, but rather how it can be used.

Towards this end, each program includes a complete write-up in addition to its listing. Each write-up begins with a discussion of its subject matter, its required inputs, and its resultant output. In some cases, there are limitations in the algorithm the program employs, or in the applicability of the program. These are described next. Following this in many programs is a Program Notes section. It tells you how to make minor program changes that make the program operate in a slightly different way, accommodate more or less data, and so forth. These changes may make the difference between the program being convenient or difficult for you to use. The Program Notes section also explains any complex or tricky aspects of the way the program itself is written. Generally speaking, it addresses the technical aspects of implementing the application with a computer program.

Following this narrative material is an example of the program in use. Wherever possible, we set this example in a more or less real-life situation. An example which states a situation that can be resolved by

using the program is more instructive than a list of raw data which you can plug into the program. The point of doing this is not to exercise our imaginations in concocting these situations, but to exercise your imagination in visualizing potential uses of the program. The examples demonstrate as many program features as they can in a problem of reasonable size. We provide the correct answers to the unknowns of the example. The answers may be in narrative form, or they may be an inherent part of the sample run, which comes next. The sample run shows the dialogue that occurs between the user and the computer when the program is used to answer the questions posed in the example. Compare the user's inputs and the computer's outputs in the sample run with the problem stated in the example. You should be able to determine how you would use the program to solve a similar problem.

Practice problems follow each example. Use them to gain more familiarity with different ways you can use a program. Generally, we provide only the answers to these practice problems and not sample runs.

The complete BASIC program listing comes next. The listings are documented with in-line remarks. The remarks make it easier for you to figure out how the program works, if you are so inclined. The remarks (which always begin with the BASIC command REM) are not essential to program operation but they will facilitate your understanding of it.

Finally, we list references for most programs. Investigate these books, articles, etc. if you wish to read more about the subject matter of the program.

How to Use These Programs

Follow the steps listed below to use any of these programs.

1. Read the program write-up and familiarize yourself with how the program works. Read the cited references if they will give you a better understanding of the subject matter which the program addresses. Be sure the program does what you need it to do before going any further.

2. Type the program listing into your computer. Since the remark statements (those that begin with REM) are not essential to program operation, you need not type them in. By doing so, you will save time and programs will take less space, and the programs may even run marginally faster. But if you plan to modify a program extensively, you may be better off including its remarks, since they can be very helpful in tracing program logic flow during debugging.

3. Check your program listing carefully for accuracy. Compare it line-by-line and character-by-character with the published listing. Correct any discrepancies.

4. Save the program on tape or disk. Do it now, before you run the program. That way you can easily retrieve it in the event that anything happens while you are running it.

5. Run the example exactly as shown in the sample run. If you have done everything right to this point, the results should be very similar to those published.

6. If your answers differ markedly from ours, or your program does not run at all (i.e., you get some sort of error message), it is time for some detective work. First, double-check and triple-check your listing against the published one. We cannot overemphasize the importance of this scrutiny. Check for missing program lines and incorrect line numbers. Make sure you have entered the right letter or digit. It is often easy to confuse zeros and O's, ones and I's, two's and Z's, fives and S's, and U's and V's.

By now, your programs should be running correctly. If not, have someone else look over your program. Often another set of eyes can see things that you will miss repeatedly. Try putting the program aside for a while and coming back to it. After a short break, errors you didn't see before may be glaringly obvious.

7. As a further test of your program, run the practice problems. Compare your answers with those in the book.

Income Averaging

This program calculates U.S. federal income tax using the income averaging method (Form 1040, Schedule G). It determines whether a taxpayer qualifies for income averaging, and if so, it displays the entries to complete Schedule G. The program is based on 1980 tax forms, tax rates, and tax laws. It is devised to be used for as many years in the future as the law, rates, and forms remain the same as in 1980.

To use the program, you must enter the taxpayer's name, the taxable year, and the taxpayer's filing status that year (that is, single, married filing jointly, married filing separately, unmarried head of household, or qualifying widow(er)). You then enter the taxpayer's base period income — the four years preceding the taxable year. For 1977 and later, this is the amount from line 34 of Form 1040, or line 11 of Form 1040A (line 10 on the 1977 and 1978 Forms 1040A). You must also enter the number of exemptions for each year 1977 and later, when the program asks for them. For any years of the four-year base period before 1977, you enter the taxable income directly. We should emphasize that you should enter an income figure — even a negative figure — for each year, and you should enter the total number of exemptions claimed each year (when requested), *even though the taxpayer had no net income or even though it was a negative taxable income.*

Note that even though Schedule G directs that line 3 may not be less than zero, whenever the Internal Revenue Service has been confronted with the legislative history of the applicable section of the Internal Revenue Code, it has backed off, and permitted a negative figure on line 3. This program takes advantage of that fact. One the other hand, note that line 6 on Schedule G may not be less than zero, and the program takes account of that, too.

The program then asks you for other applicable income amounts (for example, excluded foreign income) and the taxable income from Schedule TC for the taxable year. It then determines whether income averaging is permissible. If so, it displays the amounts you need in order to fill out Schedule G (1980 format).

Program Notes

The program rounds all calculations to the nearest penny. Some taxpayers prefer to work only to the nearest dollar. To put whole dollar calculations into effect, change lines 39 and 40 as shown below, and when the program asks you to enter dollar amounts, enter them in whole dollars only.

```
39 REM ROUND OFF TO WHOLE DOLLARS
40 DEF FNR (X) = INT (X + 0.5)
```

The 1980 Schedule G reproduced below shows how the elements of array A( ) correspond to the lines and columns of Schedule G, from A(1), the taxable year in the upper right corner, to A(44), the computed tax amount. Note that variables A(5), A(9), and A(14) are in hatched boxes (the IRS intends that they remain blank in 1980). For 1980, the program accounts for that by making them all zero. As years pass, the hatching will pass off to the right, and entries will be required in those boxes.

Example

John and Mary Brown are filing a joint tax form. They have one dependent. Line 34 of their 1979 Form 1040 is $16,699.00. Line 34 of their 1978 and 1977 1040 Forms shows $10,270.00 and $12,600.00. Their taxable income for 1976 was $11,133.00. Their foreign income for 1979 and 1976 was $5,300.00 and $5,000.00. They have no penalty under section 72(m)(5) and no community income. Their taxable income for 1980 was $37,900.00. How would you use this program to help fill out their Schedule G for 1980?

SCHEDULE G (Form 1040)
Department of the Treasury
Internal Revenue Service

Income Averaging

▶See instructions on back.
▶ Attach to Form 1040.

A(1)
21

Name(s) as shown on Form 1040 | Your social security number

Base Period Income and Adjustments	(a) 1st preceding base period year 1979	(b) 2d preceding base period year 1978	(c) 3rd preceding base period year 1977	(d) 4th preceding base period year 1976
1 Enter amount from: Form 1040 (1977, 1978, and 1979)—line 34 Form 1040A (1977 and 1978)—line 10 Form 1040A (1979)—line 11	A(2)	A(3)	A(4)	A(5)
2 a Multiply $750 by your total number of exemptions in 1977 and 1978		A(7)	A(8)	A(9)
b Multiply $1,000 by your total number of exemptions in 1979	A(6)			
3 Taxable income (subtract line 2a or 2b from line 1). If less than zero, enter zero	A(10)	A(11)	A(12)	A(13)
4 Income earned outside of the United States or within U.S. possessions and excluded under sections 911 and 931	A(16)	A(17)	A(18)	A(19)
5 On your 1980 Form 1040, if you checked box (2 or 5 enter $3,200; 1 or 4 enter $2,200; 3 enter $1,600)			A(14)	A(15)
6 Base period income (add lines 3, 4 and 5)	A(22)	A(23)	A(24)	A(25)

Computation of Averageable Income

7 Taxable income for 1980 from Schedule TC (Form 1040), Part I, line 3	7	A(26)		
8 Certain amounts received by owner-employees subject to a penalty under section 72(m)(5)	8	A(20)		
9 Subtract line 8 from line 7	9	A(27)		
10 Excess community income	10	A(21)		
11 Adjusted taxable income (subtract line 10 from line 9). If less than zero, enter zero			11	A(28)
12 Add columns (a) through (d), line 6, and enter here	12	A(29)		
13 Enter 30% of line 12			13	A(30)
14 Averageable income (subtract line 13 from line 11)			14	A(31)

If line 14 is $3,000 or less, do not complete the rest of this form. You do not qualify for income averaging. G

Computation of Tax

15 Amount from line 13			15	A(32)
16 20% of line 14			16	A(33)
17 Total (add lines 15 and 16)			17	A(34)
18 Excess community income from line 10			18	A(21)
19 Total (add lines 17 and 18)			19	A(35)
20 Tax on amount on line 19 (see caution below)			20	A(36)
21 Tax on amount on line 17 (see caution below)	21	A(37)		
22 Tax on amount on line 15 (see caution below)	22	A(38)		
23 Subtract line 22 from line 21	23	A(39)		
24 Multiply the amount on line 23 by 4			24	A(40)
Note: *If no entry was made on line 8 above, skip lines 25 through 27 and go to line 28.*				
25 Tax on amount on line 7 (see caution below)	25	A(41)		
26 Tax on amount on line 9 (see caution below)	26	A(42)		
27 Subtract line 26 from line 25			27	A(43)
28 Tax (add lines 20, 24, and 27). Enter here and on Schedule TC (Form 1040), Part I, line 4 and check Schedule G box			28	A(44)

Caution: Use Tax Rate Schedule X, Y or Z from the Form 1040 instructions to figure your tax on lines 20, 21, 22, 25 and 26. Do not use the tax tables.

Answer:

```
INCOME AVERAGING
TAXPAYER'S NAME IS:
?JOHN AND MARY BROWN

TAXABLE YEAR:
?1980

ENTER FILING STATUS--
 --1 FOR SINGLE
 --2 FOR MARRIED/JOINT
 --3 FOR MARRIED/SEPARATE
 --4 FOR HEAD OF HOUSEHOLD
 --5 FOR QUALIFYING WIDOW(ER)
?2

ENTER THE INCOME FIGURE CORRESPONDING
  TO LINE 34 ON FORM 1040, OR ON FORM
  1040A, CORRESPONDING TO LINE 11(1979)
  OR LINE 10(1977-1978).....
FOR THE YEAR 1979
?16699

HOW MANY EXEMPTIONS CLAIMED THAT YEAR?
?3

ENTER THE INCOME FIGURE CORRESPONDING
  TO LINE 34 ON FORM 1040, OR ON FORM
  1040A, CORRESPONDING TO LINE 11(1979)
  OR LINE 10(1977-1978).....
FOR THE YEAR 1978
?10270

HOW MANY EXEMPTIONS CLAIMED THAT YEAR?
?3

ENTER THE INCOME FIGURE CORRESPONDING
  TO LINE 34 ON FORM 1040, OR ON FORM
  1040A, CORRESPONDING TO LINE 11(1979)
  OR LINE 10(1977-1978).....
FOR THE YEAR 1977
?12600

HOW MANY EXEMPTIONS CLAIMED THAT YEAR?
?3

ENTER TAXABLE INCOME FOR YEAR 1976
?11133

MOST TAXPAYERS DON'T HAVE EXCLUDED
  FOREIGN INCOME, PENALIZED AMOUNTS
  UNDER CODE SEC 72(M)(5), OR EXCESS
  COMMUNITY INCOME.  DO YOU HAVE ANY
  OF THESE ITEMS?  (Y/N)
?Y
```

```
EXCLUDED FOREIGN INCOME--YEAR 1979
?5300
                    SAME--YEAR 1978
?0
                    SAME--YEAR 1977
?0
                    SAME--YEAR 1976
?5000


ENTER PENALIZED AMOUNTS, SEC. 72(M)(5)
?0
ENTER EXCESS COMMUNITY INCOME
?0

ENTER TAXABLE INCOME FOR YEAR 1980
?37900

FOR JOHN AND MARY BROWN,1980 TAX,
USING INCOME AVERAGING,
COMES TO 7718.69

THE FOLLOWING REPRESENTS THE FILLED-IN
SCHEDULE G, USING THE 1980 FORMAT:


************** SCHEDULE G ***********

JOHN AND MARY BROWN --1980
FILING STATUS: MARR./JOINT

ENTER 'C' TO CONTINUE?C
BASE PERIOD INCOME AND ADJUSTMENTS
LINE 1-          1979 : $16699
                 1978 : $10270
                 1977 : $12600
                 1976 : $0
LINE 2A-         1978 : $2250
                 1977 : $2250
LINE 2B-         1979 : $3000
LINE 3-          1979 : $13699
                 1978 : $8020
                 1977 : $10350
                 1976 : $11133
LINE 4-          1979 : $5300
                 1978 : $0
                 1977 : $0
                 1976 : $5000
LINE 5-          1976 : $3200
LINE 6-          1979 : $18999
                 1978 : $8020
                 1977 : $10350
                 1976 : $19333

ENTER 'C' TO CONTINUE?C
```

```
COMPUTATION OF AVERAGEABLE INCOME
    AND COMPUTATION OF TAX
LINE  7 : $37900
LINE  8 : $0
LINE  9 : $37900
LINE 10 : $0
LINE 11 : $37900
LINE 12 : $56702
LINE 13 : $17010.6
LINE 14 : $20889.4
LINE 15 : $17010.6
LINE 16 : $4177.88
LINE 17 : $21188.48
ENTER 'C' TO CONTINUE?C
LINE 18 : $0
LINE 19 : $21188.48
LINE 20 : $3549.77
LINE 21 : $3549.77
LINE 22 : $2507.54
LINE 23 : $1042.23
LINE 24 : $4168.92
LINE 25 : $0
LINE 26 : $0
LINE 27 : $0
LINE 28 : $7718.69
********** END OF SCHEDULE G **********
ENTER 'C' TO CONTINUE WITH NEXT TAXPAYER?X
```

Practice Problems

1. Hester Prynne is single, head of household, and has one dependent. Line 34 of her 1979 Form 1040 is $13,988.39. Line 10 of her 1978 Form 1040A shows $12,650.10. Her taxable income for 1977 was $9,212.58; for 1976 it was $8,775.39. In 1979, she had $1,996.50 excluded under section 911. Her taxable income in 1980 is $25,300.17, and she has $1,100.00 subject to penalty under section 72(m)(5). How should she fill out her 1980 Schedule G?

Answer:

```
FOR HESTER PRYNNE,1980 TAX,
USING INCOME AVERAGING,
COMES TO 5115.8

THE FOLLOWING REPRESENTS THE FILLED-IN
SCHEDULE G, USING THE 1980 FORMAT:

********** SCHEDULE G **********

HESTER PRYNNE --1980
FILING STATUS: UNM. HEAD OF HOUSEHOLD

ENTER 'C' TO CONTINUE?C
BASE PERIOD INCOME AND ADJUSTMENTS
LINE 1-          1979 : $13988.39
                 1978 : $12650.1
```

```
                    1977 : $9212.58
                    1976 : $0
LINE 2A-            1978 : $1500
                    1977 : $1500
LINE 2B-            1979 : $2000
LINE 3-             1979 : $11988.39
                    1978 : $11150.1
                    1977 : $7712.58
                    1976 : $8775.39
LINE 4-             1979 : $1996.5
                    1978 : $0
                    1977 : $0
                    1976 : $0
LINE 5-             1976 : $2200
LINE 6-             1979 : $13984.89
                    1978 : $11150.1
                    1977 : $7712.58
                    1976 : $10975.39

ENTER 'C' TO CONTINUE?C

COMPUTATION OF AVERAGEABLE INCOME
    AND COMPUTATION OF TAX
LINE  7 : $25300.17
LINE  8 : $1100
LINE  9 : $24200.17
LINE 10 : $0
LINE 11 : $24200.17
LINE 12 : $43822.96
LINE 13 : $13146.89
LINE 14 : $11053.28
LINE 15 : $13146.89
LINE 16 : $2210.66
LINE 17 : $15357.55
ENTER 'C' TO CONTINUE?C
LINE 18 : $0
LINE 19 : $15357.55
LINE 20 : $2568.96
LINE 21 : $2568.96
LINE 22 : $2031.25
LINE 23 : $537.71
LINE 24 : $2150.84
LINE 25 : $5599.06
LINE 26 : $5203.06
LINE 27 : $396
LINE 28 : $5115.8
********** END OF SCHEDULE G **********
ENTER 'C' TO CONTINUE WITH NEXT TAXPAYER?X
```

2. Billy Budd is single and has no dependents. Line 34 of his 1979 Form 1040 is $45,130.75. Line 34 of his 1978 Form 1040 is $48,968.20. In 1977 and 1976, his taxable incomes were $37,500.00 and $38,105.05. He had $10,000.00 of excludable foreign income in 1979, $3,000.00 in 1978, $2,500.00 in 1977, and $2,000.00 in 1976. He has no excess community income and nothing subject to section

72(m)(5) penalty. His income for 1980 is $57,762.53. How would he complete Schedule G, if he is eligible for income averaging?
Answer:

```
BILLY BUDD
DOES NOT QUALIFY FOR AVERAGING.
AVERAGEABLE INCOME FOR 1980
IS $1691.33- WHICH IS $3000 OR LESS.
ENTER 'C' TO CONTINUE WITH NEXT TAXPAYER?X
```

Program Listing

```
1  REM  ///// INCOME AVERAGING /////
8  REM      A() HOLDS SCHEDULE G AMOUNTS
9  REM  C() AND R() ARE FOR TAX RATE SCHEDULES
10  DIM A(45),C(4,16),R(4,16)
19  REM  READ TAX RATE SCHEDULES
20  GOSUB 6900
39  REM  ROUNDOFF FUNCTION
40  DEF  FN R(X) =  INT (100 * X + 0.5) / 100
49  REM  CLEAR SCHEDULE G FOR NEXT TAXPAYER
50  FOR I = 1 TO 45
60 A(I) = 0
70  NEXT I
79  REM  CLEAR SCREEN
80  HOME
85  PRINT "INCOME AVERAGING"
90  PRINT "TAXPAYER'S NAME IS:"
100  INPUT Z$
105  PRINT
110  PRINT "TAXABLE YEAR:"
120  INPUT A(1)
125  PRINT
130  PRINT "ENTER FILING STATUS--"
140  PRINT " --1 FOR SINGLE"
150  PRINT " --2 FOR MARRIED/JOINT"
160  PRINT " --3 FOR MARRIED/SEPARATE"
170  PRINT " --4 FOR HEAD OF HOUSEHOLD"
180  PRINT " --5 FOR QUALIFYING WIDOW(ER)"
190  INPUT F
200  PRINT
256  REM
257  REM  **** BASE PERIOD INCOME AND ADJUSTMENTS ****
258  REM
268  REM  ENTER INCOME AMOUNTS--
269  REM  PROCEDURE IS DIFFERENT BEFORE 1977
270  FOR J = 1 TO 4
280  IF A(1) - J > 1976 THEN 320
290  PRINT "ENTER TAXABLE INCOME FOR YEAR ";A(1) - J
300  INPUT A(J + 9)
305  PRINT
310  GOTO 750
320  PRINT "ENTER THE INCOME FIGURE CORRESPONDING"
330  PRINT "  TO LINE 34 ON FORM 1040, OR ON FORM"
```

```
340  PRINT "  1040A, CORRESPONDING TO LINE 11(1979)"
350  PRINT "  OR LINE 10(1977-1978)....."
360  PRINT "FOR THE YEAR ";A(1) - J
370  INPUT A(J + 1)
380  PRINT
470  PRINT "HOW MANY EXEMPTIONS CLAIMED THAT YEAR?"
480  INPUT B
485  PRINT
488  REM  EXEMPTIONS ARE $1000 EACH 1979 AND AFTER,
489  REM  $750 EACH BEFORE THAT
490 A(J + 5) = 1000 * B
500  IF A(1) - J > 1978 THEN 740
510 A(J + 5) = 750 * B
740 A(J + 9) = A(J + 1) - A(J + 5)
750  NEXT J
866  REM  5. FROM FILING STATUS, DETERMINE ZERO
867  REM       BRACKET AMOUNT FOR 1975 AND 1976
868  REM  IF TAX YEAR IS 1981 OF LATER,IGNORE
869  REM  ZERO BRACKET AMOUNTS
870  IF A(1) > 1980 THEN 1010
890  IF F = 1 OR F = 4 THEN 900
893  IF F = 2 OR F = 5 THEN 920
897  IF F = 3 THEN 940
899  REM   SINGLE HEAD OF HOUSEHOLD
900 A(15) = 2200
910  GOTO 960
919  REM  MARRIED/JOINT OR WIDOW(ER)
920 A(15) = 3200
930  GOTO 960
939  REM  MARRIED/SEPARATE
940 A(15) = 1600
949  REM  1975 SAME AS 1976
958  REM  IF TAX YEAR IS 1980, IGNORE 1975
959  REM  ZERO BRACKET AMOUNT
960  IF A(1) = 1980 THEN 1010
970 A(14) = A(15)
1010  PRINT "MOST TAXPAYERS DON'T HAVE EXCLUDED"
1020  PRINT "  FOREIGN INCOME, PENALIZED AMOUNTS"
1030  PRINT "  UNDER CODE SEC 72(M)(5), OR EXCESS"
1040  PRINT "  COMMUNITY INCOME.  DO YOU HAVE ANY"
1050  PRINT "  OF THESE ITEMS?  (Y/N)"
1060  INPUT W$
1070  IF W$ = "N" THEN 1200
1080  PRINT "EXCLUDED FOREIGN INCOME--YEAR ";A(1) - 1
1090  INPUT A(16)
1100  PRINT "                      SAME--YEAR ";A(1) - 2
1110  INPUT A(17)
1120  PRINT "                      SAME--YEAR ";A(1) - 3
1130  INPUT A(18)
1140  PRINT "                      SAME--YEAR ";A(1) - 4
1150  INPUT A(19)
1155  PRINT
1160  PRINT "ENTER PENALIZED AMOUNTS, SEC. 72(M)(5)"
1170  INPUT A(20)
1180  PRINT "ENTER EXCESS COMMUNITY INCOME"
```

```
1190  INPUT A(21)
1195  PRINT
1199  REM  ADD UP BASE PERIOD INCOME COLUMNS A-D
1200 A(22) = A(10) + A(16)
1210 A(23) = A(11) + A(17)
1220 A(24) = A(12) + A(18) + A(14)
1230 A(25) = A(13) + A(19) + A(15)
1238  REM  BASE PERIOD INCOME CANNOT BE NEGATIVE
1239  REM  IN ANY YEAR
1240  FOR I = 22 TO 25
1250  IF A(I) > 0 THEN 1280
1270 A(I) = 0
1280  NEXT I
1286  REM
1287  REM  **** COMPUTATION OF AVERAGEABLE INCOME ****
1288  REM
1289  REM  7. TAXABLE INCOME FROM SCHEDULE TC
1290  PRINT "ENTER TAXABLE INCOME FOR YEAR ";A(1)
1300  INPUT A(26)
1305  PRINT
1309  REM  9. SUBTRACT LINE 8 FROM LINE 7
1310 A(27) = A(26) - A(20)
1318  REM  10. EXCESS COMMUNITY INCOME IS A(21)
1319  REM   11. ADJUSTED TAXABLE INCOME
1320 A(28) = A(27) - A(21)
1329  REM  LINE 11 CANNOT BE NEGATIVE
1330  IF A(28) >  = 0 THEN 1360
1350 A(28) = 0
1359  REM  12. TOTAL BASE PERIOD INCOME
1360 A(29) = A(22) + A(23) + A(24) + A(25)
1379  REM  13. 30% OF LINE 12
1380 A(30) =  FN R(A(29) * .3)
1389  REM  14. AVERAGEABLE INCOME
1390 A(31) = A(28) - A(30)
1400  IF A(31) >  = 3000 THEN 1450
1420  PRINT Z$
1425  PRINT "DOES NOT QUALIFY FOR AVERAGING."
1430  PRINT "AVERAGEABLE INCOME FOR ";A(1)
1435  PRINT "IS $";A(31);"- WHICH IS $3000 OR LESS."
1440  GOTO 2170
1449  REM  15. AMOUNT FROM LINE 13
1450 A(32) = A(30)
1469  REM  16. 20% OF LINE 14
1470 A(33) =  FN R(A(31) * .2)
1479  REM  17. TOTAL (ADD LINES 15 AND 16)
1480 A(34) = A(32) + A(33)
1488  REM  18. EXCESS COMMUNITY INCOME IS A(21)
1489  REM  19. TOTAL (ADD LINES 17 AND 18)
1490 A(35) = A(34) + A(21)
1499  REM  20. TAX ON LINE 19 AMOUNT
1500 S = A(35)
1510  GOSUB 6000
1520 A(36) = T
1529  REM  21. TAX ON LINE 17 AMOUNT
1530 S = A(34)
```

```
1540  GOSUB 6000
1550 A(37) = T
1559  REM  22. TAX ON LINE 15 AMOUNT
1560 S = A(32)
1570  GOSUB 6000
1580 A(38) = T
1589  REM  23. SUBTRACT LINE 22 FROM LINE 21
1590 A(39) = A(37) - A(38)
1599  REM  24. MULTIPLY LINE 23 AMOUNT BY 4
1600 A(40) = 4 * A(39)
1608  REM  -IF THERE'S NO SECTION 72(M)(5) PENALTY
1609  REM  -INCOME, SKIP TO LINE 28
1610  IF A(20) = 0 THEN 1690
1619  REM  25. TAX ON LINE 7 AMOUNT
1620 S = A(26)
1630  GOSUB 6000
1640 A(41) = T
1649  REM  26. TAX ON LINE 9 AMOUNT
1650 S = A(27)
1660  GOSUB 6000
1670 A(42) = T
1679  REM  27. SUBTRACT LINE 26 FROM LINE 25
1680 A(43) = A(41) - A(42)
1689  REM  28. TAX (ADD LINES 20, 24, AND 27)
1690 A(44) = A(36) + A(40) + A(43)
1692  REM
1693  REM  **** PRINT SCHEDULE G ****
1694  REM
1695  PRINT "FOR ";Z$;",";A(1);" TAX,"
1700  PRINT "USING INCOME AVERAGING,"
1710  PRINT "COMES TO ";A(44)
1720  PRINT
1730  PRINT "THE FOLLOWING REPRESENTS THE FILLED-IN"
1740  PRINT "SCHEDULE G, USING THE 1980 FORMAT:"
1750  PRINT
1755  PRINT "************ SCHEDULE G **********"
1759  PRINT
1760  PRINT Z$;" --";A(1)
1770  PRINT "FILING STATUS: ";
1780  IF F = 2 THEN 1810
1782  IF F = 3 THEN 1830
1784  IF F = 4 THEN 1850
1786  IF F = 5 THEN 1870
1788  REM  OTHERWISE F=1
1790  PRINT "SINGLE"
1800  GOTO 1880
1810  PRINT "MARR./JOINT"
1820  GOTO 1880
1830  PRINT "MARR./SEP."
1840  GOTO 1880
1850  PRINT "UNM. HEAD OF HOUSEHOLD"
1860  GOTO 1880
1870  PRINT "QUAL. WIDOW(ER)"
1880  PRINT
1889  REM   WAIT FOR OPERATOR CUE TO CONTINUE
```

```
1890  GOSUB 5800
1895  PRINT "BASE PERIOD INCOME AND ADJUSTMENTS"
1899  REM  PRINT LINES 1, 2, AND 3
1900  FOR I = 2 TO 10 STEP 4
1905  IF I = 6 AND A(1) >  = 1980 THEN  GOSUB 5750
1910  IF I <  > 6 OR A(1) < 1980 THEN  GOSUB 5700
1915  NEXT I
1919  REM  PRINT LINE 4
1920 I = 16
1930  GOSUB 5700
1939  REM  PRINT LINE 5, IF IT'S APPLICABLE
1940  PRINT "LINE 5- ";
1950  IF A(14) = 0 THEN 1970
1960  PRINT ,A(1) - 3;" : $";A(14)
1970  IF A(15) = 0 THEN 1990
1980  PRINT ,A(1) - 4;" : $";A(15)
1989  REM   PRINT LINE 6
1990 I = 22
2000  GOSUB 5700
2005  PRINT
2009  REM  WAIT FOR OPERATOR CUE TO CONTINUE
2010  GOSUB 5800
2015  PRINT
2020  PRINT "COMPUTATION OF AVERAGEABLE INCOME"
2030  PRINT "    AND COMPUTATION OF TAX"
2040  PRINT "LINE  7 : $";A(26)
2050  PRINT "LINE  8 : $";A(20)
2060  PRINT "LINE  9 : $";A(27)
2070  PRINT "LINE 10 : $";A(21)
2080  FOR J = 11 TO 17
2090  PRINT "LINE ";J;" : $";A(J + 17)
2100  NEXT J
2109  REM   WAIT FOR OPERATOR CUE TO CONTINUE
2110  GOSUB 5800
2120  PRINT "LINE 18 : $";A(21)
2130  FOR J = 19 TO 28
2140  PRINT "LINE ";J;" : $";A(J + 16)
2150  NEXT J
2160  PRINT "********** END OF SCHEDULE G **********"
2168  REM  WAIT BEFORE ERASING SCREEN FOR
2169  REM  NEXT TAXPAYER
2170  PRINT "ENTER 'C' TO CONTINUE WITH NEXT TAXPAYER";
2180  INPUT W$
2190  IF W$ = "C" THEN 50
3000  END
5697  REM
5698  REM  *** SUBROUTINE TO PRINT ALL OF LINE 1,2,3,4,OR 6 ***
5699  REM
5700  PRINT "LINE "; INT ((I - 2) / 4) + 1;"- ";
5710  FOR J = 0 TO 3
5720  PRINT ,A(1) - J - 1;" : $";A(I + J)
5730  NEXT J
5740  RETURN
5745  REM  SUBROUTINE TO PRINT OUT LINE 2 A AND B
5750  PRINT "LINE 2A-        1978 : $";A(7)
```

```
5760  PRINT "                    1977 : $";A(8)
5770  PRINT "LINE 2B-            1979 : $";A(6)
5780  RETURN
5795  REM
5796  REM  *** SUBROUTINE TO WAIT FOR OPERATOR CUE
5797  REM       TO CONTINUE SINCE ENTIRE SCHEDULE G
5798  REM       WON'T FIT ON ONE SCREEN ***
5799  REM
5800  PRINT "ENTER 'C' TO CONTINUE";
5810  INPUT W$
5820  RETURN
5994  REM
5995  REM  *** SUBROUTINE TO CALCULATE TAX ON AMOUNT S ***
5996  REM
5999  REM  INITIALIZE TAX TO ZERO
6000 T = 0
6002  REM  SINGLE HAS 16 BRACKETS, ALL OTHERS HAVE 15
6003 K = 15
6004  IF F > 1 THEN 6010
6005 K = 16
6009  REM  DETERMINE WHETHER TO USE SCHED. X,Y, OR Z
6010 I = F
6019  REM  WIDOW(ER) SAVE AS MARRIED/JOINT
6020  IF F < 5 THEN 6040
6030 I = 2
6039  REM  START WITH ZERO BRACKET AMOUNT
6040 J = 1
6049  REM  IS INCOME <= ZERO BRACKET AMOUNT?
6050  IF S <  = C(I,J) THEN 6130
6059  REM  IS INCOME > THIS BRACKET'S CEILING?
6060  IF S > C(I,J + 1) THEN 6090
6068  REM  FOUND MAX TAX BRACKET--
6069  REM  --TAX BALANCE OF INCOME
6070 T = T + (S - C(I,J)) * R(I,J) / 100
6080  GOTO 6130
6089  REM  ACCUMULATE TAX FROM THIS BRACKET
6090 T = T + (C(I,J + 1) - C(I,J)) * R(I,J) / 100
6099  REM  PROCEED TO NEXT BRACKET
6100 J = J + 1
6110  IF J < K THEN 6060
6119  REM  TAX BALANCE OF INCOME AT HIGHEST RATE
6120 T = T + (C(I,J) - C(I,J - 1)) * R(I,J) / 100
6129  REM  ROUND TAX AMOUNT
6130 T =  FN R(T)
6140  RETURN
6897  REM
6898  REM  ***   SUBROUTINE TO READ TAX RATES   ***
6899  REM
6900  RESTORE
6909  REM  FIRST SCHED X
6910  FOR J = 1 TO 16
6920  READ R(1,J),C(1,J)
6930  NEXT J
6939  REM  THEN SCHEDS Y & Z
6940  FOR I = 2 TO 4
```

```
6950  FOR J = 1 TO 15
6960  READ R(I,J),C(I,J)
6970  NEXT J
6980  NEXT I
6985  RETURN
6990  REM
6991  REM  ***** 1979 TAX RATE SCHEDULES X,Y, AND Z *****
6992  REM
6993  REM  FOR EACH TABLE BELOW, GET RATE AND
6994  REM  CUTOFF DATA PAIR FROM THE RIGHTMOST TWO
6995  REM  COLUMNS OF THE APPROPRIATE SCHEDULE
6996  REM
6997  REM  -----SCHEDULE X-----
6998  REM
7000  DATA  14,2300,16,3400,18,4400,19,6500,21,8500
7005  DATA  24,10800,26,12900,30,15000,34,18200
7010  DATA  39,23500,44,28800,49,34100,55,41500
7020  DATA  63,55300,68,81800,70,108300
7027  REM
7028  REM  -----SCHEDULE Y (JOINT/WIDOW)-----
7029  REM
7030  DATA  14,3400,16,5500,18,7600,21,11900,24,16000,28
7040  DATA  20200,32,24600,37,29900,43,35200,49,45800,54
7050  DATA  60000,59,85600,64,109400,68,162400,70,215400
7057  REM
7058  REM  -----SCHEDULE Y (SEPARATE)-----
7059  REM
7060  DATA  14,1700,16,2750,18,3800,21,5950,24,8000,28,10100
7070  DATA  32,12300,37,14950,43,17600,49,22900,54,30000
7080  DATA  59,42800,64,54700,68,81200,70,107700
7087  REM
7088  REM  -----SCHEDULE Z-----
7089  REM
7090  DATA  14,2300,16,4400,18,6500,22,8700,24,11800,26,15000
7100  DATA  31,18200,36,23500,42,28800,46,34100,54,44700,59
7110  DATA  60600,63,81800,68,108300,70,161300
9999  END
```

References

U.S. Internal Revenue Service Code, Sections 1301-05.

U.S. Public Law 91-172, Section 311(b) amending Internal Revenue Code Section 1302.

U.S. Treasury Department, Internal Revenue Service. *Income Averaging,* publication number 506.

U.S. Treasury Department, Internal Revenue Service. Regulations, Sections 1.1301-0 to 1304-6, especially the last sentence of 1.1302-02(b)(1).

Current Value of a Treasury Bill

Treasury bills differ from other investment vehicles in that they are bought and sold at a discount from their face value. The rate will vary as the bill approaches maturity. Also, discounts are figured as if a year were 360 days; the annual percentage rate, or yield, is calculated using a 365/366-day year.

To use this program, enter the T-bill's face value, issue and maturity dates in MONTH, DAY, YEAR format, using one or two numbers for each value (be sure to separate each value with a comma). Then enter the current date and current price bid. The program provides the current value as a dollar amount.

Example

A $10,000 T-bill was sold 1/10/80 to mature on 4/10/80. On 1/17/80, government securities dealers were quoting a bid price of 12.09%. How much was the bill worth?

Answer: The bill was worth $9,717.90

```
CURRENT VALUE OF A TREASURY BILL

               FACE VALUE ($)?10000
        ISSUE DATE (MM,DD,YY)?1,10,80
     MATURITY DATE (MM,DD,YY)?4,10,80
      TODAY'S DATE (MM,DD,YY)?1,17,80
        CURRENT PRICE BID (%)?12.09

CURRENT VALUE = $9717.9

WOULD YOU LIKE TO RE-RUN THIS PROGRAM
   USING NEW DATA (Y/N)?N
```

Practice Problems

1. A one-year bill issued 2/16/80 with a face value of $50,000 was sold 4/10/80 at a 7.35% discount. What was the selling price?

Answer: The bill sold for $46,815.00.

2. Diego bought a $1 million bill on 1/25/80 that matures 7/25/80. On 4/10/80 he noted that dealers were offering 15.54% on his issue. For how much could Diego sell his bill on that day?

Answer: The bill was worth $954,243.33.

Program Listing

```
10  PRINT "CURRENT VALUE OF A TREASURY BILL"
20  DEF  FN A(X) =  INT (X * 100 + .5) / 100
30  PRINT
40  PRINT "                  FACE VALUE ($)";
50  INPUT P
60  PRINT "          ISSUE DATE (MM,DD,YY)";
70  INPUT M,D,Y
```

```
80  GOSUB 340
90  REM  -- X3 = ABSOLUTE NUMBER OF DAYS FROM IMAGINARY DATE
100  REM  --       00/00/00 TO ISSUE DATE
110 X3 = A4
120  PRINT "       MATURITY DATE (MM,DD,YY)";
130  INPUT M,D,Y
140  GOSUB 340
150  REM   -- X4 = TOTAL NUMBER OF DAYS IN PERIOD
160 X4 =  ABS (X3 - A4)
170  PRINT "        TODAY'S DATE (MM,DD,YY)";
180  INPUT M,D,Y
190  GOSUB 340
200  REM  -- X3 = NUMBER OF DAYS FROM ISSUE TO TODAY
210 X3 =  ABS (X3 - A4)
220  PRINT "          CURRENT PRICE BID (%)";
230  INPUT B
240  REM  -- X4 = NUMBER OF DAYS LEFT UNTIL MATURITY
250 X4 = X4 - X3
260  PRINT
270  PRINT "CURRENT VALUE = $"; FN A(P - ((P / 1E4) * (B * (X4 / 360)
     * 100)))
280  PRINT
290  PRINT "WOULD YOU LIKE TO RE-RUN THIS PROGRAM"
295  PRINT "   USING NEW DATA (Y/N)";
300  INPUT Z$
310  IF Z$ = "Y" THEN 30
320  IF Z$ = "N" THEN 450
330  GOTO 290
340  REM  -- SUBROUTING TO DETERMINE NUMBER OF DAYS BETWEEN IMAGINARY
350  REM  -- DATE 00/00/00 AND MM/DD/YY USING 365/366 DAY YEAR.
360  REM  -- REF. ACCOUNTS PAYABLE & ACCOUNTS RECEIVABLE (WANG),
365  REM  --      P.255
370  RESTORE
380  DATA  0,3,3,6,8,11,13,16,19,21,24,26
390  FOR I1 = 1 TO M
400  READ A4
410  NEXT I1
420 A4 = A4 + Y * 365 +  INT (Y / 4) + 1 + (M - 1) * 28 + D
430  IF  INT (Y / 4) = Y / 4 AND M < 3 THEN A4 = A4 - 1
440  RETURN
450  END
```

References

U.S. Department of Treasury. *Information about Treasury Bills Sold at Original Issue,* Form PD 800-D (rev. June 1978).

U.S. Federal Reserve. *Marketable Securities of the United States Government — U.S. Treasury Bills, Notes, and Bonds,* circular No. LLM 185.

Accrued Interest on Bonds

This program computes accrued interest to date on a bond. The program performs calculations using either a 365/366-day standard year, or a 360-day year method (used by some federal agency notes and bonds). Sometimes a bond is issued after the first period has begun. Because this results in a first coupon payment of less than the normal amount, some issues skip that payment and include it with the second period's payment. In this case, you would respond "Y" for Yes when the program asks if this coupon involves a long first period, and enter the additional dates requested.

To use the program, select the type of year the bond calculations will use, then enter the coupon rate and the number of coupons per year. If this coupon involves a long first period, enter a "Y" and enter the date the first period began, the date the bond was acquired, and the date the first coupon would normally have been paid had this not been a long coupon. If this coupon is normal or short, enter "N" and then enter the beginning date for this period. For both long and normal or short coupons, you now enter the date the current period ends, and the settlement date. The program will output the accrued interest in percent of par value.

Example

What is the accrued interest for settlement on 9/10/79, for an 8.25% note due 8/31/81 and issued 8/29/79, with a long first coupon? The coupon dates are 2/28 and 8/31. The first period began on 2/28/79. (Since 1980 is a leap year, the end of the current period is 2/29/80.)

Answer: Accrued interest is 0.271485308% of par value.

```
ACCRUED INTEREST ON BONDS

COMPUTE USING:
               1) 360 DAY YEAR
               2) 365/366 DAY YEAR
               3) END PROGRAM

               WHICH ?2

COUPON RATE (%) ?8.25

NUMBER OF COUPONS PER YEAR ?2

DOES THIS COUPON INCLUDE A
LONG FIRST YEAR PERIOD (Y/N) ?Y

BEGINNING OF FIRST PERIOD
(MM,DD,YY) ?2,28,79
ISSUE DATE (MM,DD,YY) ?8,29,79
FIRST COUPON DATE (MM,DD,YY) ?8,31,79
END OF CURRENT PERIOD
(MM,DD,YY) ?2,29,80
SETTLEMENT DATE (MM,DD,YY) ?9,10,79

ACCRUED INTEREST IS .271485308% OF PAR.

WOULD YOU LIKE TO RE-RUN PROGRAM
USING NEW DATA (Y/N) ?N
```

Practice Problem

What is the accrued interest for settlement on 6/3/80, of a Federal Home Loan Bank Bond at 7.375% due 8/25/82? The coupon payment dates are 2/25 and 8/25. (FHLB bonds use a 360-day year for calculations.)

Answer: 2.00763889% of par.

Program Listing

```
10  PRINT "ACCRUED INTEREST ON BONDS"
20  PRINT
30  PRINT "COMPUTE USING:"
40  PRINT "                1) 360 DAY YEAR"
50  PRINT "                2) 365/366 DAY YEAR"
60  PRINT "                3) END PROGRAM"
70  PRINT
80  PRINT "                WHICH ";
90  INPUT T
100  IF T = 1 THEN 130
110  IF T = 3 THEN 820
120  IF T <  > 2 THEN 80
130  PRINT
140  PRINT "COUPON RATE (%) ";
150  INPUT I
160  PRINT
170  PRINT "NUMBER OF COUPONS PER YEAR ";
180  INPUT N
190 X1 = 0
200  PRINT
210  PRINT "DOES THIS COUPON INCLUDE A"
215  PRINT "LONG FIRST YEAR PERIOD (Y/N) ";
220  INPUT Z$
230  IF Z$ = "N" THEN 410
240  IF Z$ <  > "Y" THEN 210
250  REM  -- SKIP THIS SECTION IF FIRST PERIOD IS NOT LONG
260  PRINT
270  PRINT "BEGINNING OF FIRST PERIOD"
275  PRINT "(MM,DD,YY) ";
280  GOSUB 650
290 X2 = A4
300  REM  -- ISSUE DATE IS DATE CURRENT BONDHOLDER OBTAINED THE BOND
310  PRINT "ISSUE DATE (MM,DD,YY) ";
320  GOSUB 650
330  REM  -- X1 = NUMBER OF DAYS FROM ISSUE TO END OF PARTIAL PERIOD
340 X1 =  ABS (X2 - A4)
350  PRINT "FIRST COUPON DATE (MM,DD,YY) ";
360  GOSUB 650
370  REM  -- X2 = TOTAL NUMBER OF DAYS IN FIRST PERIOD
380 X2 =  ABS (X2 - A4)
390 X1 = (X2 - X1) / X2
400  GOTO 460
410  PRINT
420  PRINT "BEGINNING OF CURRENT PERIOD "
425  PRINT "(MM,DD,YY) ";
```

```
430  GOSUB 650
440  REM  -- X3 = ABSOLUTE NUMBER OF DAYS FROM IMAGINARY DATE
450  REM  --      00/00/00 TO BEGINNING OF CURRENT PERIOD
460 X3 = A4
470  PRINT "END OF CURRENT PERIOD"
475  PRINT "(MM,DD,YY) ";
480  GOSUB 650
490  REM  -- X4 = TOTAL NUMBER OF DAYS IN CURRENT PERIOD
500 X4 =  ABS (X3 - A4)
510  PRINT "SETTLEMENT DATE (MM,DD,YY) ";
520  GOSUB 650
530  REM  -- X3 = NUMBER OF DAYS FROM BEGINNING OF
540  REM  --      CURRENT PERIOD TO SETTLEMENT DATE
550 X3 =  ABS (X3 - A4)
560 X3 = (X3 / X4) + X1
570  PRINT
580  PRINT "ACCRUED INTEREST IS ";(I / N) * X3;"% OF PAR."
590  PRINT
600  PRINT "WOULD YOU LIKE TO RE-RUN PROGRAM"
605  PRINT "USING NEW DATA (Y/N) ";
610  INPUT Z$
620  IF Z$ = "Y" THEN 20
630  IF Z$ = "N" THEN 820
640  GOTO 600
650  INPUT M,D,Y
660  IF T = 1 THEN 800
670  REM  -- SUBROUTINE TO DETERMINE NUMBER OF DAYS BETWEEN
675  REM  -- IMAGINARY DATE 00/00/00 AND MM/DD/YY USING 365/366
680  REM  -- DAY YEAR.  REF. ACCOUNTS PAYABLE & ACCOUNTS
690  REM  -- RECEIVABLE (WANG), P.255
700  RESTORE
710  DATA  0,3,3,6,8,11,13,16,19,21,24,26
720  FOR I1 = 1 TO M
730  READ A4
740  NEXT I1
750 A4 = A4 + Y * 365 +  INT (Y / 4) + 1 + (M - 1) * 28 + D
760  IF  INT (Y / 4) <  > Y / 4 OR M > 2 THEN 770
764 A4 = A4 - 1
770  RETURN
780  REM  -- SUBROUTINE TO COMPUTE NUMBER OF DAYS FROM
790  REM  -- IMAGINARY DATE 00/00/00 TO MM/DD/YY USING 360 YEAR.
800 A4 = (Y * 360) + (M * 30) + D
810  RETURN
820  END
```

Reference

Stigum, Marcia. *The Money Market: Myth, Reality, and Practice.* Homewood, Ill.: Dow Jones-Irwin, 1978. Pages 538-47.

Continuous Interest Compounding

This program calculates the future value of an investment for which interest is compounded continuously. You must enter the interest rate, the number of years that interest will accrue, and the amount of the initial deposit. The total value is based on the following formula:

$$T = De^{IN}$$

where:
T = total value after N years
D = initial investment
I = interest rate
e 2.718281828... (base of natural logarithms)

Example

Dan deposits $800.00 at 7½% interest, compounded continuously. How much will his account be worth in ten years?

Answer: $1,693.60

```
CONTINUOUS INTEREST COMPOUNDING
ENTER THE ANNUAL INTEREST RATE
TO BE PAID ON THE ACCOUNT
?7.5
ENTER THE NUMBER OF YEARS OF FRACTIONS
OF YEARS THAT INTEREST WILL ACCRUE
?10
ENTER YOUR INITIAL DEPOSIT
?800
WITH CONTINUOUS COMPOUNDING A DEPOSIT OF
$800 GROWS IN 10 YEARS AT 7.5% TO
$1693.6
```

Practice Problems

1. If George invests $5,000.00 at 9%, compounded continuously, how much will he have in seven years and three months? (Enter 7 years 3 months as 7.25 years.)

Answer: $9,601.68

2. Dr. Williams invests $70.00 for his niece on the day she is born. How much will she get when she turns 21, at 6¼% compounded continuously?

Answer: $260.08

Program Listing

```
10  PRINT "CONTINUOUS INTEREST COMPOUNDING"
20  PRINT "ENTER THE ANNUAL INTEREST RATE"
30  PRINT "TO BE PAID ON THE ACCOUNT"
40  INPUT I
```

```
50  IF I <  = 0 THEN 20
60  PRINT "ENTER THE NUMBER OF YEARS OF FRACTIONS"
70  PRINT "OF YEARS THAT INTEREST WILL ACCRUE"
80  INPUT N
90  IF N <  = 0 THEN 60
100  PRINT "ENTER YOUR INITIAL DEPOSIT"
110  INPUT D
120  IF D <  = 0 THEN 100
130  PRINT "WITH CONTINUOUS COMPOUNDING A DEPOSIT OF"
140  PRINT "$";D;" GROWS IN ";N;" YEARS AT ";I;"% TO "
150  PRINT "$"; INT (100 * (D *  EXP (I / 100 * N)) + .5) / 100
160  END
```

Rule of 78's Interest

This program computes the interest for each month of a loan in accordance with the rule of 78's. You enter the total interest which would have been earned had the loan continued to maturity, and the number of months in the original period of the loan. The program then prints out a table, with the number of each month, the interest earned during that month by the rule, the interest earned so far, and the balance of (unearned) interest remaining at the end of that month.

Example

A 24-month loan calls for total interest of $10,000.00. What is the interest for each month of the loan?
Answer:

```
RULE OF 78'S INTEREST
ENTER TOTAL INTEREST TO BE EARNED
TO MATURITY OF THE LOAN
?10000
ENTER NO. OF MONTHS DURATION
OF THE LOAN TO MATURITY
?24
MONTH     MONTH'S  ACCUM.    BAL. OF
OF LOAN   INTEREST INT.      INTEREST
1         800      800       9200
2         766.67   1566.67   8433.33
3         733.33   2300      7700
4         700      3000      7000
5         666.67   3666.67   6333.33
6         633.33   4300      5700
7         600      4900      5100
8         566.67   5466.67   4533.33
9         533.33   6000      4000
10        500      6500      3500
11        466.67   6966.67   3033.33
12        433.33   7400      2600
13        400      7800      2200
14        366.67   8166.67   1833.33
15        333.33   8500      1500
16        300      8800      1200
17        266.67   9066.67   933.33
18        233.33   9300      700
19        200      9500      500
20        166.67   9666.67   333.33
21        133.33   9800      200
22        100      9900      100
23        66.67    9966.67   33.33
24        33.33    10000     0
PENNY BREAKAGE ADJUSTED IN LAST MONTH
```

Practice Problems

1. Laurie took out a 36-month loan. Her total interest was $3,614.59. What was the balance of unearned interest if she terminated the loan after two years?
Answer: $423.33

2. Bob Johnson pays off a three-year loan two years early. If the total interest would have been $180.00, how much interest did he actually pay?
Answer: $98.94

Program Listing

```
5  PRINT "RULE OF 78'S INTEREST"
9  REM  ROUNDOFF FUNCTION
10  DEF  FN R(X) =  INT (100 * X + .5) / 100
20  PRINT "ENTER TOTAL INTEREST TO BE EARNED"
30  PRINT "TO MATURITY OF THE LOAN"
40  INPUT I
60  PRINT "ENTER NO. OF MONTHS DURATION"
70  PRINT "OF THE LOAN TO MATURITY"
80  INPUT T
100 T1 = T * (T + 1) / 2
110  PRINT "MONTH   MONTH'S  ACCUM.    BAL. OF"
120  PRINT "OF LOAN INTEREST INT.      INTEREST"
130 A = 0
139  REM  PRINT TABLE
140  FOR M = 1 TO T - 1
170 J =  FN R((T - M + 1) * I / T1)
180 A = A + J
190 B = I - A
240  PRINT M; TAB( 9);J; TAB( 18);A; TAB( 27); FN R(B)
250  NEXT M
255  PRINT T; TAB( 9); FN R(B); TAB( 18);A + B; TAB( 27);0
260  PRINT "PENNY BREAKAGE ADJUSTED IN LAST MONTH"
270  END
```

Present Value of a Tax Deduction

When evaluating an investment, the value of the tax savings is often a consideration. This program calculates the amount of any savings you might realize by deducting interest payments.

You must enter the tax rate, the interest rate on the debt, the term of the debt (in years), and the amount of interest to be paid during each year of the term.

Program Notes

If the level of debt will be constant throughout the term of the investment, you may want to change the program to calculate interest amounts as a percentage of a fixed dollar debt amount. Make these changes.

```
90  PRINT "NUMBER OF PERIODS";
100  INPUT N
102  PRINT "ENTER AMOUNT OF DEBT ($)";
104  INPUT Z
110 P = 0
120  FOR J = 1 TO N
  .
  .
  .
180  PRINT Z * K
190 P = P + (Z * K * T) / ((1 + K) ^ J)
200  NEXT J
```

Example

What is the present value of the tax savings on projected interest payments of $4,000, $3,500, $4,500, $4,000, and $5,000 over the next five years if the tax rate is 48% and the interest rate on that debt will be 19%?

Answer: If the five interest payments are deducted from taxable income, the present value of the taxes saved is $6,044.74.

```
PRESENT VALUE OF AN INTEREST TAX
DEDUCTION

WHAT IS THE TAX RATE (%) ?48
ENTER INTEREST RATE (%) ?19
NUMBER OF PERIODS ?5
INTEREST AMOUNT FOR PERIOD ($) 1 ?4000
                               2 ?3500
                               3 ?4500
                               4 ?4000
                               5 ?5000

PRESENT VALUE OF DEDUCTION = $6044.74

WOULD YOU LIKE TO RE-RUN THIS PROGRAM
WITH NEW DATA (Y/N) ?N
```

Practice Problems

1. If Nick buys a new truck for the shipping business he plans to start, the principal will be $6,250.00 and the interest rate 16%. Nick will make interest payments of $1,000.00, $900.00, and $800.00 during the three-year term of the loan. If his new company will be in a 33% tax bracket, what is the present value of the taxes he will not have to pay when he deducts the interest payments?

Answer: The present value of the tax savings realized by deducting the interest payments is $674.34.

2. If the tax rate is 30% and the interest rate is 15%, what is the present value of taxes saved by deducting interest payments of $45.00, $40.00, $35.00, and $30.00 during the next four years?

Answer: The present value of the tax savings here is $32.86.

Program Listing

```
10  PRINT "PRESENT VALUE OF AN INTEREST TAX
15  PRINT "DEDUCTION"
20  PRINT
30  PRINT "WHAT IS THE TAX RATE (%) ";
40  INPUT T
50 T = T / 100
60  PRINT "ENTER INTEREST RATE (%) ";
70  INPUT K
80 K = K / 100
90  PRINT "NUMBER OF PERIODS ";
100  INPUT N
110 P = 0
120  FOR J = 1 TO N
130  IF J > 1 THEN 160
140  PRINT "INTEREST AMOUNT FOR PERIOD ($) ";
150  GOTO 170
160  PRINT "                                    ";
170  PRINT J;" ";
180  INPUT Z
190 P = P + (Z * T) / ((1 + K) ^ J)
200  NEXT J
210  PRINT
220  PRINT "PRESENT VALUE OF DEDUCTION = $";
225  PRINT  INT (P * 100 + .5) / 100
230  PRINT
240  PRINT "WOULD YOU LIKE TO RE-RUN THIS PROGRAM"
245  PRINT "WITH NEW DATA (Y/N) ";
250  INPUT Z$
260  IF Z$ = "Y" THEN 20
270  IF Z$ <  > "N" THEN 240
280  END
```

Reference

Solomon and Pringle. *An Introduction to Financial Management.* Santa Monica, Calif.: Goodyear Publishing Co., 1977. Pages 376-78.

Future Value of an Investment (Uneven Cash Flow)

Often it is useful to project the future (or terminal) value of monies to be received from an investment. The accept/reject criterion stipulates you should reject any investment whose future value of all cash flows, including the initial investment, is less than zero. This program computes that value, based on the term (in years), the growth rate, and the cash flow amounts for each year. The growth rate should be the rate at which you have alternative opportunities to invest.

Example

Aunt Lonna wants to start a college fund for her nephew, Brian. She plans to put $200.00 into savings this year, $350.00 next year, and $250.00 the following year. The interest rate is 6%. What will Brian's fund be worth at the end of the third year?

Answer: Brian's fund will be worth $845.72.

```
FUTURE VALUE OF AN INVESTMENT

  NUMBER OF CASH FLOWS ?3
       GROWTH RATE (%) ?6

(ENTER INFLOWS AS POSITIVE,
 OUTFLOWS AS NEGATIVE)
AMOUNT OF CASH FLOW 1 ?200
                    2 ?350
                    3 ?250

FUTURE VALUE AT END OF PERIOD 3 = $845.72

DO YOU WANT TO RE-RUN THIS PROGRAM
WITH NEW DATA? (Y/N)?N
```

Practice Problems

1. What will the value of $25,000 be in eight years if another $25,000 is invested in year three and $10,000 is withdrawn during the fifth year? The growth rate is 15%.

   Answer: $101,575.68

2. If the growth rate above was 18%, what would the future value be?

   Answer: $120,400.47

Program Listing

```
10  PRINT "FUTURE VALUE OF AN INVESTMENT"
20  DEF  FN A(X) =  INT (X * 100 + .5) / 100
30  PRINT
40  PRINT "  NUMBER OF CASH FLOWS ";
```

```
50  INPUT N
60  PRINT "        GROWTH RATE (%) ";
70  INPUT R
80 R = R / 100
90  PRINT
100 T = 0
110  PRINT "(ENTER INFLOWS AS POSITIVE,"
115  PRINT " OUTFLOWS AS NEGATIVE)"
120  FOR J = 1 TO N
130  IF J > 1 THEN 160
140  PRINT "AMOUNT OF CASH FLOW ";
150  GOTO 170
160  PRINT "                          ";
170  PRINT J;" ";
180  INPUT C
190  REM  ADD FUTURE VALUES OF EACH YEAR BASED ON RATE OF R
200 T = T +  FN A(C * (1 + R) ^ (N - J))
210  NEXT J
220  PRINT
230  PRINT "FUTURE VALUE AT END OF PERIOD ";N;" = $";T
240  REM  RESTART OF END PROGRAM?
250  PRINT
260  PRINT "DO YOU WANT TO RE-RUN THIS PROGRAM"
265  PRINT "WITH NEW DATA? (Y/N)";
270  INPUT Z$
280  IF Z$ = "Y" THEN 30
290  IF Z$ <  > "N" THEN 260
300  END
```

Reference

Solomon and Pringle. *An Introduction to Financial Management.* Santa Monica, Calif.: Goodyear Publishing, 1977.

Net Present Value of an Investment

Net Present Value (NPV) is defined as the present value of all cash flows associated with an investment, including the initial outlay. The NPV accept/reject criterion for an investment is to accept any investment whose NPV is greater than zero.

To use this program, you first enter the amount of the initial outlay, the term of the investment (in years), the required rate of return, and the cash flow amounts for each year.

Program Notes

To obtain the present value of an investment, enter an initial investment of zero.

Example

Jack has an investment opportunity that requires an initial investment of $10,000 and offers cash returns of $3,000, $5,000, and $4,000 over the next three years. Jack wants at least 15% return on his money. What is the NPV of this investment? Should Jack accept?

Answer: The NPV of this investment is −$980.52. Jack should not accept.

```
NET PRESENT VALUE

INVESTMENT ?10000
NUMBER OF YEARS ?3
REQUIRED RATE OF RETURN (%) ?15

ENTER CASH FLOW AMOUNTS EACH YEAR
(ENTER OUTFLOWS AS NEGATIVE).

INFLOW FOR YEAR 1 ?3000
               2 ?5000
               3 ?4000

NET PRESENT VALUE = $ -980.52

DO YOU WANT TO RE-RUN THIS PROGRAM
WITH NEW DATA; (Y/N)?N
```

Practice Problems

1. Doris holds a note for $1,000.00 which matures in two years, but she wants to invest that money now in new sound equipment. Her bank will buy the note at a 10% discount. What price is the bank offering? (Hint: This is a *present* value calculation.)

Answer: The bank will pay Doris $826.45 for the note.

2. What is the NPV of a $1,500 investment which offers returns of $800.00 year 1, $900.00 year 2, requires $1,000 more to be invested year 3, returns $900.00 year 4, and $800.00 year 5? Comparable five-year investments currently offer a 15% return.

Answer: The NPV of this investment is $130.98, quite acceptable.

Program Listing

```
10  PRINT "NET PRESENT VALUE"
20  DEF  FN A(X) =  INT (X * 100 + .5) / 100
30  REM  ADD DIM C(N) STATEMENT AT LINE 40
35  REM  IF MAXIMUM NUMBER OF CASH FLOWS IF > 10
40  REM
50  PRINT
60  PRINT "INVESTMENT ";
70  INPUT CO
80 CO =  -  FN A(CO)
90  PRINT "NUMBER OF YEARS ";
100  INPUT N
110  PRINT "REQUIRED RATE OF RETURN (%) ";
120  INPUT R
130 R = R / 100 + 1
140 F = 0
150  PRINT
160  PRINT "ENTER CASH FLOW AMOUNTS EACH YEAR"
165  PRINT "(ENTER OUTFLOWS AS NEGATIVE)."
170  PRINT
180  FOR J = 1 TO N
190  IF J > 1 THEN 220
200  PRINT "INFLOW FOR YEAR ";
210  GOTO 230
220  PRINT "                  ";
230  PRINT J;" ";
240  INPUT C(J)
260  NEXT J
270 T = CO
280  REM  ADD PRESENT VALUES FOR EACH YEAR BASED ON RATE OF R
290  FOR J = 1 TO N
300 T = T +  FN A(C(J) / (R ^ J))
310  NEXT J
320  PRINT
330  PRINT "NET PRESENT VALUE = $ "; FN A(T)
340  REM  RESTART OF END PROGRAM?
350  PRINT
360  PRINT "DO YOU WANT TO RE-RUN THIS PROGRAM"
365  PRINT "WITH NEW DATA; (Y/N)";
370  INPUT Z$
380  IF Z$ = "Y" THEN 50
390  IF Z$ <  > "N" THEN 360
400  END
```

References

Rosen, Lawrence R. *Dow Jones-Irwin Guide to Interest.* Homewood, Ill.: Dow Jones-Irwin, Inc., 1974.

Solomon and Pringle. *An Introduction to Financial Management.* Santa Monica, Calif.: Goodyear Publishing, 1977. Pages 261-62.

Lease/Buy Decision

This program computes the present value of the cost to lease, and the present value of the cost to buy. Any difference between those amounts is the advantage of leasing or of buying. It is assumed that the asset would be financed over the same period of time that it would be leased.

To use the program, enter the price of the asset, the interest rate, the term in years, the salvage value at the end of that term, the tax rate, annual amount of loan payments, and the annual amount of lease payments. The program outputs the present value of the cost to buy, the present value of the cost to lease, and the difference between those amounts.

While this program may be instructive in pointing out decision factors you may have overlooked, it is not meant to replace your judgment. Capital planning requirements and lease/loan terms must ultimately guide your decision. In general, depreciation and salvage value reduce the cost of buying. However, if an asset is subject to rapid obsolescence, leasing may be the less expensive choice.

Program Notes

This program is actually a modified version of the Net Present Value of an Investment program. As such, you may find it instructive of modifications you may make to any of the programs in this book.

Example

Acme Landscaping has need for a small truck for everyday use. They are considering buying a truck for $6,000. Salvage value after four years is estimated to be $2,000. The bank will lend $6,000 at 16% interest to be repaid in four equal installments of $2,145. The lease will cost $2,000 per year. Taxes are 40%, and straight-line depreciation of $1,000 per year will be used. What is the present value of the cost to buy? What is the present value of the cost to lease? Should Acme lease or buy?

Answer: The present value of the loan is $3,011.90. The present value of the lease is $3,357.82. Acme should buy the truck.

```
LEASE/BUY DECISION

ENTER THE COST TO ACQUIRE ASSET
(PRINCIPAL OF LOAN) ?6000
ENTER THE INTEREST RATE (%) ?16
ENTER THE TERM IN YEARS ?4
WHAT IS THE SALVAGE VALUE
AT THE END OF 4 YEARS ?2000

WHAT IS THE TAX RATE (%) ?40
ENTER THE ANNUAL AMOUNT
OF LOAN PAYMENTS ?2145
ENTER THE ANNUAL AMOUNT
OF LEASE PAYMENTS ?2000

ENTER THE DEPRECIATION AMOUNT
FOR EACH YEAR

YEAR NUMBER 1 ?1000
            2 ?1000
```

```
              3 ?1000
              4 ?1000

PRESENT VALUE OF COST OF LOAN =$3011.9
PRESENT VALUE OF COST OF LEASE =$3357.82

ADVANTAGE OF BUYING =$345.92

WOULD YOU LIKE TO RE-RUN THIS PROGRAM
WITH NEW DATA (Y/N)?N
```

Practice Problems

1. In the above example, what if the lease is $1,200 per year?
Answer: Leasing would be the best choice. The present value of the lease would be $2,014.69. The leasing advantage would be $997.21.

2. Industrial Supply Company needs a computer for their in-house use. The model they want will cost $30,000, to be financed at 17% interest over five years. After five years ISC plans to sell the computer for $10,000 and buy a larger model. The tax rate is 48%, annual loan payments will be $9,375.00, and a five-year lease on the equipment would cost $3,500.00 per year. Depreciation would be $6,000.00 the first year, $5,000 year 2, $4,000 year 3, $3,000 year 4, and $2,000 year 5. What is the advantage of leasing or buying?
Answer: ISC would realize an advantage of $7,362.24 if they leased the new computer.

Program Listing

```
10  PRINT "LEASE/BUY DECISION"
20  REM   - FUNCTION TO ROUND TO NEAREST HUNDREDTH
30  DEF  FN A(X) =  INT (X * 100 + 0.5) / 100
40  PRINT
50  PRINT "ENTER THE COST TO ACQUIRE ASSET"
55  PRINT "(PRINCIPAL OF LOAN) ";
60  INPUT B1
70  PRINT "ENTER THE INTEREST RATE (%) ";
80  INPUT I1
90  REM  - CONVERT INTEREST RATE TO DECIMAL
100 I1 = I1 / 100
110  PRINT "ENTER THE TERM IN YEARS ";
120  INPUT Y1
130  PRINT "WHAT IS THE SALVAGE VALUE"
135  PRINT "AT THE END OF ";Y1;" YEARS ";
140  INPUT S1
150  PRINT
160  PRINT "WHAT IS THE TAX RATE (%) ";
170  INPUT R1
180  REM  - CONVERT TAX RATE TO DECIMAL
190 R1 = R1 / 100
200  PRINT "ENTER THE ANNUAL AMOUNT"
205  PRINT "OF LOAN PAYMENTS ";
210  INPUT A1
220  PRINT "ENTER THE ANNUAL AMOUNT"
225  PRINT "OF LEASE PAYMENTS ";
230  INPUT A2
```

```
240  REM  - RESET TOTAL AMOUNTS TO ZERO
250 T1 = 0
260 L1 = 0
270  PRINT
280  PRINT "ENTER THE DEPRECIATION AMOUNT"
285  PRINT "FOR EACH YEAR"
290  PRINT
300  REM  - LOOP TO INPUT, CALCULATE, AND ACCUMULATE
305  REM  - VALUES EACH YEAR
310  FOR Z = 1 TO Y1
320  IF Z > 1 THEN 350
330  PRINT "YEAR NUMBER ";
340  GOTO 360
350  PRINT "             ";
360  PRINT Z;" ";
370  INPUT D1
380  REM  - CALCULATE INTEREST AMOUNT FOR EACH YEAR
390 B0 =  ABS (B1 -  FN A(B1 * (1 + I1)))
400  REM  - CONVERT D1 TO PRESENT VALUE OF COST
405  REM  - OF OWNING EACH YEAR
410 D1 =  FN A((A1 -  FN A((D1 + B0) * R1)) / ((1 + I1) ^ Z))
420  REM  - SUBTRACT ANNUAL PAYMENT,
425  REM  - ADD ANNUAL INTEREST TO PRINCIPAL
430 B1 = B1 - A1 + B0
440  REM  - SUM PRESENT VALUE AMOUNTS OF EACH YEAR
450 T1 = T1 + D1
460  REM  - COMPUTE PRESENT VALUE OF COST TO LEASE FOR EACH YEAR
470 L1 = L1 +  FN A((A2 - (A2 * R1)) / (1 + I1) ^ Z)
480  NEXT Z
490  REM  - SUBTRACT PRESENT VALUE OF SALVAGE VALUE
495  REM  - FROM TOTAL COST TO OWN
500 T1 = T1 -  FN A(S1 / (1 + I1) ^ Y1)
510  REM  - OUTPUT RESULTS
520  PRINT
530  PRINT "PRESENT VALUE OF COST OF LOAN =$"; FN A(T1)
540  PRINT "PRESENT VALUE OF COST OF LEASE =$"; FN A(L1)
550  PRINT
560  IF L1 < T1 THEN 590
570  PRINT "ADVANTAGE OF BUYING =$"; FN A(L1 - T1)
580  GOTO 600
590  PRINT "ADVANTAGE OF LEASING =$"; FN A(T1 - L1)
600  PRINT
610  REM  - RESTART OR END PROGRAM?
620  PRINT "WOULD YOU LIKE TO RE-RUN THIS PROGRAM"
625  PRINT "WITH NEW DATA (Y/N)";
630  INPUT Z$
640  IF Z$ = "Y" THEN 40
650  IF Z$ <  > "N" THEN 620
660  END
```

Reference

Chase and Aquilano. *Production and Operations Management.* Homewood, Ill.: Richard D. Irwin, Inc., 1977. Pages 138-40.

Syndicated Investment Analysis

This program evaluates tax savings and net cash flows from an investment by a syndicate, or group of investors, to a participating investor. The program considers the investor's tax bracket, as well as the proportion of the original investment, participation in cash income, taxable income/loss, and tax credits.

To use this program, enter the length of the analysis in years and the first year of syndication. Then, for each year, enter the cash income for the syndicate, followed by its taxable income. Enter the year (1,2, and so forth) and total investment for that year by the syndicate. Then, enter the year and amount of investment or other tax credits (entered as a negative number), or credit recapture (entered as a positive number). Next, enter allocation percentages for the investor: percentage of total investment, cash, income, and taxable income (or loss) and credits. The final entry is the investor's tax bracket, entered as a percentage.

The program then prints its analysis, which shows the investor his/her original investment, cash income, taxable income, tax saving (tax savings are negative; tax paid is positive), net end-of-year cash flow and cumulative net cash flows. You may repeat the analysis for different tax brackets when the program asks for a new tax bracket to consider. (All other investment factors remain as you last entered them.) Enter a tax bracket of 999 to respecify the percentage allocations. Enter an investment allocation percentage of 999 to end the program.

Program Notes

The program is set for 40 years of projections. You can change this amount by modifying line 20 as follows:

```
20 N9 = I
```

Make sure that you replace the expression I with a constant equal to the maximum number of years.

Example

Consider this syndicated investment: An income property with a $35,000 down payment which will generate $4,500 cash over the first four years, $5,200 over the next four years, and $5,500 over the remaining five years. The investment earns a $3,500 investment tax credit in the first year. Taxable income will start at −$3,800 and increase by $1,100 per year for the life of the investment.

The investor is in the 55% tax bracket, and is contributing 30% of the original cash outlay. Participation is 30% on cash income and taxable income. How will this investor run the program?

Answer: The printout below shows the investor's portion of cash income, tax savings, net and cumulative cash flow. At the end of the investment projection, cumulative cash to this investor is $4,432, and the investment is sheltered until the end of 1985, when a tax on $109 must be paid.

```
SYNDICATED INVESTMENT ANALYSIS

FOR HOW MANY YEARS DO YOU WANT
THIS PROJECTION (LIMIT: 40) ?13

ENTER THE FIRST YEAR OF
SYNDICATION (E.G. 1981) ?1980
```

```
FOR ENTIRE SYNDICATE, ENTER CASH INCOME
FOR EACH YEAR OF PROJECTION
YEAR 1 CASH INCOME = ?4500
YEAR 2 CASH INCOME = ?4500
YEAR 3 CASH INCOME = ?4500
YEAR 4 CASH INCOME = ?4500
YEAR 5 CASH INCOME = ?5200
YEAR 6 CASH INCOME = ?5200
YEAR 7 CASH INCOME = ?5200
YEAR 8 CASH INCOME = ?5200
YEAR 9 CASH INCOME = ?5500
YEAR 10 CASH INCOME = ?5500
YEAR 11 CASH INCOME = ?5500
YEAR 12 CASH INCOME = ?5500
YEAR 13 CASH INCOME = ?5500

FOR ENTIRE SYNDICATE, ENTER TAXABLE
INCOME FOR EACH YEAR OF PROJECTION
POSITIVE FOR INCOME NEGATIVE FOR LOSS
YEAR 1 TAXABLE = ?-3800
YEAR 2 TAXABLE = ?-2700
YEAR 3 TAXABLE = ?-1600
YEAR 4 TAXABLE = ?-500
YEAR 5 TAXABLE = ?600
YEAR 6 TAXABLE = ?1700
YEAR 7 TAXABLE = ?2800
YEAR 8 TAXABLE = ?3900
YEAR 9 TAXABLE = ?5000
YEAR 10 TAXABLE = ?6100
YEAR 11 TAXABLE = ?7200
YEAR 12 TAXABLE = ?8300
YEAR 13 TAXABLE = ?9400

ENTER YEAR OF VENTURE (1, 2, ETC.) AND
AMOUNT OF INVESTMENT BY ENTIRE GROUP
OF INVESTORS THAT YEAR.  AFTER LAST
YEAR, ENTER 99999,0
?1,35000
?99999,0

ENTER YEAR OF VENTURE (1, 2, ETC.) AND
AMOUNT OF INVESTMENT CREDIT OF OTHER
SIMILAR CREDIT FOR ENTIRE SYNDICATE
(AS NEGATIVE),AND CREDIT RECAPTURE
(AS POSITIVE ) FIGURE.  AFTER LAST
ENTRY, ENTER 99999,0
?1,-3500
?99999,0

ENTER PERCENTAGE ALLOCATIONS (0-100%)
FOR THIS INVESTOR...
PCT. OF INVESTMENT (999=END) ?30
      PCT. OF CASH INCOME ?30
   PCT. OF TAXABLE INCOME
   (OR LOSS), AND CREDITS ?30
```

```
ENTER TAX BRACKET
(999=CHANGE ALLOCATIONS) ?55
RESULTS FOR INVESTOR IN 55
% TAX BRACKET

YEAR  INVEST-  CASH   TAX    NET CUMULATIVE
      MENT  INCOME SAVING CASH   CASH

1980  10500  1350  -1677  -7473  -7473
1981  0      1350  -446   1796   -5677
1982  0      1350  -264   1614   -4063

1983  0      1350  -82    1432   -2631
1984  0      1560  99     1461   -1170
1985  0      1560  281    1279   109

1986  0      1560  462    1098   1207
1987  0      1560  644    916    2123
1988  0      1650  825    825    2948

1989  0      1650  1007   643    3591
1990  0      1650  1188   462    4053
1991  0      1650  1370   280    4333

1992  0      1650  1551   99     4432
THIS SCHEDULE DISREGARDS MINIMUM TAX,
DISALLOWANCE OF INVESTMENT INTEREST
EXPENSE, CODE SEC.183, ETC.


ENTER TAX BRACKET
(999=CHANGE ALLOCATIONS) ?999

ENTER PERCENTAGE ALLOCATIONS (0-100%)
FOR THIS INVESTOR...
PCT. OF INVESTMENT (999=END) ?999
```

Practice Problems

1. Alvin wants to start a musical career with his brothers Simon and Theodore. Alvin is in the 40% tax bracket. He will contribute 45% of the $30,000 needed to build a recording studio. He will participate 20% in the cash earnings, and 45% in the taxable earnings of the company. Alvin expects that the studio will generate $8,000 cash per year for the first two years. A further investment of $15,000 will come up in year 3 for new equipment. The studio's taxable earnings will start at $4,200, increasing by $1,000 each year. Cash income for the recording studio will increase to $12,000 per year from year 3 to year 10 (the last year of projection).

What will Alvin's cumulative cash flow be from this investment? In what year will Alvin have to start paying taxes on his share of the investment? Assume that the studio will earn a 10% investment tax credit for the initial cash outlay as well as the $15,000 in year 3.

Answer: Alvin's cumulative cash flow will be $3,635 at the end of year 10. Assuming the first year is 1980, Alvin will have to start paying taxes on this investment in 1985 ($144).

2. Fred wants to start a helicopter tour service. He is in the 65% tax bracket, and will participate in all aspects of the syndicate at 51%. The initial investment for a four-passenger helicopter is $12,500. Fred plans on trading up to a six-passenger helicopter after three years. The group will receive a $6,500 tax credit in year 1. If they trade up in year 3, they will receive an $8,500 tax credit, and will have to invest another $19,000. They will sell the four-passenger helicopter in year 4, losing $4,167 from credit recapture. Cash income will start at $40,000 per year, growing to $48,000 per year at the start of year 3, up until year 8 (the final year of projection). Taxable income starts at −$9,000, growing by $2,000 every year.

What will the total cumulative cash flow be for the eight years of projection? How will the credit recapture affect him in year 4?

Answer: Total cumulative cash flow will be $182,441. Fred will have to pay $1,131 in taxes in year 4, due to the credit recapture.

Program Listing

```
1  PRINT "SYNDICATED INVESTMENT ANALYSIS"
2  PRINT
9  REM  ROUND-OFF FUNCTION
10  DEF  FN R(X) =  INT (X + 0.5)
18  REM  N9 = MAXIMUM YEARS FOR PROJECTION
19  REM        AND MAXIMUM DIMENSION FOR LINE 30
20 N9 = 40
30  DIM C(N9),J(N9),T(N9),U(N9)
200  PRINT "FOR HOW MANY YEARS DO YOU WANT"
210  PRINT "THIS PROJECTION (LIMIT: ";N9;") ";
220  INPUT Y
225  IF Y > N9 THEN 200
230  PRINT
240  PRINT "ENTER THE FIRST YEAR OF"
250  PRINT "SYNDICATION (E.G. 1981) ";
260  INPUT Y1
270  PRINT
280  PRINT "FOR ENTIRE SYNDICATE, ENTER CASH INCOME"
290  PRINT "FOR EACH YEAR OF PROJECTION "
300  FOR I = 1 TO Y
310  PRINT "YEAR ";I;" CASH INCOME = ";
320  INPUT C(I)
340  NEXT I
350  PRINT
360  PRINT "FOR ENTIRE SYNDICATE, ENTER TAXABLE"
370  PRINT "INCOME FOR EACH YEAR OF PROJECTION"
380  PRINT "POSITIVE FOR INCOME NEGATIVE FOR LOSS"
390  FOR I = 1 TO Y
400  PRINT "YEAR ";I;" TAXABLE = ";
410  INPUT T(I)
430  NEXT I
440  PRINT
450  PRINT "ENTER YEAR OF VENTURE (1, 2, ETC.) AND"
460  PRINT "AMOUNT OF INVESTMENT BY ENTIRE GROUP"
470  PRINT "OF INVESTORS THAT YEAR.  AFTER LAST"
480  PRINT "YEAR, ENTER 99999,0"
490  INPUT I,XO
500  IF I = 99999 THEN 530
505 J(I) = XO
520  GOTO 490
```

```
530  PRINT
540  PRINT "ENTER YEAR OF VENTURE (1, 2, ETC.) AND"
550  PRINT "AMOUNT OF INVESTMENT CREDIT OF OTHER
560  PRINT "SIMILAR CREDIT FOR ENTIRE SYNDICATE"
570  PRINT "(AS NEGATIVE),AND CREDIT RECAPTURE"
580  PRINT "(AS POSITIVE ) FIGURE.  AFTER LAST"
590  PRINT "ENTRY, ENTER 99999,0"
600  INPUT I,XO
610  IF I = 99999 THEN 640
615 U(I) = XO
630  GOTO 600
640  PRINT
645  PRINT "ENTER PERCENTAGE ALLOCATIONS (0-100%)
650  PRINT "FOR THIS INVESTOR..."
655  PRINT "PCT. OF INVESTMENT (999=END) ";
660  INPUT P1
665  IF P1 > 998 THEN 2170
670 P1 = P1 / 100
675  PRINT "      PCT. OF CASH INCOME ";
680  INPUT P2
685 P2 = P2 / 100
690  PRINT "   PCT. OF TAXABLE INCOME "
693  PRINT "   (OR LOSS), AND CREDITS ";
695  INPUT P3
700 P3 = P3 / 100
705  PRINT
710  PRINT "ENTER TAX BRACKET"
715  PRINT "(999=CHANGE ALLOCATIONS) ";
720  INPUT T1
725  IF T1 > 998 THEN 640
750  PRINT "RESULTS FOR INVESTOR IN ";T1
753  PRINT "% TAX BRACKET"
755 T1 = T1 / 100
760  PRINT
770  PRINT "YEAR"; TAB( 6);"INVEST-"; TAB( 14);"CASH"; TAB( 20);
775  PRINT "TAX"; TAB( 26);"NET"; TAB( 30);"CUMULATIVE"
780  PRINT  TAB( 7);"MENT"; TAB( 12);"INCOME"; TAB( 19);"SAVING";
785  PRINT  TAB( 26);"CASH"; TAB( 33);"CASH"
800  PRINT
810 S1 = 0
820  FOR I = 1 TO Y
850 K =  FN R(P1 * J(I))
870 D =  FN R(P2 * C(I))
890 V =  FN R(P3 * T(I) * T1 + P3 * U(I))
900 S = D - K - V
910 S1 = S1 + S
920  PRINT Y1 + I - 1; TAB( 6);K; TAB( 13);D; TAB( 19);
925  PRINT V; TAB( 26);S; TAB( 33);S1
940  IF I / 3 <  >  INT (I / 3) THEN 960
950  PRINT
960  NEXT I
2099  REM  PRINT DISCLAIMER/BLANK LINES
2100  PRINT "THIS SCHEDULE DISREGARDS MINIMUM TAX,"
2110  PRINT "DISALLOWANCE OF INVESTMENT INTEREST"
2120  PRINT "EXPENSE, CODE SEC.183, ETC."
```

```
2130  PRINT
2140  PRINT
2150  PRINT
2160  GOTO 710
2170  END
```

Depreciation Switch

An accelerated depreciation method provides for greatest depreciation in the earlier years. At some point, switching to a straight-line depreciation will allow a larger amount to be depreciated in later years than could be done by continuing to use the accelerated method.

Calculations are made using a fixed cost of $1 million. The actual cost of the asset involved is unimportant. The million-dollar cost serves only to separate close calculations. Enter the depreciation method to use for this asset, in percent (that is, 125, 150, 200, and so forth); the useful life of the asset, in years; and the number of months of depreciation the first year of the useful life (a full first year should be entered as 12 months).

Example

Champion Products acquired a plastic injection machine that has a useful life of five years. Six months' depreciation remains in this fiscal year, and Champion plans to use 200% declining balance depreciation. When should they switch from declining balance method to straight-line depreciation in order to maximize the amounts depreciated?

Answer: Champion should switch methods in the fifth year.

```
DEPRECIATION SWITCH

ENTER METHOD, IN PERCENT (0=END) ?200
ENTER USEFUL LIFE OF ASSET,
IN YEARS ?5
ENTER NUMBER OF MONTHS DEPRECIATION
LEFT IN FIRST YEAR ?6

YEAR OF SWITCH = 5


ENTER METHOD, IN PERCENT (0=END) ?0
```

Practice Problems

1. In the above example, what if 12 months of depreciation remains in the current fiscal year?
Answer: The switch should be effected in the fourth year.

2. Using 150% depreciation, when should an asset with an eight-year life be depreciated by the straight-line method, assuming a full year's depreciation remains in the first year?
Answer: The switch to straight-line should be made in the fourth year.

Program Listing

```
10  PRINT "DEPRECIATION SWITCH"
20  REM  - USE MILLION DOLLAR COST TO
30  REM  - SEPARATE CLOSE CALCULATIONS
40 C = 1E + 6
```

```
50  REM  - RESET ACCUMULATED DEPRECIATION TO ZERO
60 A = 0
70  PRINT
80  PRINT "ENTER METHOD, IN PERCENT (0=END) ";
90  INPUT T
100  IF T = 0 THEN 350
110 T = T / 100
120  PRINT "ENTER USEFUL LIFE OF ASSET,"
125  PRINT "IN YEARS ";
130  INPUT L
140  IF L >  = 3 THEN 170
150  PRINT "LIMIT 3 YEARS MINIMUM LIFE,"
155  PRINT "PLEASE RE-ENTER."
160  GOTO 120
170  PRINT "ENTER NUMBER OF MONTHS DEPRECIATION"
175  PRINT "LEFT IN FIRST YEAR ";
180  INPUT M
190 Y = 1
200  REM  - CALCULATE DEPRECIATION ACCUMULATED IN THE FIRST YEAR
210 A =  INT (((M / 12) * (T / L) * C) * 100 + 0.5) / 100
220 Y = Y + 1
230  REM  - COMPUTE AMOUNT OF DEPRECIATION THIS YEAR
240 D =  INT (((T / L) * (C - A)) * 100 + 0.5) / 100
250  REM  - IF DEPRECIATION IS LESS THAN VALUE
260  REM  - DIVIDED BY REMAINING LIFE, PRINT YEAR NUMBER
270  IF D < (C - A) / (L - Y + 1 + (12 - M) / 12) THEN 310
280  REM  - IF NOT, INCREMENT ACCUMULATED DEPRECIATION
290 A = A + D
300  GOTO 220
310  PRINT
320  PRINT "YEAR OF SWITCH = ";Y
330  PRINT
340  GOTO 60
350  END
```

References

U.S. Internal Revenue Service Code, Section 167(b) and Section 167(e)(1).

U.S. Treasury Department, Internal Revenue Service. Regulations, Sections 1.167(b)-0, 1.167(b)-1, 1.167(b)-2, and 1.167(e)-1.

Apportionment by Ratios

This program divides a quantity into the proportion that each of a group of numbers bears to the sum of that group. You are first asked for the number of decimal places that you wish shown from whole numbers down to 13 decimal places (if your computer is that accurate). You then enter the value to be apportioned, and the number of parts into which it is to be divided. You then enter each component of the group to be used as the basis for apportionment. The program prints out a table that shows each of these amounts, the percentage each is of the group total, and the corresponding apportioned amount. At the conclusion, it prints the totals of these three columns.

Example

Ten employees at Widgets, Inc., are receiving bonuses from a $30,000 pool. If each receives a share proportionate to his salary, how much does each one get?

Name	Salary
Abelson	$54,000
Boucher	$47,000
Charleston	$40,000
Dryden	$33,500
Evans	$29,750
Freisner	$26,000
Goodine	$24,500
Holloway	$21,000
Ishikawa	$17,500
Johnson	$15,000

Answer:

```
APPORTIONMENT BY RATIOS
ENTER THE NUMBER OF DECIMAL
PLACES OF ROUNDING YOU WANT:
0 FOR WHOLE NUMBERS, 1 FOR TENTHS, ETC.
UP TO 9.
?2
ENTER TOTAL TO BE APPORTIONED
?30000
ENTER NUMBER OF PORTIONS
?10
ENTER AMOUNT 1
?54000
ENTER AMOUNT 2
?47000
ENTER AMOUNT 3
?40000
ENTER AMOUNT 4
?33500
ENTER AMOUNT 5
?29750
ENTER AMOUNT 6
?26000
ENTER AMOUNT 7
```

```
?24500
ENTER AMOUNT 8
?21000
ENTER AMOUNT 9
?17500
ENTER AMOUNT 10
?15000
         AMOUNT       PERCENT      APPORTIONED

         54000        17.52        5255.47
         47000        15.25        4574.21
         40000        12.98        3892.94
         33500        10.87        3260.34
         29750        9.65         2895.38
         26000        8.43         2530.41
         24500        7.95         2384.43
         21000        6.81         2043.8
         17500        5.68         1703.16
         15000        4.86         1459.86

TOTALS 308250         100.00       30000

LAST ITEM ADUSTED WHERE NECESSARY
```

Practice Problems

1. A mayor running for re-election wants to divide his campaign workers among the city's six districts based on the population of each district. He has 42 campaign workers, and the districts are populated as follows: District 1: 29,842; District 2: 17,420; District 3: 14,625; District 4: 24,314; District 5: 21,209; District 6: 18,956. How many workers should he place in each district?

Answer: District 1: 10; District 2: 6; Disrict 3: 5; District 4: 8; District 5: 7; District 6: 6.

2. A winery has 120 bottles of wine that it wants to distribute among its employees. If the wine is divided in proportion to each employee's seniority, how much wine does each employee get?

Name	Years Employed
Jones	22
Romero	18
Lippitt	14
Doyle	8
Peterson	4
Covey	2
Miller	2
Bennett	1

Answer: Jones: 37 bottles; Romero: 30 bottles; Lippitt: 24 bottles; Doyle: 14 bottles; Peterson: 7 bottles; Covey: 3 bottles; Miller: 3 bottles; Bennett: 2 bottles.

Program Listing

```
10  PRINT "APPORTIONMENT BY RATIOS"
20  DIM A(100)
30  PRINT "ENTER THE NUMBER OF DECIMAL"
40  PRINT "PLACES OF ROUNDING YOU WANT:"
50  PRINT "0 FOR WHOLE NUMBERS, 1 FOR TENTHS, ETC."
```

```
60  PRINT "UP TO 9."
70  INPUT R1
80  PRINT "ENTER TOTAL TO BE APPORTIONED"
90  INPUT S2
100  PRINT "ENTER NUMBER OF PORTIONS"
110  INPUT N
119  REM  ENTER RATIO AMOUNTS ONE BY ONE
120  FOR I = 1 TO N
130  PRINT "ENTER AMOUNT ";I
140  INPUT A(I)
150 S1 = S1 + A(I)
160  NEXT I
170  PRINT  TAB( 8);"AMOUNT"; TAB( 19);"PERCENT"; TAB( 30);"APPORTIONED"
180  PRINT
190  FOR I = 1 TO N - 1
200 P =  INT (10000 * A(I) / S1 + 0.5) / 100
210 P1 = P1 + P
220 R =  INT ((S2 * A(I) / S1) * 10 ^ (R1) + 0.5) / 10 ^ (R1)
230 S3 = S3 + R
240  PRINT  TAB( 8);A(I); TAB( 19);P; TAB( 30);R
250  NEXT I
252 PR =  INT ((100 - P1) * 10 ^ (R1) + 0.5) / 10 ^ (R1)
254 SR =  INT ((S2 - S3) * 10 ^ (R1) + 0.5) / 10 ^ (R1)
260  PRINT  TAB( 8);A(N); TAB( 19);PR; TAB( 30);SR
270  PRINT
280  PRINT "TOTALS"; TAB( 8);S1; TAB( 19);"100.00"; TAB( 30);S2
290  PRINT
300  PRINT "LAST ITEM ADJUSTED WHERE NECESSARY"
310  END
```

Internal Rate of Return

Internal Rate of Return (IRR) is the rate at which the sum of all cash flows discount to the amount of the initial investment. This program finds the rate by using a half-interval search.

To use the program, enter the amount of the initial investment, then the term of the investment (in years), and the cash flow amount for each year. Enter outflows (funds you invest) as negative numbers. Enter an initial investment of zero to end the program.

IRR can also be used to compute the yield to maturity of a bond by entering the price of the bond as the initial investment, the number of years to maturity as the term, coupon amounts for each year they will be received as the cash flow amounts for those years (enter the total amount to be received in each year), and coupon amount(s) plus the maturity value of the bond in the last year (when the bond will mature). The IRR returned by the program is the yield to maturity of the bond.

Program Notes

The half-interval search at lines 320 to 540 will find rates of return between 0% and 99%. If this range is not wide enough to suit your needs, change the initial values of variable L at line 330 and H at line 340. These are the low and high search limits. Make sure that upon the first execution of line 370, the value of (L+H)/2 is not zero, as that will cause premature exit from the search algorithm.

Example

Bob T. has an opportunity to invest in a venture. An initial investment of $10,000 is needed, with cash returns of $4,000, $5,000, and $3,000 over the next three years. His required rate of return is 15%. Should Bob accept this investment?

Answer: No. The IRR of this investment is 10.1331%. The accept/reject criterion stipulates rejection of any investment whose IRR is less than the required rate of return.

```
INTERNAL RATE OF RETURN

ENTER THE AMOUNT OF THE INITIAL
INVESTMENT (0 TO END) ?10000

NUMBER OF CASH FLOW PERIODS ?3

(ENTER INFLOWS AS POSITIVE,
OUTFLOWS AS NEGATIVE AMOUNTS)
CASH FLOW FOR PERIOD 1 ?4000
                     2 ?5000
                     3 ?3000


INTERNAL RATE OF RETURN = 10.1331%


ENTER THE AMOUNT OF THE INITIAL
INVESTMENT (0 TO END) ?0
```

Practice Problem

A new bond issue offers a coupon rate of 8.25% and matures in 7 years. What is the yield to maturity of a $10,000 bond if the price is $8,500?

Answer: The yield to maturity is 11.4831%.

Program Listing

```
10  PRINT "INTERNAL RATE OF RETURN"
20  REM  FUNCTION TO ROUND TO NEAREST HUNDREDTH
30  DEF  FN A(X) =  INT (X * 100 + 0.5) / 100
40  REM   FUNCTION TO ROUND TO NEAREST TEN-THOUSANDTH
50  DEF  FN B(X) =  INT (X * 1E4 + 0.5) / 1E4
60  REM  CHANGE SIZE OF ARRAY C() IF NECESSARY
70  DIM C(12)
80  PRINT
90  PRINT "ENTER THE AMOUNT OF THE INITIAL"
95  PRINT "INVESTMENT (0 TO END) ";
100  INPUT I
110  REM  END PROGRAM?
120  IF I = 0 THEN 590
130  PRINT
140  PRINT "NUMBER OF CASH FLOW PERIODS ";
150  INPUT N
160  REM  RESTART IF NUMBER OF CASH FLOW PERIODS IS INVALID
170  IF N < 1 THEN 80
180  REM  LOOP TO INPUT AND SUM CASH FLOW AMOUNT(S)
190 F = 0
200  PRINT
210  PRINT "(ENTER INFLOWS AS POSITIVE,
215  PRINT "OUTFLOWS AS NEGATIVE AMOUNTS)"
220  FOR J = 1 TO N
230  IF J > 1 THEN 260
240  PRINT "CASH FLOW FOR PERIOD ";
250  GOTO 270
260  PRINT "                          ";
270  PRINT J;" ";
280  INPUT C(J)
300  NEXT J
310  PRINT
320  REM  INITIALIZE VALUES
330 L = 0
340 H = 1
350 R1 = 0
360  REM  GUESS RATE = (HIGH RATE + LOW RATE) / 2
370 R = (L + H) / 2
380  REM  EXIT IF RATE REMAINS UNCHANGED
390  IF R = R1 THEN 550
400  REM  SET LAST GUESS TO CURRENT GUESS
410 R1 = R
420  REM  ADD PRESENT VALUES FOR EACH YEAR BASED ON RATE OF
430 T = 0
440  FOR J = 1 TO N
```

```
450 T = T +  FN A(C(J) / ((R + 1) ^ J))
460  NEXT J
470  REM  IF TOTAL PRESENT VALUES EQUAL INVESTMENT, EXIT
480  IF T = I THEN 550
490  REM  SET HIGH OF LOW RATE TO CURRENT GUESS
500  IF I > T THEN 530
510 L = R
520  GOTO 370
530 H = R
540  GOTO 370
550  PRINT
560  PRINT "INTERNAL RATE OF RETURN = "; FN B(R * 100);"%"
570  PRINT
580  GOTO 80
590  END
```

References

Chase and Aquilano. *Production and Operations Management.* Homewood, Ill.: Richard D. Irwin, Inc., 1977. Pages 131-32.

Rosen, Lawrence R. *The Dow Jones-Irwin Guide to Interest.* Homewood, Ill.: Dow Jones-Irwin, 1974.

Solomon and Pringle. *An Introduction to Financial Management.* Santa Monica, Calif.: Goodyear Publishing, 1977. Pages 257-61.

Financial Management Rate of Return

Financial Management Rate of Return (FMRR) differs from Internal Rate of Return in several respects. For some investments, particularly real estate ventures, FMRR will provide a more realistic value than IRR. FMRR assumes only cash flows after financing and taxes are considered, and it ignores the fact that other sources of funds may be available.

To use the program, you enter the term of the investment (in years), then a liquid investment rate. This is a rate at which funds can be invested in any amount, at a guaranteed after-tax rate, and withdrawn as needed (such as a savings account). You also enter a "safe" fixed investment rate. "Safe" means the return on the investment will be at least that high. This investment can be a real estate project or other fixed investment of comparable risk at after-tax rates above the liquid rate, such as certificates of deposit or Treasury bills. The fixed investment should have a minimum amount that can be invested. Enter this amount, too.

The program will indicate points where you will be expected to invest funds in the liquid and fixed investments, the actual initial investment you will need to make (the difference between that amount and the original initial investment must be invested at the fixed rate at the beginning of the first year), the actual total return on the investment, and the rate at which the actual total return discounts to the actual initial investment (the FMRR).

Example

Horatio plans to buy an apartment house. The terms require $10,000 down payment to be made now, and payments of $50,000 to be made next year and the following year. Cash flows indicate that at the end of years 3 and 5, Horatio can expect to receive $30,000 from his investment. He plans to remodel the building during year 4, at an estimated cost of $20,000. Finally, in year 6 he plans to sell the building for $250,000. The liquid investment rate available is 5%, and a minimum $10,000 fixed investment will earn at least 10%. What is the FMRR on Horatio's investment?

Answer: 19.348% (The IRR of this investment is 25.2%.)

```
'FINANCIAL MANAGEMENT' RATE OF RETURN

NUMBER OF YEARS ?6
LIQUID INVESTMENT INTEREST RATE ?5
'SAFE' FIXED INVESTMENT
INTEREST RATE ?10
MINIMUM AMOUNT OF FIXED
INVESTMENT ?10000

(ENTER INFLOWS AS POSITIVE,
OUTFLOWS AS NEGATIVE.)

ENTER CASH FLOW AMOUNT FOR YEAR
          0 ?-10000
          1 ?-50000
          2 ?-50000
          3 ?30000
          4 ?-20000
          5 ?30000
          6 ?250000
```

```
LIQUID INVESTMENT OF $19047
TO BE MADE AT END OF YEAR 3

FIXED INVESTMENT OF $10952
TO BE MADE AT END OF YEAR 3
FIXED INVESTMENT OF $30000
TO BE MADE AT END OF YEAR 5

ACTUAL TOTAL INITIAL INVESTMENT
                = $102971
TOTAL RETURN ON INVESTMENT
                = $297577

'FINANCIAL MANAGEMENT'
RATE OF RETURN = 19.348%

WOULD YOU LIKE TO RE-RUN THIS PROGRAM
WITH NEW DATA? (Y/N) ?N
```

Practice Problems

1. What is the FMRR on a 6-year project if the liquid rate is 7.25%, the fixed rate is 15% (with a minimum investment of $10,000), and the initial investment is $100,000? Cash flows will be $30,000 inflow year 1, $45,000 outflow year 2, and $50,000 inflows during each of the remaining 4 years of the term.

   Answer: The FMRR is 11.783%.

2. On a 4-year investment, requiring $10,000 initially and cash flows of −$2,500, $5,000, −$2,500, and $25,000 during the term, what is the FMRR? The liquid rate is 8.5%, and a minimum $1,000 fixed investment will return at least 13%.

   Answer: The FMRR is 23.303%.

Program Listing

```
10  PRINT "'FINANCIAL MANAGEMENT' RATE OF RETURN"
20  REM  FUNCTION TO ROUND TO NEAREST THOUSANDTH
30  DEF  FN B(X) =  INT (X * 1E3 + 0.5) / 1E3
40  REM  -- CHANGE DIMENSION OF ARRAY C()
45  REM  -- TO MAXIMUM NUMBER OF YEARS
50  DIM C(12)
60  PRINT
70  PRINT "NUMBER OF YEARS ";
80  INPUT N
90  PRINT "LIQUID INVESTMENT INTEREST RATE ";
100  INPUT R1
110 R1 = R1 / 100 + 1
120  PRINT "'SAFE' FIXED INVESTMENT "
125  PRINT "INTEREST RATE ";
130  INPUT R2
140 R2 = R2 / 100 + 1
150  PRINT "MINIMUM AMOUNT OF FIXED "
155  PRINT "INVESTMENT ";
```

```
160  INPUT M
170  PRINT
180  PRINT "(ENTER INFLOWS AS POSITIVE,"
185  PRINT "OUTFLOWS AS NEGATIVE.)"
190  PRINT
200  PRINT "ENTER CASH FLOW AMOUNT FOR YEAR"
205  PRINT "          0 ";
210  INPUT CO
220  FOR J = 1 TO N
230  PRINT "          ";J;" ";
240  INPUT C(J)
250  NEXT J
260  PRINT
270  REM  REMOVE ALL FUTURE OUTFLOWS BY UTILIZING
280  REM  PRIOR INFLOWS WHERE POSSIBLE
290  REM
300  REM  FIRST, FIND OUTFLOWS
310  FOR J = 1 TO N - 1
320  REM  SKIP OVER INFLOWS AND ZERO AMOUNTS
330  IF C(J) >  = 0 THEN 520
340  REM  OUTFLOW FOUND
350 A = C(J)
360  REM  NOW FIND PRIOR INFLOW(S)
370 K = 0
380 K = K + 1
390  IF K = J THEN 520
400  IF C(J - K) <  = 0 THEN 380
410  REM  INFLOW FOUND, REMOVE AMOUNT NEEDED
415  REM  TO ZERO OUTFLOW IF POSSIBLE
420 C(J - K) = C(J - K) +  INT (A / R1 ^ K)
430  IF C(J - K) >  = 0 THEN 490
440  REM  IF NOT ENOUGH MONEY AVAILABLE,
445  REM  CORRECT TO ZERO THE INFLOW
450 A = A +  INT ( ABS (C(J - K)) * R1 ^ K)
460 C(J - K) = 0
470 C(J) = A
480  GOTO 500
490 C(J) = 0
500  PRINT "LIQUID INVESTMENT OF $"; INT ( ABS (A / R1 ^ K))
510  PRINT "TO BE MADE AT END OF YEAR ";J - K
520  NEXT J
530  PRINT
540  REM  DISCOUNT REMAINING OUTFLOWS TO
545  REM  PRESENT AT LIQUID INTEREST RATE
550  FOR J = 1 TO N - 1
560  IF C(J) >  = 0 THEN 590
570 CO = CO +  FN B(C(J) / R1 ^ J)
580 C(J) = 0
590  NEXT J
600 CO =  INT ( ABS (CO) + 0.5)
610  REM  COMPOUND FORWARD ALL REMAINING
620  REM  INFLOWS GREATER THAN MINIMUM
625  REM  FIXED INVESTMENT AMOUNT
630  FOR J = 1 TO N - 1
640  IF C(J) < M THEN 670
```

```
650 C(N) = C(N) +  FN B(C(J) * R2 ^ (N - J))
660  PRINT "FIXED INVESTMENT OF $";C(J)
665  PRINT "TO BE MADE AT END OF YEAR ";J
670  NEXT J
680  PRINT
690 C(N) =  INT ( ABS (C(N)) + 0.5)
700  PRINT "ACTUAL TOTAL INITIAL INVESTMENT"
705  PRINT "                  = $";CO
710  PRINT "TOTAL RETURN ON INVESTMENT"
715  PRINT "                  = $";C(N)
720  REM  INITIALIZE LOW AND HIGH
725  REM  GUESSES, SET LAST GUESS TO ZERO
730 L = 0
740 H = 1
750 R0 = 0
760 R = (H + L) / 2
770  REM  EXIT IF RATE REMAINS UNCHANGED
780  IF R = R0 THEN 910
790  REM  SET LAST GUESS TO CURRENT GUESS
800 R0 = R
810  REM  CALCULATE PRESENT VALUE OF
815  REM  FUTURE VALUE BASED ON RATE OF R
820 T =  INT (C(N) / ((R + 1) ^ N))
830  REM  IF PRESENT VALUE EQUALS INVESTMENT,EXIT
840  IF T = CO THEN 910
850  IF T > CO THEN 890
860  REM  SET HIGH OR LOW GUESS TO CURRENT GUESS
870 H = R
880  GOTO 760
890 L = R
900  GOTO 760
910  PRINT
920  PRINT "'FINANCIAL MANAGEMENT'"
925  PRINT "RATE OF RETURN = "; FN B(R * 100);"%"
930  REM  RESTART OF END PROGRAM?
940  PRINT
950  PRINT "WOULD YOU LIKE TO RE-RUN THIS PROGRAM"
955  PRINT "WITH NEW DATA? (Y/N) ";
960  INPUT Z$
970  IF Z$ = "Y" THEN 60
980  IF Z$ <  > "N" THEN 950
990  END
```

References

Determination and Usage of FM Rate of Return. Detroit: Realtron Corporation, 1973.

Messner, Schreiber, and Lyon. *Marketing Investment Real Estate Finance Taxation Techniques.* Chicago: Realtors National Marketing Institute of the National Association of Realtors, 1975.

Financial Statement Ratio Analysis

This program calculates 22 ratios of interest to an investor, based on data you enter from a firm's financial statements. They indicate a firm's profitability, liquidity, activity, and capital structure. You should only compare the ratios of a firm with others in the same industry, or against an industry average. To use the program, enter the name of the firm which you are analyzing, the date of financial statement and selected dollar amounts from it. You also need to enter the number of common shares outstanding, market price per share and dividends paid per share.

Example

Jim would like to invest in an issue of common stock from a manufacturer of computer equipment. Its financial statements are shown below. Wimpytron has 7,000 shares of common stock outstanding at a market price of $17.50 per share. Dividends of $1.25 per share were paid to stockholders of record from July 1979 through June 1980.

WIMPYTRON, Inc.
Balance Sheet as of July 1, 1980
(figures in thousands of dollars)

ASSETS			LIABILITIES AND EQUITY		
Cash	$ 50		Accounts Payable	$ 75	
Accounts Receivable	100		Notes Payable	155	
Marketable Securities	20		Total Current Liabilities		$230
Inventory	200				
Total Current Assets		$370	Long-Term Debt		190
			STOCKHOLDERS' EQUITY		
Plant and Equipment	500		Common Stock	40	
Less: Depreciation	30		Retained Earnings	380	
Total Fixed Assets		470			420
TOTAL ASSETS		$840	TOTAL LIABILITIES AND EQUITY		$840

WIMPYTRON, Inc.
Income Statement as of July 1, 1980
(figures in thousands of dollars)

Net Sales		$985
Cost of Goods Sold		
Beginning Inventory	$380	
Purchases	200	
Less: Ending Inventory	200	
Total Cost of Goods Sold		380
Gross Margin		$605
Selling Expenses	150	
General & Administrative Expenses	220	
Interest Expense	70	
Total Expenses		440
Income Before Taxes		155
Income Taxes		78
Net Earnings After Taxes		$ 73

How would you run the program to analyze this firm?

```
FINANCIAL STATEMENT RATIO ANALYSIS

   NAME OF FIRM ?WIMPYTRON INC.
 MONTH/DAY/YEAR ?JULY 1 1981
------INCOME STATEMENT------
ENTER AMOUNTS FOR:
                        NET SALES ?985000
               BEGINNING INVENTORY ?380000
                  ENDING INVENTORY ?200000
                COST OF GOODS SOLD ?380000
                  INTEREST EXPENSE ?70000
                    PRE-TAX INCOME ?155000
                      INCOME TAXES ?78000
-------BALANCE SHEET-------
ENTER AMOUNTS FOR:
                               CASH ?50000
             ACCOUNTS RECEIVABLE ?100000
   NOTES & MARKETABLE SECURITIES ?20000
                     TOTAL ASSETS ?840000
             CURRENT LIABILITIES ?230000
           STOCKHOLDERS' EQUITY ?420000

ALSO ENTER:
       COMMON SHARES OUTSTANDING ?7000
          MARKET PRICE PER SHARE ?17.5
             DIVIDENDS PER SHARE ?1.25

--EVALUATION OF WIMPYTRON INC.--
    BY RATIO ANALYSIS
    ENDING PERIOD: JULY 1 1981

-----PROFITABLITY-----
                RETURN OF ASSETS 9.2%
                RETURN ON EQUITY 18.3%
      RETURN OF INVESTED CAPITAL 12.6%
              EARNINGS PER SHARE $11
                 OPERATING RATIO .843:1

   -----LIQUIDITY-----
             NET WORKING CAPITAL $140000
        ACID TEST (QUICK) RATIO .739:1
                  CURRENT RATIO 1.609:1

   -----ACTIVITY-----
                   SALES PER DAY $2698.63
     DAYS SALES OUTSTANDING 37.056  DAYS
         INVENTORY TURNOVER 1.31  TIMES

   -----INDEBTEDNESS-----
     CREDITORS' INTEREST IN FIRM 50%
           TIMES INTEREST EARNED 4.329
                 DEBT TO EQUITY 1:1
     LONG-TERM DEBT TO NET WORTH .452:1
       LONG-TERM DEBT TO CAPITAL .311:1
```

```
   -----EQUITY-----
  STOCKHOLDERS' INTEREST IN FIRM 50%
                    PAYOUT RATIO .114:1
                  EARNINGS YIELD 62.9%
                BOOK VALUE/SHARE $60
             PRICE/EARNINGS RATIO 1.591:1
                  DIVIDEND YIELD 7.1%

DO YOU WANT ANOTHER ANALYSIS (Y/N) ?N
```

Practice Problems

1. Suppose the balance sheet is altered so the stockholders' equity is $390,000. (The long-term debt will be changed by the program.) What ratios will change, and what will their new values be?

Answer: Return on equity, 19.7%; creditors' interest, 53.6%; debt to equity, 1.154:1; long-term debt to net worth, 0.564:1; long-term debt to capital, 0.361:1; stockholders' interest, 46.4%; book value, $55.714.

2. If you interchange the amounts for accounts receivable and cash, what ratios will change and what will their new values be?

Answer: Days sales outstanding changes to 18.528 days. All others ratios remain unchanged.

Program Listing

```
10  PRINT "FINANCIAL STATEMENT RATIO ANALYSIS"
20  DIM D(20)
30  REM
40  REM  D(1)  = NET SALES
50  REM  D(2)  = BEGINNING INVENTORY
60  REM  D(3)  = ENDING INVENTORY
70  REM  D(4)  = COST OF GOODS SOLD
80  REM  D(5)  = INTEREST EXPENSE
90  REM  D(6)  = PRETAX INCOME
100  REM  D(7) = TAXES
110  REM  D(8) = CASH
120  REM  D(9) = ACCOUNTS RECEIVABLE
130  REM  D(10)= NOTES RECEIVABLE
140  REM  D(11)= TOTAL ASSETS
150  REM  D(12)= CURRENT LIABILITIES
160  REM  D(13)= EQUITY
170  REM  D(14)= SHARES OUTSTANDING
180  REM  D(15)= MARKET PRICE PER SHARE
190  REM  D(16)= DIVIDENDS PAID
200  REM
210  DATA   "NET SALES","BEGINNING INVENTORY"
215  DATA  "ENDING INVENTORY"
220  DATA  "COST OF GOODS SOLD","INTEREST EXPENSE"
230  DATA  "PRE-TAX INCOME","INCOME TAXES","CASH"
240  DATA  "ACCOUNTS RECEIVABLE"
245  DATA  "NOTES & MARKETABLE SECURITIES"
250  DATA   "TOTAL ASSETS","CURRENT LIABLITIES"
260  DATA  "STOCKHOLDERS' EQUITY"
265  DATA  "COMMON SHARES OUTSTANDING"
```

```
270  DATA  "MARKET PRICE PER SHARE"
275  DATA  "DIVIDENDS PER SHARE"
280  PRINT
290  PRINT "   NAME OF FIRM ";
300  INPUT N$
310  PRINT " MONTH/DAY/YEAR ";
320  INPUT D$
330  REM  ENTER INCOME STATEMENT ACCOUNTS
340  RESTORE
350  PRINT "------INCOME STATEMENT------"
360  PRINT "ENTER AMOUNTS FOR:"
370  FOR I = 1 TO 7
380  GOSUB 1620
390  NEXT I
400  REM  ENTER BALANCE SHEET ACCOUNTS
410  PRINT "-------BALANCE SHEET-------"
420  PRINT "ENTER AMOUNTS FOR:"
430  FOR I = 8 TO 13
440  GOSUB 1620
450  NEXT I
460  PRINT
470  PRINT "ALSO ENTER:"
480  FOR I = 14 TO 16
490  GOSUB 1620
500  NEXT I
510  PRINT
520  PRINT "--EVALUATION OF ";N$;"--"
525  PRINT "    BY RATIO ANALYSIS"
530  PRINT "    ENDING PERIOD: ";D$
540  PRINT
550  PRINT "-----PROFITABILITY-----"
560 T$ = "RETURN OF ASSETS"
570 X1 = 2
580 XO = (D(6) - D(7)) / D(11)
590  GOSUB 1670
600 T$ = "RETURN ON EQUITY"
610 XO = (D(6) - D(7)) / D(13)
620  GOSUB 1670
630 T$ = "RETURN OF INVESTED CAPITAL"
640 X1 = 2
650 XO = (D(6) - D(7)) / (D(11) - D(12))
660  GOSUB 1670
670 T$ = "EARNINGS PER SHARE"
680 X1 = 3
690 XO = (D(6) - D(7)) / D(14)
700  GOSUB 1670
710 T$ = "OPERATING RATIO"
720 X1 = 1
730 XO = (D(1) - D(6)) / D(1)
740  GOSUB 1670
750  PRINT
760  PRINT "   -----LIQUIDITY-----"
770 T$ = "NET WORKING CAPITAL"
780 X1 = 3
790  REM  CALCULATE CURRENT ASSETS
```

```
800 C1 = D(8) + D(9) + D(10) + D(3)
810  REM  CALCULATE LONG-TERM DEBT
820 L0 = D(11) - D(12) - D(13)
830 X0 = C1 - D(12)
840  GOSUB 1670
850 T$ = "ACID TEST (QUICK) RATIO"
860 X1 = 1
870 X0 = (C1 - D(3)) / D(12)
880  GOSUB 1670
890 T$ = "CURRENT RATIO"
900 X0 = C1 / D(12)
910  GOSUB 1670
920  PRINT
930  PRINT "    -----ACTIVITY-----"
940 T$ = "SALES PER DAY"
950 X1 = 3
960 X0 = D(1) / 365
970  GOSUB 1670
980 T$ = "DAYS SALES OUTSTANDING"
990 X1 = 0
1000 X0 = D(9) / (D(1) / 365)
1005 X0 =  INT (X0 * 1000 + 0.5) / 1000
1010  PRINT  TAB( 5);T$;" ";X0;" ";
1020  PRINT " DAYS"
1030  REM  IF NO INVENTORY DATA, SKIP PRINTING
1040  IF D(2) + D(3) = 0 THEN 1090
1050 T$ = "INVENTORY TURNOVER"
1060 X0 = D(4) / ((D(2) + D(3)) / 2)
1065 X0 =  INT (X0 * 1000 + 0.5) / 1000
1070  PRINT  TAB( 9);T$;" ";X0;" ";
1080  PRINT " TIMES"
1090  PRINT
1100  PRINT "    -----INDEBTEDNESS-----"
1110 T$ = "CREDITORS' INTEREST IN FIRM"
1120 X1 = 2
1130 X0 = (D(11) - D(13)) / D(11)
1140  GOSUB 1670
1150 T$ = "TIMES INTEREST EARNED"
1160 X1 = 0
1170 X0 = (D(6) + D(7) + D(5)) / D(5)
1180  GOSUB 1670
1185  PRINT
1190 T$ = "DEBT TO EQUITY"
1200 X1 = 1
1210 X0 = (D(11) - D(13)) / D(13)
1220  GOSUB 1670
1230 T$ = "LONG-TERM DEBT TO NET WORTH"
1240 X0 = L0 / D(13)
1250  GOSUB 1670
1260 T$ = "LONG-TERM DEBT TO CAPITAL"
1270 X0 = L0 / (L0 + D(13))
1280  GOSUB 1670
1290  PRINT
1300  PRINT "    -----EQUITY-----"
1310 T$ = "STOCKHOLDERS' INTEREST IN FIRM"
```

```
1320 X1 = 2
1330 XO = (D(13)) / D(11)
1340  GOSUB 1670
1350 T$ = "PAYOUT RATIO"
1360 X1 = 1
1370 XO = D(16) / ((D(6) - D(7)) / D(14))
1380  GOSUB 1670
1390 T$ = "EARNINGS YIELD"
1400 X1 = 2
1410 XO = ((D(6) - D(7)) / D(14)) / D(15)
1420  GOSUB 1670
1430 T$ = "BOOK VALUE/SHARE"
1440 X1 = 3
1450 XO = D(13) / D(14)
1460  GOSUB 1670
1470 T$ = "PRICE/EARNINGS RATIO"
1480 X1 = 1
1490 XO = D(15) / ((D(6) - D(7)) / D(14))
1500  GOSUB 1670
1510 T$ = "DIVIDEND YIELD"
1520 X1 = 2
1530 XO = D(16) / D(15)
1540  GOSUB 1670
1550  PRINT
1560  PRINT "DO YOU WANT ANOTHER ANALYSIS (Y/N) ";
1570  INPUT T$
1580  IF T$ = "Y" THEN 280
1590  IF T$ <  > "N" THEN 1560
1600  GOTO 1840
1610  REM  DATA ENTRY ROUTINE
1620  READ T$
1630  PRINT  TAB( 31 -  LEN (T$));" ";T$;" ";
1640  INPUT D(I)
1650  RETURN
1660  REM  SUBROUTINE TO PRINT RATIOS & TURNOVER DATA
1670  PRINT  TAB( 31 -  LEN (T$));" ";T$;
1680 XO =  INT (XO * 1000 + 0.5) / 1000
1690  REM  RATIO FORMAT IF X1=1
1700  IF X1 = 1 THEN 1780
1710  REM  RATE FORMAT IF X1=2
1720  IF X1 = 2 THEN 1800
1730  REM  DOLLAR FORMAT IF X1=3
1740  IF X1 = 3 THEN 1820
1750  REM  DEFAULT TO NO FORMAT IF X1=0
1760  PRINT " ";XO;
1770  RETURN
1780  PRINT " ";XO;":1"
1790  RETURN
1800  PRINT " ";XO * 100;"%"
1810  RETURN
1820  PRINT " $";XO
1830  RETURN
1840  END
```

References

Slavin, Albert, and Reynolds, Isaac. *Basic Accounting* (3rd ed.). Hinsdale, Ill.: Dryden Press, 1975.

Solomon, Ezra. *An Introduction to Financial Management,* Santa Monica: Goodyear Publishing Company, 1977.

Profit Sharing Contributions

This program calculates the profit sharing contributions for up to 250 employees. Some profit sharing plans are not "integrated" (that is, the contribution made for each employee is exactly proportionate to his salary). If his compensation is 5% of the total compensation of all participants, then he is allotted 5% of the total contribution for that year, and so on.

Integrated profit sharing plans are less straightforward. In this case, a salary level no higher than the current Social Security wage base ($22,900 in 1979, $25,900 in 1980) is chosen as the integration level. Each employee whose salary exceeds the integration level receives a percentage (not more than 7%) of the amount by which his earnings exceed the integration level. The remainder of the total contribution is distributed proportionate to salary. If the integrated portion of the total contribution exceeds the total, it is reduced proportionately. If this happens, those whose salary is less than the integration level receive nothing.

This program handles both integrated and non-integrated plans of up to 250 participants. You first enter the name and salary of each employee/participant. After you enter the last employee's name and salary, enter anything for the name, and −1 for the salary when the program requests them. The program then prints out the total of the salaries, and the usual 15% limit on contributions. You then enter the amount of the contribution as a decimal fraction of the total compensation. You are asked if the plan is integrated and, if so, what the integration level and percentage are.

The program then prints a table showing each employee's name, salary, and the amount of his allocation, divided into integrated and non-integrated portions. The program prints the totals for all employees, and then allows you to go back and change some or all of the data.

Example

The following employees are all participants in a profit sharing plan:

Name	Salary
Connell	$150,000
Johnson	22,900
Smith	15,000
Jones	12,000
Brown	10,000

Assuming a 15% company contribution, what allocation would be made to each employee in a non-integrated plan?

Answer:

```
PROFIT SHARING CONTRIBUTIONS
ENTER EACH EMPLOYEE'S NAME AND SALARY
ENTER -1 AS THE SALARY TO END ENTRY
?CONNELL,150000
?JOHNSON,22900
?SMITH,15000
?JONES,12000
?BROWN,10000
?A,-1
TOTAL COMPENSATION = 209900
15% LIMITATION = 31485
P/S % CONTRIBUTION AS A DECIMAL = ?0.15
```

```
IS PLAN INTEGRATED? (Y/N)?N
                  INTEGRATED NON-INTEG.
NAME        SALARY PORTION   PORTION  TOTAL
CONNELL     150000 0          22500   22500
JOHNSON      22900 0           3435    3435
SMITH        15000 0           2250    2250
JONES        12000 0           1800    1800
BROWN        10000 0           1500    1500

TOTALS      209900 0          31485   31485

WANT DIFFERENT SALARIES? (Y/N) ?N
DIFFERENT CONTRIBUTION? (Y/N) ?N
CHANGE WHETHER INTEGRATED? (Y/N) ?N
DIFFERENT INTEGRATION LEVEL? (Y/N) ?N
DIFFERENT INTEGRATION %?(Y/N) ?N
```

Practice Problems

1. For the same group of employees, what would be the allocations in a plan integrated at 3% over $15,000?

Answer: Connell: $23,486.40; Johnson: $3,204.29; Smith: $1,943.64; Jones: $1,554.91; Brown: $1,295.76.

2. If the plan is integrated at 7% over $22,900, what are the allocations for these same employees?

Answer: Connell: $25,038.97; Johnson: $2,464.34; Smith: $1,614.20; Jones: $1,291.36; Brown: $1,076.13.

Program Listing

```
5  PRINT "PROFIT SHARING CONTRIBUTIONS"
9  REM  ROUNDOFF FUNCTION
10  DEF  FN R(X) =  INT (100 * X + 0.5) / 100
20  DIM A$(250),B(250),C(250),D(250)
120  PRINT "ENTER EACH EMPLOYEE'S NAME AND SALARY"
140  PRINT "ENTER -1 AS THE SALARY TO END ENTRY"
150 K = 0
160 J = 1
170  INPUT A$(J),B(J)
180  IF B(J) =  - 1 THEN 240
190 K = K + B(J)
200 J = J + 1
210  GOTO 170
240 J = J - 1
250  PRINT "TOTAL COMPENSATION = ";K
260  PRINT "15% LIMITATION = "; FN R(K * 0.15)
270  PRINT "P/S % CONTRIBUTION AS A DECIMAL = ";
280  INPUT M
290  IF M >  = 1 OR M <  = 0 THEN 270
300  PRINT "IS PLAN INTEGRATED? (Y/N)";
310  INPUT Y$
320  IF Y$ = "N" THEN 640
```

```
330  IF Y$ <  > "Y" THEN 300
360  PRINT "INTEGRATION LEVEL = "
370  INPUT L
390  PRINT "INTEGRATION % AS A DECIMAL = ";
400  INPUT P
420 S = 0
430 H = 0
439  REM  CALCULATE INTEGRATED PORTION FOR EACH EMPLOYEE
440  FOR I = 1 TO J
450  IF B(I) > L THEN 460
453 C(I) = 0
456  GOTO 490
460 C(I) =  FN R(P * (B(I) - L))
470 S = S + 1
480 H = H + C(I)
490  NEXT I
500  IF H < M * K THEN 650
510  IF H > M * K THEN 520
512  FOR I = 1 TO J
514 D(I) = 0
516  NEXT I
518  GOTO 760
520 R = 0
530 T = 0
539  REM  REDUCE INTEGRATED AMOUNT TO TOTAL CONTRIBUTION
540  FOR I = 1 TO J
550  IF C(I) = 0 THEN 620
560 T = T + 1
570  IF T = S THEN 610
580 C(I) =  FN R(C(I) * M * K / H)
590 R = R + C(I)
600  GOTO 620
610 C(I) = M * K - R
620  NEXT I
630  GOTO 760
640 H = 0
642  FOR I = 1 TO J
644 C(I) = 0
646  NEXT I
650 G = M - H / K
669  REM  CALCULATE NON-INTEGRATED PORTION
670  FOR I = 1 TO J
690 D(I) =  FN R(B(I) * G)
720  NEXT I
760 Q = 0
770 X = 0
780  PRINT "                    INTEGRATED NON-INTEG."
790  PRINT "NAME       SALARY PORTION   PORTION  TOTAL"
799  REM  PRINT OUT RESULTS
800  FOR I = 1 TO J
820 X = X + C(I) + D(I)
830 Q = Q + D(I)
840  PRINT A$(I); TAB( 11);B(I); TAB( 18);C(I); TAB( 28);
845  PRINT D(I); TAB( 36);C(I) + D(I)
850  NEXT I
```

```
855  PRINT
860  PRINT "TOTALS"; TAB( 11);K; TAB( 18);
870  IF H >  = M * K THEN 900
880  PRINT H; TAB( 28)
890  GOTO 910
900  PRINT M * K; TAB( 28)
910  PRINT Q; TAB( 36);X
920  PRINT
930  PRINT "WANT DIFFERENT SALARIES? (Y/N) ";
940  INPUT Z$
950  IF Z$ = "Y" THEN 120
960  PRINT "DIFFERENT CONTRIBUTION? (Y/N) ";
970  INPUT Z$
980  IF Z$ = "Y" THEN 270
990  PRINT "CHANGE WHETHER INTEGRATED? (Y/N) ";
1010  INPUT Z$
1020  IF Z$ = "Y" THEN 300
1030  PRINT "DIFFERENT INTEGRATION LEVEL? (Y/N) ";
1040  INPUT Z$
1050  IF Z$ = "Y" THEN 360
1060  PRINT "DIFFERENT INTEGRATION %?(Y/N) ";
1070  INPUT Z$
1080  IF Z$ = "Y" THEN 390
1090  END
```

Reference

U.S. Internal Revenue Service Code, Sections 401-04.

Checkbook Reconciliation

This program can remove a considerable burden from you each time you reconcile your checking account. Since the computer performs all of the addition and subtraction, the chance for errors to occur is greatly reduced.

You must enter the ending balance from your statement, then each deposit or credit made since the statement date. After you have entered all outstanding deposits and credits, enter zero. This signals the program to continue to the next section, entry of outstanding checks. Enter check and other debit amounts as you did for deposits, and enter zero when all outstanding checks and debits have been entered.

You should enter only positive dollar amounts for each response. The exception is that you may enter negative amounts for your previous balance and your checkbook balance.

If your account won't balance, check all of your entries to make sure they are complete and correct. Do your check register entries match the amounts on the cancelled checks? Have you entered all checks, deposits, and automatic debits and credits? If you can't find any mistakes, call your bank.

Example

Janet's checking account statement does not show the $600.00 paycheck she deposited yesterday. She also wrote two checks that aren't shown either, one for $87.32, and one for $250.00. If the ending balance from the statement is $348.55, Janet's check register shows a balance of $614.54, and service charges on the statement are $3.31, what is her adjusted account balance? Is Janet's account balanced?

Answer: Janet's adjusted balance is $611.23. Her account is balanced.

```
CHECKBOOK RECONCILIATION

WHAT IS THE ENDING BALANCE
FROM THE STATEMENT ?348.55

ENTER THE AMOUNT OF EACH DEPOSIT
NOT SHOWN ON THE STATEMENT
(ENTER ZERO WHEN ALL OUTSTANDING
DEPOSITS ARE ENTERED)
?600
?0

ENTER THE AMOUNT OF EACH CHECK
NOT SHOWN ON THE STATEMENT
(ENTER ZERO WHEN ALL OUTSTANDING
CHECKS ARE ENTERED)
?87.32
?250
?0

ACCOUNT BALANCE = $611.23

ENTER YOUR CHECKBOOK BALANCE ?614.54
ENTER THE AMOUNT OF SERVICE CHARGES ?3.31

ADJUSTED ACCOUNT BALANCE = $611.23
```

```
WOULD YOU LIKE TO RE-RUN THIS PROGRAM
WITH NEW DATA? (Y/N) ?N
```

Practice Problems

1. Ending balance is $352.13. Not shown on the statement are three deposits of $100.00 each, and six checks amounting to $159.21, $25.00, $14.75, $29.54, $45.67, and $22.50. What is the account balance? The checkbook balance is $358.97. Service charges on this statement are $3.51. What is the adjusted account balance? Does the account balance?

Answer: The account balance is $355.46. The adjusted account balance is $355.46. Yes, the account does balance.

2. Ending balance is −$17.39. One deposit of $250.00 is outstanding, as are three checks: $50.00, $25.00, and $12.98. A pre-authorized withdrawal of $35.00 also has occurred, but is not shown on this statement. What is the account balance? If the checkbook balance is $118.99, and service charges are $9.36, what is the adjusted account balance? Is the account balanced?

Answer: The account balance is $109.63. The adjusted account balance is $109.63. Yes, the account is balanced.

Program Listing

```
10  PRINT "CHECKBOOK RECONCILIATION"
20  REM  - FUNCTION TO DETERMINE IF POSITIVE
25  REM  - DOLLAR AMOUNT WAS ENTERED
30  DEF  FN B(X) =  INT (X * 100 + 0.5) / 100 *  SGN (X)
40  PRINT
50  PRINT "WHAT IS THE ENDING BALANCE"
55  PRINT "FROM THE STATEMENT ";
60  INPUT E
70  REM  - SPECIAL TEST FOR VALID INPUT
75  REM  - (NEGATIVE NUMBER ALLOWED)
77 X = E * 100
80  IF X =  INT (X) THEN 120
90  REM  - INVALID AMOUNT. DISPLAY ERROR,
95  REM  - LOOP TO RE-ENTER
100  GOSUB 680
110  GOTO 50
120  PRINT
130  PRINT "ENTER THE AMOUNT OF EACH DEPOSIT"
135  PRINT "NOT SHOWN ON THE STATEMENT"
140  PRINT "(ENTER ZERO WHEN ALL OUTSTANDING"
145  PRINT "DEPOSITS ARE ENTERED)"
150 D = 0
160  INPUT A
170  REM  - ALL DEPOSITS ENTERED?
180  IF A = 0 THEN 260
190  REM  - NO, TEST FOR VALID ENTRY
200  IF ( FN B(A) = A) THEN 240
210  REM  - INVALID, PRINT STANDARD ERROR,
215  REM  - LOOP TO RE-ENTER
220  GOSUB 720
230  GOTO 160
240 D = D + A
```

```
250  GOTO 160
260  PRINT
270  PRINT "ENTER THE AMOUNT OF EACH CHECK"
275  PRINT "NOT SHOWN ON THE STATEMENT"
280  PRINT "(ENTER ZERO WHEN ALL OUTSTANDING "
285  PRINT "CHECKS ARE ENTERED)"
290 C = 0
300  INPUT A
310  REM  - ALL OUTSTANDING CHECKS ENTERED?
320  IF A = 0 THEN 400
330  REM  - NO, TEST FOR VALID ENTRY
340  IF ( FN B(A) = A) THEN 380
350  REM  - INVALID, PRINT STANDARD ERROR,
355  REM  - LOOP TO RE-ENTER
360  GOSUB 720
370  GOTO 300
380 C = C + A
390  GOTO 300
400  PRINT
405 Y =  INT ((E + D - C) * 100 + 0.5) / 100
410  PRINT "ACCOUNT BALANCE = $";Y
420  PRINT
430  PRINT "ENTER YOUR CHECKBOOK BALANCE ";
440  INPUT B
450  PRINT "ENTER THE AMOUNT OF SERVICE CHARGES ";
460  INPUT S
470  REM  - TEST FOR VALID ENTRY
480  IF  FN B(S) = S THEN 520
490  REM  - INVALID, PRINT STANDARD ERROR,
495  REM  - LOOP TO RE-ENTER
500  GOSUB 720
510  GOTO 450
520  PRINT
525 X =  INT ((B - S) * 100 + 0.5) / 100
530  PRINT "ADJUSTED ACCOUNT BALANCE = $";X
540  IF Y = X THEN 620
550  PRINT
560  PRINT "YOUR ACCOUNT IS OUT OF BALANCE."
570  PRINT "MAKE SURE YOU HAVE INCLUDED"
575  PRINT "ALL TRANSACTIONS AGAINST THIS ACCOUNT,"
580  PRINT "INCLUDING AUTOMATIC DEPOSITS AND"
590  PRINT "INTEREST PAYMENTS, AS WELL AS"
595  PRINT "PRE-AUTHORIZED WITHDRAWALS."
600  PRINT
610  REM
620  PRINT
630  PRINT "WOULD YOU LIKE TO RE-RUN THIS PROGRAM"
635  PRINT "WITH NEW DATA? (Y/N) ";
640  INPUT Z$
650  IF Z$ = "Y" THEN 40
660  IF Z$ = "N" THEN 760
670  GOTO 630
680  PRINT
690  PRINT "ERROR: ENTER A VALID DOLLAR AMOUNT ONLY"
700  PRINT
```

```
710  RETURN
720  PRINT
730  PRINT "ERROR: ENTER A POSITIVE"
735  PRINT "       DOLLAR AMOUNT ONLY"
740  PRINT
750  RETURN
760  END
```

Home Budgeting

This program sets up a cash budget for personal use, allowing for a variety of expenses which can occur at many different times. Once you enter the income and expense information which the program requests, day-by-day details of income and expenses print as they occur. The program also allows you to use credit cards as a means of paying expenses when the cash you have is insufficient to meet your obligations. Or, if you wish, you can delay them until the next time they come up.

To use the program, enter the date where the budget will begin. The program then guides you through a series of entries, starting with net income(s), followed by secured loans, credit cards and, finally, normal living expenses. If your budget does not include items which the programs asks for, just enter zero for those items. The program will then skip to the next budget item.

Whenever you have a budget item to enter, you will have to enter its periodic amount, how often it occurs, and when it will occur next. The exceptions to this are secured loans and credit cards, which ask for more information. The periodic amount is the amount you regularly receive as income, or pay as an expense. When you enter how often the budget item occurs, it must be an integer from 1 to 99, inclusive. This number tells the program how many times per year the item occurs (1=yearly, 2=semiannually, 4=quarterly, 6=bimonthly, 12=monthly, 24=semimonthly, 26=biweekly, and 52=weekly).

If the next date for the budget item happens to be the same as the budget start date, enter zero. Otherwise, enter the next date as one number (for example, 91580=Sept. 15, 1980). You can enter a date months or even years after the budget start date if you like. When the program performs its cash flow analysis, it will "activate" future income or expenses when it reaches the date you specify.

With secured loans, you have to enter the remaining balance of the loan as well as the periodic amount, frequency and next date. When you enter credit card information, you will input the annual percentage rate for the card, the remaining balance, and its authorized credit limit. The program automatically calculates the number and amount of remaining payments for each credit card, and displays them. If you want to change the payment which the program calculates, just specify a new periodic payment of a higher or lower amount. Note: the program will calculate an even stream of payments to make budgeting more predictable. When the remaining balance of the credit card goes below the calculated payment amount during the cash flow analysis, only the remaining balance is paid.

Once you have entered all of the budget items, the program will ask how much cash you have on hand. Enter this amount, and the program will begin its cash flow analysis. At the end of each month's detail, total cash inflows and outflows are printed. At this point, you can choose to go to the next month's analysis or stop the program.

Because you will be entering a significant amount of data in order to run this program, you should know how to correct data entry errors. You can only correct errors which you make on the current budget item (that is, you cannot backtrack to the fifth item when you are on the tenth).

On a current budget item, you can move as far back as the periodic amount entry by entering −1. For example, you notice that you have entered the wrong periodic amount for salary 1, and the program now wants you to enter the next date for this item. Rather than entering the next date for salary 1, enter −1. The computer will accept this entry and then ask you to enter the periodic amount for salary 1 again.

Program Notes

Home budgeting/cash flow allows for a maximum of 3 incomes, 3 loans, 5 credit cards and 25 expense items. At present, the program will allocate cash to loans first, then credit cards, and finally other expenses. The expenses are arranged in descending order of importance (that is, if a loan, charge card and restaurant expense all appear on the same day, the program will allocate cash to the loan first and to the resturant expense last).

Changing this program to allow for more budget items is a three-step process. First, change line 20, substituting the terms A, B, C, and D in parentheses with actual numbers. These items are explained below.

```
20 DIM D(12), IO(A,2), CO(B,3), C1(C,5), C1$(C), E0(D,2)
```

A = Maximum number of incomes
B = Maximum number of secured loans
C = Maximum number of charge cards
D = Maximum number of expense items

The second step is to put descriptions of the extra budget items in the DATA statements at the beginning of the program. You can add any extra loans by placing DATA statements between lines 90 and 100 which contain descriptions of the loans. Note: you do not need to change DATA statements to allow for more incomes or more credit cards. To add more expenses, add DATA statements anywhere from lines 110 through 180.

The third and last step is to change FOR/NEXT loops in the program. If you change the number of secured loans, be sure to also change lines 530 and 1080 of the program. Currently they are set for three iterations. Change the number 3 in these two statements to the new number of secured loans. If you have added or eliminated expense items, you will need to change lines 750 and 1360. Change the number 25 in these two statements to the new number of expense items.

Example

```
HOME BUDGETING/CASH FLOW MODEL

DATE TO START ANALYSIS FROM:
ENTER MONTH-DAY-YEAR (MMDDYY) ?90180
```

Start analysis on Sept. 1, 1980.

```
------NET SALARY 1------
PERIODIC AMOUNT FOR INCOME ?512
HOW MANY TIMES PER YEAR ?26
ENTER MONTH-DAY-YEAR (MMDDYY) ?90580
------NET SALARY 2------
PERIODIC AMOUNT FOR INCOME ?100
HOW MANY TIMES PER YEAR ?4
ENTER MONTH-DAY-YEAR (MMDDYY) ?100180
------NET SALARY 3------
PERIODIC AMOUNT FOR INCOME ?0
```

First net income is $512.00, paid bi-weekly. The next paycheck will be on Sept. 5, 1980.

Finish entering income data.

```
PERIODIC AMOUNT FOR MORTGAGE ?0

PERIODIC AMOUNT FOR CAR LOAN ?80
HOW MANY TIMES PER YEAR ?12
ENTER MONTH-DAY-YEAR (MMDDYY) ?-1
CURRENT BALANCE ?-1
PERIODIC AMOUNT FOR CAR LOAN ?-1

PERIODIC AMOUNT FOR CAR LOAN ?95
HOW MANY TIMES PER YEAR ?12
ENTER MONTH-DAY-YEAR (MMDDYY) ?91580
CURRENT BALANCE ?1290

PERIODIC AMOUNT FOR OTHER LOAN ?0
```

Car loan payment was incorrect. −1 entry used to back up to the incorrect entry.

```
NAME OF CREDIT CARD 1(RETURN TO END)
?VISA
ANNUAL INTEREST RATE ?18
CURRENT BALANCE ?525
CREDIT LIMIT ?100

12 PAYMENTS OF $52.5
NEEDED TO PAY DEBT
```

Enter credit card 1.
Note: calculation of payments allows for interest over 12 payments.

```
CHANGE AMOUNT (Y/N) ?Y
ENTER DESIRED PAYMENT AMOUNT ?35

18 PAYMENTS OF $35
NEEDED TO PAY DEBT

CHANGE AMOUNT (Y/N) ?N
```

Payment was changed to a lower amount.

```
ENTER NEXT VISA BILLING DATE:
ENTER MONTH-DAY-YEAR (MMDDYY) ?92080
NAME OF CREDIT CARD 2(RETURN TO END)
?MASTERCHARGE
ANNUAL INTEREST RATE ?18
CURRENT BALANCE ?230
CREDIT LIMIT ?500

12 PAYMENTS OF $23
NEEDED TO PAY DEBT

CHANGE AMOUNT (Y/N) ?N
ENTER NEXT MASTERCHARGE BILLING DATE:
ENTER MONTH-DAY-YEAR (MMDDYY) ?92480
NAME OF CREDIT CARD 3(RETURN TO END)
```

Finish entering credit card data.

```
PERIODIC AMOUNT FOR PROPERTY TAX ?0

PERIODIC AMOUNT FOR RENT ?300
HOW MANY TIMES PER YEAR ?12
ENTER MONTH-DAY-YEAR (MMDDYY) ?90180

PERIODIC AMOUNT FOR LIFE INSURANCE ?12.5
HOW MANY TIMES PER YEAR ?12
ENTER MONTH-DAY-YEAR (MMDDYY) ?92480

PERIODIC AMOUNT FOR HOUSE INSURANCE ?0

PERIODIC AMOUNT FOR CAR INSURANCE ?125
HOW MANY TIMES PER YEAR ?4
ENTER MONTH-DAY-YEAR (MMDDYY) ?102180
```

Quarterly expense.

```
PERIODIC AMOUNT FOR TELEPHONE ?35
HOW MANY TIMES PER YEAR ?12
ENTER MONTH-DAY-YEAR (MMDDYY) ?90880

PERIODIC AMOUNT FOR GAS & ELECTRIC ?17
HOW MANY TIMES PER YEAR ?12
ENTER MONTH-DAY-YEAR (MMDDYY) ?91880
```

```
PERIODIC AMOUNT FOR WATER ?0

PERIODIC AMOUNT FOR TRASH PICKUP ?0

PERIODIC AMOUNT FOR GROCERIES ?25
HOW MANY TIMES PER YEAR ?52
ENTER MONTH-DAY-YEAR (MMDDYY) ?90580

PERIODIC AMOUNT FOR CLOTHING ?40
HOW MANY TIMES PER YEAR ?4
ENTER MONTH-DAY-YEAR (MMDDYY) ?110180

PERIODIC AMOUNT FOR PHYSICIAN ?30
HOW MANY TIMES PER YEAR ?4
ENTER MONTH-DAY-YEAR (MMDDYY) ?120180

PERIODIC AMOUNT FOR DENTIST ?0

PERIODIC AMOUNT FOR DRUGS ?0

PERIODIC AMOUNT FOR TUITION ?0

PERIODIC AMOUNT FOR CHILD CARE ?0

PERIODIC AMOUNT FOR GAS/OIL ?15
HOW MANY TIMES PER YEAR ?52
ENTER MONTH-DAY-YEAR (MMDDYY) ?90680

PERIODIC AMOUNT FOR AUTO REPAIR ?40
HOW MANY TIMES PER YEAR ?3
ENTER MONTH-DAY-YEAR (MMDDYY) ?10181

PERIODIC AMOUNT FOR COMMUTING ?0

PERIODIC AMOUNT FOR MEDICAL PLAN ?0

PERIODIC AMOUNT FOR HOME REPAIR ?0

PERIODIC AMOUNT FOR RESTAURANTS ?15
HOW MANY TIMES PER YEAR ?52
ENTER MONTH-DAY-YEAR (MMDDYY) ?0

PERIODIC AMOUNT FOR MOVIES/CONCERTS ?10
HOW MANY TIMES PER YEAR ?26
ENTER MONTH-DAY-YEAR (MMDDYY) ?0

PERIODIC AMOUNT FOR SUBSCRIPTIONS ?0

PERIODIC AMOUNT FOR MISCELLANEOUS ?18
HOW MANY TIMES PER YEAR ?52
ENTER MONTH-DAY-YEAR (MMDDYY) ?91580
ENTER CASH ON HAND ?400
```

Weekly expense.

Expense occurs every 4 months.

Next date for this item is the same as the budget start date.

Cash available at start of analysis.

```
CASH FLOWS FOR 9/80
OPENING CASH BALANCE $400
 1        RENT                    -300
 1        RESTAURANTS             -15
 1        MOVIES/CONCERTS         -10
FRI 5     INCOME 1          512
FRI 5     GROCERIES               -25
SAT 6     GAS/OIL                 -15
MON 8     TELEPHONE               -35
MON 8     RESTAURANTS             -15
FRI 12    GROCERIES               -25
SAT 13    GAS/OIL                 -15
MON 15    CAR LOAN PAYMENT        -95
MON 15    RESTAURANTS             -15
MON 15    MOVIES/CONCERTS         -10
MON 15    MISCELLANEOUS           -18
THU 18    GAS & ELECTRIC          -17
FRI 19    INCOME 1          512
FRI 19    GROCERIES               -25
SAT 20    VISA                    -35
SAT 20    GAS/OIL                 -15
MON 22    RESTAURANTS             -15
MON 22    MISCELLANEOUS           -18
WED 24    MASTERCHARGE            -23
WED 24    LIFE INSURANCE          -12.5
FRI 26    GROCERIES               -25
SAT 27    GAS/OIL                 -15
MON 29    RESTAURANTS             -15
MON 29    MOVIES/CONCERTS         -10
MON 29    MISCELLANEOUS           -18
          CASH IN:1024   CASH OUT:836.5
```

Total monthly cash income and expenses.

```
DO YOU WANT TO SEE THE NEXT MONTH
(Y/N) ?Y
CASH FLOWS FOR 10/80
OPENING CASH BALANCE $587.5
WED 1     INCOME 2          100
WED 1     RENT                    -300
THU 2     INCOME 1          512
THU 2     GROCERIES               -25
FRI 3     GAS/OIL                 -15
SUN 5     RESTAURANTS             -15
SUN 5     MISCELLANEOUS           -18
WED 8     TELEPHONE               -35
THU 9     GROCERIES               -25
FRI 10    GAS/OIL                 -15
SUN 12    RESTAURANTS             -15
SUN 12    MOVIES/CONCERTS         -10
SUN 12    MISCELLANEOUS           -18
WED 15    CAR LOAN PAYMENT        -95
THU 16    INCOME 1          512
THU 16    GROCERIES               -25
FRI 17    GAS/OIL                 -15
SAT 18    GAS & ELECTRIC          -17
SUN 19    RESTAURANTS             -15
```

```
SUN 19    MISCELLANEOUS          -18
MON 20    VISA                   -35
TUE 21    CAR INSURANCE          -125
THU 23    GROCERIES              -25
FRI 24    MASTERCHARGE           -23
FRI 24    LIFE INSURANCE         -12.5
FRI 24    GAS/OIL                -15
SUN 26    RESTAURANTS            -15
SUN 26    MOVIES/CONCERTS        -10
SUN 26    MISCELLANEOUS          -18
THU 30    INCOME 1          512
THU 30    GROCERIES              -25
FRI 31    GAS/OIL                -15
           CASH IN:1636    CASH OUT:994.5

DO YOU WANT TO SEE THE NEXT MONTH
(Y/N) ?Y
CASH FLOWS FOR 11/80
OPENING CASH BALANCE $1229
SAT 1     RENT                   -300
SAT 1     CLOTHING               -40
SUN 2     RESTAURANTS            -15
SUN 2     MISCELLANEOUS          -18
THU 6     GROCERIES              -25
FRI 7     GAS/OIL                -15
SAT 8     TELEPHONE              -35
SUN 9     RESTAURANTS            -15
SUN 9     MOVIES/CONCERTS        -10
SUN 9     MISCELLANEOUS          -18
THU 13    INCOME 1          512
THU 13    GROCERIES              -25
FRI 14    GAS/OIL                -15
SAT 15    CAR LOAN PAYMENT       -95
SUN 16    RESTAURANTS            -15
SUN 16    MISCELLANEOUS          -18
TUE 18    GAS & ELECTRIC         -17
THU 20    VISA                   -35
THU 20    GROCERIES              -25
FRI 21    GAS/OIL                -15
SUN 23    RESTAURANTS            -15
SUN 23    MOVIES/CONCERTS        -10
SUN 23    MISCELLANEOUS          -18
MON 24    MASTERCHARGE           -23
MON 24    LIFE INSURANCE         -12.5
THU 27    INCOME 1          512
THU 27    GROCERIES              -25
FRI 28    GAS/OIL                -15
SUN 30    RESTAURANTS            -15
SUN 30    MISCELLANEOUS          -18
           CASH IN:1024    CASH OUT:902.5
```

```
DO YOU WANT TO SEE THE NEXT MONTH
(Y/N) ?Y
CASH FLOWS FOR 12/80
OPENING CASH BALANCE $1350.5
MON 1     RENT                    -300
MON 1     PHYSICIAN               -30
THU 4     GROCERIES               -25
FRI 5     GAS/OIL                 -15
SUN 7     RESTAURANTS             -15
SUN 7     MOVIES/CONCERTS         -10
SUN 7     MISCELLANEOUS           -18
MON 8     TELEPHONE               -35
THU 11    INCOME 1          512
THU 11    GROCERIES               -25
FRI 12    GAS/OIL                 -15
SUN 14    RESTAURANTS             -15
SUN 14    MISCELLANEOUS           -18
MON 15    CAR LOAN PAYMENT        -95
THU 18    GAS & ELECTRIC          -17
THU 18    GROCERIES               -25
FRI 19    GAS/OIL                 -15
SAT 20    VISA                    -35
SUN 21    RESTAURANTS             -15
SUN 21    MOVIES/CONCERTS         -10
SUN 21    MISCELLANEOUS           -18
WED 24    MASTERCHARGE            -23
WED 24    LIFE INSURANCE          -12.5
THU 25    INCOME 1          512
THU 25    GROCERIES               -25
FRI 26    GAS/OIL                 -15
SUN 28    RESTAURANTS             -15
SUN 28    MISCELLANEOUS           -18
           CASH IN:1024    CASH OUT:859.5

DO YOU WANT TO SEE THE NEXT MONTH
(Y/N) ?Y
CASH FLOWS FOR 1/81
OPENING CASH BALANCE $1515
THU 1     INCOME 2          100
THU 1     RENT                    -300
THU 1     GROCERIES               -25
THU 1     AUTO REPAIR             -40
FRI 2     GAS/OIL                 -15
SUN 4     RESTAURANTS             -15
SUN 4     MOVIES/CONCERTS         -10
SUN 4     MISCELLANEOUS           -18
THU 8     INCOME 1          512
THU 8     TELEPHONE               -35
THU 8     GROCERIES               -25
FRI 9     GAS/OIL                 -15
SUN 11    RESTAURANTS             -15
SUN 11    MISCELLANEOUS           -18
THU 15    CAR LOAN PAYMENT        -95
THU 15    GROCERIES               -25
FRI 16    GAS/OIL                 -15
```

```
SUN 18    GAS & ELECTRIC          -17
SUN 18    RESTAURANTS             -15
SUN 18    MOVIES/CONCERTS         -10
SUN 18    MISCELLANEOUS           -18
TUE 20    VISA                    -35
WED 21    CAR INSURANCE           -125
THU 22    INCOME 1          512
THU 22    GROCERIES               -25
FRI 23    GAS/OIL                 -15
SAT 24    MASTERCHARGE            -23
SAT 24    LIFE INSURANCE          -12.5
SUN 25    RESTAURANTS             -15
SUN 25    MISCELLANEOUS           -18
THU 29    GROCERIES               -25
FRI 30    GAS/OIL                 -15
           CASH IN:1124    CASH OUT:1034.5

DO YOU WANT TO SEE THE NEXT MONTH
(Y/N) ?Y
CASH FLOWS FOR 2/81
OPENING CASH BALANCE $1604.5
SUN 1     RENT                    -300
SUN 1     CLOTHING                -40
SUN 1     RESTAURANTS             -15
SUN 1     MOVIES/CONCERTS         -10
SUN 1     MISCELLANEOUS           -18
THU 5     INCOME 1          512
THU 5     GROCERIES               -25
FRI 6     GAS/OIL                 -15
SUN 8     TELEPHONE               -35
SUN 8     RESTAURANTS             -15
SUN 8     MISCELLANEOUS           -18
THU 12    GROCERIES               -25
FRI 13    GAS/OIL                 -15
SUN 15    CAR LOAN PAYMENT        -95
SUN 15    RESTAURANTS             -15
SUN 15    MOVIES/CONCERTS         -10
SUN 15    MISCELLANEOUS           -18
WED 18    GAS & ELECTRIC          -17
THU 19    INCOME 1          512
THU 19    GROCERIES               -25
FRI 20    VISA                    -35
FRI 20    GAS/OIL                 -15
SUN 22    RESTAURANTS             -15
SUN 22    MISCELLANEOUS           -18
TUE 24    MASTERCHARGE            -23
TUE 24    LIFE INSURANCE          -12.5
THU 26    GROCERIES               -25
FRI 27    GAS/OIL                 -15
           CASH IN:1024    CASH OUT:869.5
```

```
DO YOU WANT TO SEE THE NEXT MONTH
(Y/N) ?Y
CASH FLOWS FOR 3/81
OPENING CASH BALANCE $1759
SUN 1     RENT                    -300
SUN 1     PHYSICIAN               -30
SUN 1     RESTAURANTS             -15
SUN 1     MOVIES/CONCERTS         -10
SUN 1     MISCELLANEOUS           -18
THU 5     INCOME 1          512
THU 5     GROCERIES               -25
FRI 6     GAS/OIL                 -15
SUN 8     TELEPHONE               -35
SUN 8     RESTAURANTS             -15
SUN 8     MISCELLANEOUS           -18
THU 12    GROCERIES               -25
FRI 13    GAS/OIL                 -15
SUN 15    CAR LOAN PAYMENT        -95
SUN 15    RESTAURANTS             -15
SUN 15    MOVIES/CONCERTS         -10
SUN 15    MISCELLANEOUS           -18
WED 18    GAS & ELECTRIC          -17
THU 19    INCOME 1          512
THU 19    GROCERIES               -25
FRI 20    VISA                    -35
FRI 20    GAS/OIL                 -15
SUN 22    RESTAURANTS             -15
SUN 22    MISCELLANEOUS           -18
TUE 24    MASTERCHARGE            -23
TUE 24    LIFE INSURANCE          -12.5
THU 26    GROCERIES               -25
FRI 27    GAS/OIL                 -15
SUN 29    RESTAURANTS             -15
SUN 29    MOVIES/CONCERTS         -10
SUN 29    MISCELLANEOUS           -18
           CASH IN:1024    CASH OUT:902.5

DO YOU WANT TO SEE THE NEXT MONTH
(Y/N) ?Y
CASH FLOWS FOR 4/81
OPENING CASH BALANCE $1880.5
WED 1     INCOME 2          100
WED 1     RENT                    -300
THU 2     INCOME 1          512
THU 2     GROCERIES               -25
FRI 3     GAS/OIL                 -15
SUN 5     RESTAURANTS             -15
SUN 5     MISCELLANEOUS           -18
WED 8     TELEPHONE               -35
THU 9     GROCERIES               -25
FRI 10    GAS/OIL                 -15
SUN 12    RESTAURANTS             -15
SUN 12    MOVIES/CONCERTS         -10
SUN 12    MISCELLANEOUS           -18
WED 15    CAR LOAN PAYMENT        -95
```

```
THU 16    INCOME 1          512
THU 16    GROCERIES                -25
FRI 17    GAS/OIL                  -15
SAT 18    GAS & ELECTRIC           -17
SUN 19    RESTAURANTS              -15
SUN 19    MISCELLANEOUS            -18
MON 20    VISA                     -35
TUE 21    CAR INSURANCE            -125
THU 23    GROCERIES                -25
FRI 24    MASTERCHARGE             -23
FRI 24    LIFE INSURANCE           -12.5
FRI 24    GAS/OIL                  -15
SUN 26    RESTAURANTS              -15
SUN 26    MOVIES/CONCERTS          -10
SUN 26    MISCELLANEOUS            -18
THU 30    INCOME 1          512
THU 30    GROCERIES                -25
           CASH IN:1636    CASH OUT:979.5

DO YOU WANT TO SEE THE NEXT MONTH
(Y/N) ?N
```

Program Listing

```
10  REM  HOME BUDGETING/CASH FLOW ANALYSIS
20  DIM D(12),IO(3,2),CO(4,3),C1(5,5),C1$(5),EO(25,2)
30  REM  D()     --DAY OFFSET FACTORS
40  REM  IO()    --SALARIED INCOME
50  REM  C1()    --CREDIT INSTRUMENTS
60  REM  EO()    --EXPENSES
70  REM  C1$()   --DESCRIPTIONS OF CREDIT CARDS
80  REM  CO()    --FIXED-TERM LOANS
90  DATA  "MORTGAGE","CAR LOAN","OTHER LOAN"
100  REM  EXPENSES
110  DATA  "PROPERTY TAX","RENT"
120  DATA  "LIFE INSURANCE","HOUSE INSURANCE","CAR INSURANCE"
130  DATA  "TELEPHONE","GAS & ELECTRIC","WATER","TRASH PICKUP"
140  DATA  "GROCERIES","CLOTHING","PHYSICIAN","DENTIST"
150  DATA  "DRUGS","TUITION","CHILD CARE","GAS/OIL"
160  DATA   "AUTO REPAIR","COMMUTING","MEDICAL PLAN"
170  DATA   "HOME REPAIR","RESTAURANTS","MOVIES/CONCERTS"
180  DATA  "SUBSCRIPTIONS","MISCELLANEOUS"
190 D(1) = 31
200 D(2) = 28
210 D(3) = 31
220 D(4) = 30
230 D(5) = 31
240 D(6) = 30
250 D(7) = 31
260 D(8) = 31
270 D(9) = 31
280 D(10) = 31
290 D(11) = 30
```

```
300 D(12) = 31
310 D$ = "SATSUNMONTUEWEDTHUFRI"
320  PRINT "HOME BUDGETING/CASH FLOW MODEL"
330  PRINT
340  PRINT "DATE TO START ANALYSIS FROM: "
350  GOSUB 2990
360 D1 = D2
370 Y1 = Y
380 M1 = M
390 D4 = Y * 10000 + M * 100 + D2
400  PRINT
410  REM  ENTER INCOMES--AMOUNTS & FREQUENCY
420 I2 = 0
430 X$ = "INCOME
440  PRINT "------NET SALARY ";I2 + 1;"------"
450  GOSUB 2360
460  IF A2(1) = 0 THEN 510
470 I2 = I2 + 1
480 IO(I2,1) = A2(1)
490 IO(I2,2) = A2(2)
500  GOTO 440
510  PRINT
520  REM  ENTER SECURED LOANS
530  FOR I = 1 TO 3
540  READ X$
550  PRINT
560  GOSUB 2360
570  IF A2(1) = 0 THEN 640
580  IF A2(1) < 0 THEN 550
590 CO(I,1) = A2(1)
600 CO(I,2) = A2(2)
610  PRINT "CURRENT BALANCE ";
620  INPUT CO(I,3)
630  IF CO(I,3) < 1 THEN 560
640  NEXT I
650  REM   ENTER CREDIT CARDS AND DESCRIPTIONS
660  REM   MONTHLY PAYMENTS ARE ASSUMED
670  PRINT
680 K = 1
690  GOSUB 2020
700  IF C1$(K) <  = " " THEN 730
710 K = K + 1
720  GOTO 690
730 C4 = K - 1
740  REM  ENTER EXPENSES
750  FOR K = 1 TO 25
760  PRINT
770  READ X$
780  GOSUB 2360
790 EO(K,1) = A2(1)
800 EO(K,2) = A2(2)
810  NEXT K
820  REM  INPUT PRESENT CASH RESERVES
830  PRINT "ENTER CASH ON HAND ";
840  INPUT BO
```

```
850  REM  BEGIN ANALYSIS
860  PRINT
870  PRINT "CASH FLOWS FOR ";M1;"/";Y1
880  PRINT "OPENING CASH BALANCE $";B0
890 E1 = 0
900 I1 = 0
910  FOR K1 = D1 TO D(M1)
920  RESTORE
930  FOR J = 1 TO I2
940  REM  CHECK FOR INCOME
950  IF  INT (I0(J,2)) > D4 THEN 1060
960 B0 = B0 + I0(J,1)
970 I1 = I1 + I0(J,1)
980 M = M1
990 D2 = D1
1000 Y = Y1
1010 D3 =  INT ((I0(J,2) -  INT (I0(J,2))) * 100 + 0.5)
1020 A2(1) = D3 / 100
1030  GOSUB 2510
1040 I0(J,2) = A2(1) + Y * 10000 + M * 100 + D2
1050  PRINT A$;" ";D1; TAB( 9);"INCOME ";J; TAB( 25);I0(J,1)
1060  NEXT J
1070  REM  CALCULATE OUTFLOWS FOR FIXED-TERM LOANS
1080  FOR J = 1 TO 3
1090  READ X$
1100  IF C0(J,3) = 0 OR  INT (C0(J,2)) > D4 THEN 1200
1120  IF C0(J,3) > C0(J,1) THEN 1140
1130 C0(J,1) = C0(J,3)
1140 A2(1) = C0(J,1)
1150 A2(2) = C0(J,2)
1160  GOSUB 1700
1170 C0(J,2) = (C0(J,2) -  INT (C0(J,2))) + Y * 10000 + M * 100 + D2
1180  PRINT A$;" ";D1; TAB( 9);X$;" PAYMENT"; TAB( 30); - 1 * A2(1)
1190 C0(J,3) = C0(J,3) - A2(1)
1200  NEXT J
1210  REM  CALCULATE OUTFLOWS FOR CHARGE CARDS
1220  FOR J = 1 TO C4
1230  IF C1(J,5) > D4 OR C1(J,2) = 0 THEN 1340
1250  IF C1(J,2) > C1(J,4) THEN 1270
1260 C1(J,4) = C1(J,2)
1270 A2(1) = C1(J,4)
1280 A2(2) = C1(J,5) + 0.12
1290 X$ = C1$(J)
1300  GOSUB 1700
1310  PRINT A$;" ";D1; TAB( 9);C1$(J); TAB( 30); - 1 * A2(1)
1320 C1(J,2) = C1(J,2) - A2(1)
1330 C1(J,5) = Y * 10000 + M * 100 + D2
1340  NEXT J
1350  REM  CALCULATE OUTFLOWS FOR EXPENSES
1360  FOR J = 1 TO 25
1370  READ X$
1380  IF  INT (E0(J,2)) > D4 OR E0(J,1) = 0 THEN 1450
1400 A2(1) = E0(J,1)
1410 A2(2) = E0(J,2)
1420  GOSUB 1700
```

```
1430  PRINT A$;" ";D1; TAB( 9);X$; TAB( 30); - 1 * A2(1)
1440 E0(J,2) = (E0(J,2) -  INT (E0(J,2))) + Y * 10000 + M * 100 + D2
1450  NEXT J
1460 D1 = D1 + 1
1470 D4 = Y1 * 10000 + M1 * 100 + D1
1480 M = M1
1490 D2 = D1
1500 Y = Y1
1510  GOSUB 2890
1520  NEXT K1
1530 D3 = 1
1540 D2 = D(M1)
1550 M = M1
1560 Y = Y1
1570  GOSUB 2750
1580 D1 = 1
1590 M1 = M
1600 Y1 = Y
1610  GOSUB 2890
1620 D4 = Y1 * 10000 + M1 * 100 + D1
1630  PRINT  TAB( 10);"CASH IN:";I1; TAB( 25);"CASH OUT:";E1
1640  PRINT
1650  PRINT "DO YOU WANT TO SEE THE NEXT MONTH"
1655  PRINT "(Y/N) ";
1660  INPUT XO$
1670  IF XO$ = "Y" THEN 870
1680  IF XO$ = "N" THEN 3320
1690  GOTO 870
1700  REM  APPLY EXPENSES
1710 XO$ = ""
1720  IF BO - A2(1) >  = 0 THEN 1910
1725  PRINT
1730  PRINT "CASH NEEDED FOR: ";X$
1735  PRINT " $";A2(1);"ON HAND: ";BO
1737  PRINT
1740  PRINT "ENTER D=DELAY EXPENSE;"
1745  PRINT "C=USE CREDIT CARD ";
1750  INPUT XO$
1760  IF XO$ = "D" THEN 1930
1770  IF XO$ <  > "C" THEN 1740
1775  IF C4 = 1 THEN XO = 1: GOTO 1800
1780  PRINT "CREDIT CARD NUMBER (1-";C4;"OR ZERO) ";
1790  INPUT XO
1800  IF XO < 1 THEN 1740
1810  IF XO > C4 THEN 1780
1820  IF C1(XO,2) + A2(1) <  = C1(XO,3) THEN 1850
1830  PRINT "AVAILABLE ";C1$(XO);" CREDIT: $";C1(XO,3) - C1(XO,2)
1840  GOTO 1780
1850 C1(XO,2) = C1(XO,2) + A2(1)
1860 K = XO
1870 XO$ = "1"
1880  GOSUB 2160
1890 XO$ = ""
1900  GOTO 1920
1910 BO = BO - A2(1)
```

```
1920 E1 = E1 + A2(1)
1930 D3 =  INT ((A2(2) -  INT (A2(2))) * 100 + 0.5)
1940 Y =  INT (A2(2) / 10000)
1950 M =  INT ((A2(2) - Y * 10000) / 100)
1960 D2 =  INT ((A2(2) - (Y * 10000 + M * 100)))
1970  REM  CALCULATE NEXT DATE
1980  GOSUB 2510
1990  IF XO$ <  > "D" THEN 2010
2000  PRINT "EXPENSE IS DELAYED UNTIL ";M;"/";D2;"/";Y
2010  RETURN
2020  REM  ROUTINE TO ENTER CREDIT & CHARGE CARD DATA
2030  PRINT "NAME OF CREDIT CARD ";K;"(RETURN TO END)"
2040  INPUT C1$(K)
2050  IF C1$(K) <  = " " THEN 2350
2060  PRINT "ANNUAL INTEREST RATE ";
2070  INPUT C1(K,1)
2080  IF C1(K,1) < 0 THEN 2020
2090  PRINT "CURRENT BALANCE ";
2100  INPUT C1(K,2)
2110  IF C1(K,2) < 0 THEN 2060
2120  PRINT "CREDIT LIMIT ";
2130  INPUT C1(K,3)
2140  IF C1(K,3) < 0 THEN 2090
2150  IF C1(K,1) = 0 THEN 2290
2160 C1(K,4) =  INT (0.1 * C1(K,2) * 100 + 0.5) / 100
2170 IP = C1(K,1) / 100
2180 P1 = C1(K,2)
2190 A1 = C1(K,4)
2200  IF P1 <  = 0 THEN 2290
2210  GOSUB 3260
2215  PRINT
2220  PRINT A1;" PAYMENTS OF $";C1(K,4)
2225  PRINT "NEEDED TO PAY DEBT"
2227  PRINT
2230  PRINT "CHANGE AMOUNT (Y/N) ";
2240  INPUT X1$
2250  IF X1$ <  > "Y" THEN 2290
2260  PRINT "ENTER DESIRED PAYMENT AMOUNT ";
2270  INPUT C1(K,4)
2280  GOTO 2180
2290  IF XO$ = "1" THEN 2350
2300  PRINT "ENTER NEXT ";C1$(K);" BILLING DATE:"
2310 A2(2) = 0
2320  GOSUB 2470
2330  IF X1 =  - 1 THEN 2120
2340 C1(K,5) = A2(2)
2350  RETURN
2360  REM  ROUTINE TO CALCULATE EXPENSE FREQUENCIES
2370  REM  A2() ARRAY CONTAINS RESULTS
2380  PRINT "PERIODIC AMOUNT FOR ";X$;" ";
2390  INPUT A2(1)
2400  IF A2(1) <  = 0 THEN 2500
2410  PRINT "HOW MANY TIMES PER YEAR ";
2420  INPUT A2(2)
2430  IF A2(2) <  = 0 THEN 2380
```

```
2440  IF A2(2) < 100 THEN 2470
2450  PRINT "FREQUENCY CANNOT EXCEED 99 DAYS"
2460  GOTO 2410
2470  GOSUB 2990
2480  IF X1 =  - 1 THEN 2500
2490 A2(2) = A2(2) / 100 + Y * 10000 + M * 100 + D2
2500  RETURN
2510  REM  FIND NEXT MONTHLY, BIMONTHLY
2515  REM  OR QUARTERLY OCCURRENCE
2520  IF 24 / D3 <  >  INT (24 / D3) THEN 2740
2530  IF D3 = 24 THEN 2620
2540  FOR K = 1 TO 12 / D3
2550 M = M + 1
2560  IF M <  = 12 THEN 2590
2570 M = 1
2580 Y = Y + 1
2590  NEXT K
2600  RETURN
2610  REM  CALCULATE NEXT SEMIMONTHLY OCCURRENCE
2620  IF D2 <  > D(M) OR D2 <  > 1 THEN 2650
2630 D2 = 15
2640  GOTO 2690
2650  IF D2 > D(M) THEN 2680
2660 D2 = D2 + 15
2670  RETURN
2680 D2 = D2 - 15
2690 M = M + 1
2700  IF M <  = 12 THEN 2730
2710 Y = Y + 1
2720 M = 1
2730  RETURN
2740 D3 =  INT (365.25 / D3)
2750  REM  CALCULATE A DAY D3 DAYS FROM M/D2/Y
2760  IF D2 + D3 <  = D(M) THEN 2870
2770 D3 = D3 - (D(M) - D2)
2780 D2 = 0
2790 M = M + 1
2800  IF M <  = 12 THEN 2760
2810 Y = Y + 1
2820 M = 1
2830 D(2) = 28
2840  IF Y / 4 <  >  INT (Y / 4) THEN 2860
2850 D(2) = 29
2860  GOTO 2760
2870 D2 = D2 + D3
2880  RETURN
2890  REM  SUBROUTINE TO CALCULATE DAY OF WEEK
2900  IF Y > 1900 THEN 2920
2910 Y = Y + 1900
2920  IF M > 2 THEN 2945
2930 M = M + 12
2940 Y = Y - 1
2945 A = D2 + 2 * M +  INT (0.6 * (M + 1)) + Y +  INT (Y / 4)
2950 N = A -  INT (Y / 100) +  INT (Y / 400) + 2
2960 N =  INT ((N / 7 -  INT (N / 7)) * 7 + 0.5)
```

```
2970 A$ =  MID$ (D$,(N * 3) + 1,3)
2980  RETURN
2990  REM   ROUTINE TO ENTER DATE
3000  REM  DATE IS PASSED BACK IN M,D2 AND Y
3010 D(2) = 28
3020  PRINT "ENTER MONTH-DAY-YEAR (MMDDYY) ";
3030  INPUT X1
3040  IF X1 = 0 THEN 3160
3050  IF X1 =  - 1 THEN 3190
3060 M =  INT (X1 / 1E4)
3070  IF M > 12 OR M < 1 THEN 3020
3090 Y =  INT ((X1 / 100 -  INT (X1 / 100)) * 100 + 0.5)
3100  IF Y / 4 <  >  INT (Y / 4) THEN 3120
3110 D(2) = 29
3120 D2 =  INT ((X1 - (M * 1E4 + Y)) / 100)
3130  IF D2 < 1 THEN 3020
3140  IF D2 > D(M) THEN 3020
3150  GOTO 3190
3160 M = M1
3170 D2 = D1
3180 Y = Y1
3190  RETURN
3260  REM   SUBROUTINE TO DETERMINE TERM OF LOAN
3270  REM   IP=INTEREST RATE,P1=PRINCIPAL,A1=PAYMENT AMOUNT
3280  REM   REF. SOME COMMON BASIC PROGRAMS 3RD ED., P38
3285 A =  LOG (1 + (IP / 12) * 12)
3290 A1 =  - ( LOG (1 - (P1 * IP) / (12 * A1)) / A)
3300 A1 =  INT (A1 * 12 + 0.5)
3310  RETURN
3320  END
```

Critical Path Method (CPM)

This program calculates the time needed to complete a set of interrelated activities.

Before using the program, set up a CPM diagram and a precedence table. As you establish the network, make sure you include "dummy" activities in the diagram. These activities have no duration, but they may be necessary to indicate precedence of some activities over others in the network.

One feature of this program allows you to revise the network by changing activity durations and costs. In this way, you can observe changes in the critical path. Depending on the degree to which you revise the network, the path may shift by adding or eliminating activities.

Program Notes

This program currently allows 100 activities. If you want to change this, modify line 10 of the program as follows:

```
70 DIM A(I, 2), S(I), F(I), E(I, 2)
```

Replace the expression I with your maximum (for example, 15, 20, and so forth).

Negative slack time can exist for an activity. However, the program does not factor this into start times, end times or the critical path length.

Example

Washoe Valves is having its statewide sale-a-thon, a contest in which the company's three salespersons travel up Indiana, covering accounts in their territories and making as many sales as possible. At the end of their sale-a-thon, all three salespeople go to Chicago for a recap meeting.

Nance Graham, the sales manager, wants to know when each salesperson should start the trip, how much time each will spend driving and selling, and when to expect each salesperson to arrive in Chicago. Her precedence chart contains daily reimbursements to help calculate travel advances.

Activity	Nodal Sequence	Time (hours)	Cost
1. Gary drives to Terre Haute	1-2	2	30
2. Nance drives to Indianapolis	1-3	3	40
3. Lana drives to Muncie	1-4	3.5	49
4. Sell in Terre Haute	2-5	36	125
5. Sell in Indianapolis	3-6	48	320
6. Sell in Muncie	4-7	48	125
7. Gary drives to Lafayette	5-8	3	40
8. Nance drives to Chicago	6-11	5	35
9. Lana drives to Ft. Wayne, drops off valves	7-10	2	30
10. Sell in Lafayette	8-9	16	90
11. Lana drives to Chicago	9-11	4	52
12. Gary drives to Chicago	10-11	2	30

How does Nance run this program?

Answer: The minimum time needed to complete the sale-a-thon is 61 hours (the critical path length), and it will cost $966 in travel advances.

```
CRITICAL PATH METHOD

HOW MANY ACTIVITIES IN THIS NETWORK ?12

ENTER START,END NODES FOR ACT. 1 ?1,2
ENTER DURATION AND COST ?2,30

ENTER START,END NODES FOR ACT. 2 ?1,3
ENTER DURATION AND COST ?3,40

ENTER START,END NODES FOR ACT. 3 ?1,4
ENTER DURATION AND COST ?3.5,49

ENTER START,END NODES FOR ACT. 4 ?2,5
ENTER DURATION AND COST ?36,125

ENTER START,END NODES FOR ACT. 5 ?3,6
ENTER DURATION AND COST ?48,320

ENTER START,END NODES FOR ACT. 6 ?4,7
ENTER DURATION AND COST ?48,125

ENTER START,END NODES FOR ACT. 7 ?5,8
ENTER DURATION AND COST ?3,40

ENTER START,END NODES FOR ACT. 8 ?6,11
ENTER DURATION AND COST ?5,35

ENTER START,END NODES FOR ACT. 9 ?7,10
ENTER DURATION AND COST ?2,30

ENTER START,END NODES FOR ACT. 10 ?8,9
ENTER DURATION AND COST ?16,90

ENTER START,END NODES FOR ACT. 11 ?9,11
ENTER DURATION AND COST ?4,52

ENTER START,END NODES FOR ACT. 12 ?10,11
ENTER DURATION AND COST ?2,30

START END  EARLY  LATE
NODE  NODE START FINISH DUR. STACK COST
1       2     0      2      2    CRIT. 30
1       3     0      8      3    5     40
1       4     0      9      3.5  5.5   49
2       5     2      38     36   CRIT. 125
3       6     3      56     48   5     320
4       7     3.5    57     48   5.5   125
5       8     38     41     3    CRIT. 40
6       11    51     61     5    5     35
7       10    51.5   59     2    5.5   30
8       9     41     57     16   CRIT. 90
9       11    57     61     4    CRIT. 52
10      11    53.5   61     2    5.5   30
```

```
THE CRITICAL PATH LENGTH IS 61
TOTAL COST OF THIS NETWORK= 966

DO YOU WANT TO CHANGE ANY
ACTIVITY DURATIONS (Y/N) ?N
```

Practice Problems

1. Suppose Gary only spends 30 hours in Terre Haute. Will the critical path be different? Who will be able to wait before leaving, and for how long?

Answer: The critical path reduces to 56 hours. Gary can now wait one hour before leaving on his trip, and Lana can wait half an hour.

2. Nance may take her plane rather than drive. The flying time to Indianapolis is half an hour, and the time to Chicago is 45 minutes. She will have to pay a landing fee of $5 at Indianapolis, and $20 at Chicago, in addition to the costs shown above.

With this information, how long can she wait before leaving? What will the total cost be?

Answer: In the original network, Nance could wait five hours. She can now wait 11.75 hours before leaving. The total network cost is $991.

Program Listing

```
10  REM  CRITICAL PATH METHOD (CPM)
20  REM  A()=START AND END NODES FOR EACH ACTIVITY
30  REM  S()=EARLY START TIMES FOR EACH ACTIVITY
40  REM  F()=LATE FINISH TIMES FOR EACH ACTIVITY
50  REM  E()=DURATIONS AND COSTS OF NORMAL ACTIVITIES
60  REM  C()=DURATIONS AND COSTS OF CRASH ACTIVITIES
70  DIM A(100,2),S(100),F(100),E(100,2),C(100,2)
80  DEF  FN R(Z1) =  INT ((Z1 * 1000 + .5)) / 1000
90  PRINT "CRITICAL PATH METHOD"
100  PRINT
110  PRINT "HOW MANY ACTIVITIES IN THIS NETWORK ";
120  INPUT N
130  FOR I = 1 TO N
140  PRINT
150  PRINT "ENTER START,END NODES FOR ACT. ";I;" ";
160  INPUT A(I,1),A(I,2)
170  IF A(I,2) <  = A(I,1) THEN 200
190  IF A(I,2) < N THEN 260
200  PRINT "START NODE MUST BE NUMBERED LOWER"
210  PRINT " THAN END NODE, AND END NODE MUST"
220  PRINT " BE LESS THAN THE NUMBER OF ACTIVITIES."
230  PRINT "       *** TRY ENTRY AGAIN ***"
240  PRINT
250  GOTO 140
260  PRINT "ENTER DURATION AND COST ";
270  INPUT E(I,1),E(I,2)
280 S(I) = 0
290 F(I) = 0
300  NEXT I
310  REM  LOOP TO FIND EARLY START TIMES FOR NETWORK
```

```
320  FOR I = 1 TO N
330  IF S(A(I,2)) >  = S(A(I,1)) + E(I,1) THEN 350
340 S(A(I,2)) = S(A(I,1)) + E(I,1)
350  NEXT I
360 F(A(N,2)) = S(A(N,2))
370  REM  LOOP TO CALCULATE LATE FINISH TIMES FOR NETWORK
380  FOR I = N TO 1 STEP  - 1
390  IF F(A(I,1)) = 0 THEN 420
400  IF F(A(I,1)) > F(A(I,2)) - E(I,1) THEN 420
410  GOTO 430
420 F(A(I,1)) = F(A(I,2)) - E(I,1)
430  NEXT I
440 C1 = 0
450 L = 0
460  PRINT
470  REM  CALCULATE SLACK TIME IN S1
480  PRINT "START END  EARLY  LATE"
490  PRINT "NODE  NODE START FINISH DUR. STACK COST"
500  FOR I = 1 TO N
510  PRINT A(I,1); TAB( 7);A(I,2); TAB( 12);S(A(I,1)); TAB( 18);
520  PRINT F(A(I,2)); TAB( 25);E(I,1); TAB( 30);
530 S1 = F(A(I,2)) - S(A(I,1)) - E(I,1)
540  IF S1 > 0 THEN 590
545  IF L >  = F(A(I,2)) THEN 590
550  PRINT "CRIT.";
560 L = L + E(I,1)
570  GOTO 600
590  PRINT S1;
600  PRINT  TAB( 36);E(I,2)
610 C1 = C1 + E(I,2)
620  NEXT I
630  PRINT
640  PRINT "THE CRITICAL PATH LENGTH IS ";L
650  PRINT "TOTAL COST OF THIS NETWORK= ";C1
660  PRINT
670  PRINT "DO YOU WANT TO CHANGE ANY"
680  PRINT "ACTIVITY DURATIONS (Y/N) ";
690  INPUT A$
700  IF A$ = "N" THEN 870
710  IF A$ <  > "Y" THEN 660
720  PRINT
730  PRINT "WHICH ACTIVITY ";
740  INPUT I
750  IF I < 1 OR I > N THEN 720
770  PRINT "CURRENT DURATION IS ";E(I,1)
775  PRINT "COST = ";E(I,2)
780  PRINT "ENTER NEW DURATION AND COST ";
790  INPUT E(I,1),E(I,2)
800  PRINT "-----RECALCULATION NETWORK-----"
810  PRINT
820  FOR I = 1 TO N
830 S(I) = 0
840 F(I) = 0
850  NEXT I
860  GOTO 310
870  END
```

Reference

Brown, Kenneth S., and ReVelle, Jack B. *Quantitative Methods for Managerial Decisions.* Reading, Mass.: Addison-Wesley, 1979.

Program Evaluation and Review Technique (PERT)

This program calculates the minimum time needed to complete a complex project under uncertain conditions, and calculates the probability of the project's completion by a target time which you enter and can modify.

The program also calculates late start, early finish, and late finish times for each activity, as well as the slack time and standard deviation of expected activity times.

Before using the program, you must first organize the project, using PERT's graphing technique or a precedence table. To use the program, you must enter the number of activities in this project, including dummy activities. For each activity, you need to enter its start and end nodes, followed by the optimistic, most likely, and pessimistic duration estimates.

When you enter each activity, you must be sure each start node you enter is greater than the previous end node. If not, the program will ask you to reenter the start and end nodes.

Program Notes

This program is set for a maximum of 100 activities. If you want to change this, modify line 60 of the program as follows:

```
60 DIM A(I,2), S(I), F(I), E(I,2)
```

Replace the expression I with your maximum.

Negative slack time can exist for an activity. However, the program does not factor this into start times, end times, or the critical path length.

Example

Harriet just bought a Victorian house, advertised as a fixer-upper. She asked her contractor to provide her with three time estimates for each task involved in the remodeling. Her PERT chart and precedence table look like this:



Activity	Start Node	End Node	Optimistic Time	Most Likely Time	Pessimistic Time
1. Scrape exterior	1	2	1	2	4
2. Remove wallpaper	1	3	2	3	5
3. Replace cabinetry	1	4	3	4	7
4. Paint exterior	2	5	2	3	6
5. (dummy activity)	3	6	0	0	0
6. Lay kitchen floor	4	6	1	2	2.5
7. Paint exterior trim	5	7	1.5	2	4
8. Paint interior walls	6	7	2	3	3
9. Refinish wood floors	7	8	2	4	5

How will she run the program? What is the minimum time needed to complete the project? What is the probability of completing it one day sooner than expected?

Answer: The minimum time to complete the project is 12.916 days. The probability of completing the remodeling in 11.916 days is approximately 12.96%.

```
  PROGRAM EVALUATION
 AND REVIEW TECHNIQUE

   ENTER THE NUMBER OF
ACTIVITIES IN THIS NETWORK ?9

-----ACTIVITY 1-----
ENTER START NODE, END NODE ?1,2
ENTER THREE TIME ESTIMATES
FOR THIS ACTIVITY (A,M,B) ?1,2,4

-----ACTIVITY 2-----
ENTER START NODE, END NODE ?1,3
ENTER THREE TIME ESTIMATES
FOR THIS ACTIVITY (A,M,B) ?2,3,5

-----ACTIVITY 3-----
ENTER START NODE, END NODE ?1,4
ENTER THREE TIME ESTIMATES
FOR THIS ACTIVITY (A,M,B) ?3,4,7

-----ACTIVITY 4-----
ENTER START NODE, END NODE ?2,5
ENTER THREE TIME ESTIMATES
FOR THIS ACTIVITY (A,M,B) ?2,3,6

-----ACTIVITY 5-----
ENTER START NODE, END NODE ?3,6
ENTER THREE TIME ESTIMATES
FOR THIS ACTIVITY (A,M,B) ?0,0,0

-----ACTIVITY 6-----
ENTER START NODE, END NODE ?4,6
ENTER THREE TIME ESTIMATES
FOR THIS ACTIVITY (A,M,B) ?1,2,2.5

-----ACTIVITY 7-----
ENTER START NODE, END NODE ?5,7
ENTER THREE TIME ESTIMATES
FOR THIS ACTIVITY (A,M,B) ?1.5,2,4

-----ACTIVITY 8-----
ENTER START NODE, END NODE ?6,7
ENTER THREE TIME ESTIMATES
FOR THIS ACTIVITY (A,M,B) ?2,3,3

-----ACTIVITY 9-----
ENTER START NODE, END NODE ?7,8
ENTER THREE TIME ESTIMATES
FOR THIS ACTIVITY (A,M,B) ?2,4,5
----------------------------------------
```

```
ACTIVITY 1 (NODE 1 TO NODE 2)
IS A NON-CRITICAL EVENT.
EXPECTED DURATION: 2.167
STANDARD DEVIATION: .5
EARLY START: 0
LATE START: 1.333
EARLY FINISH: 2.167
LATE FINISH: 3.5
SLACK TIME: 1.333
-----------------------------------------

ACTIVITY 2 (NODE 1 TO NODE 3)
IS A NON-CRITICAL EVENT.
EXPECTED DURATION: 3.167
STANDARD DEVIATION: .5
EARLY START: 0
LATE START: 3.083
EARLY FINISH: 3.167
LATE FINISH: 6.25
SLACK TIME: 3.083
-----------------------------------------

ACTIVITY 3 (NODE 1 TO NODE 4)
IS A CRITICAL EVENT.
EXPECTED DURATION: 4.333
STANDARD DEVIATION: .667
START NO LATER THEN: 0
MUST BE COMPLETED BY: 4.33300001
-----------------------------------------

ACTIVITY 4 (NODE 2 TO NODE 5)
IS A NON-CRITICAL EVENT.
EXPECTED DURATION: 3.333
STANDARD DEVIATION: .667
EARLY START: 2.167
LATE START: 3.5
EARLY FINISH: 5.5
LATE FINISH: 6.833
SLACK TIME: 1.333
-----------------------------------------

ACTIVITY 5 (NODE 3 TO NODE 6)
IS A NON-CRITICAL EVENT.
EXPECTED DURATION: 0
STANDARD DEVIATION: 0
EARLY START: 3.167
LATE START: 6.25
EARLY FINISH: 3.167
LATE FINISH: 6.25
SLACK TIME: 3.083
-----------------------------------------

ACTIVITY 6 (NODE 4 TO NODE 6)
IS A CRITICAL EVENT.
EXPECTED DURATION: 1.917
```

```
STANDARD DEVIATION: .25
START NO LATER THEN: 4.333
MUST BE COMPLETED BY: 6.25
-----------------------------------------

ACTIVITY 7 (NODE 5 TO NODE 7)
IS A NON-CRITICAL EVENT.
EXPECTED DURATION: 2.25
STANDARD DEVIATION: .417
EARLY START: 5.5
LATE START: 6.833
EARLY FINISH: 7.75
LATE FINISH: 9.083
SLACK TIME: 1.333
-----------------------------------------

ACTIVITY 8 (NODE 6 TO NODE 7)
IS A CRITICAL EVENT.
EXPECTED DURATION: 2.833
STANDARD DEVIATION: .167
START NO LATER THEN: 6.25
MUST BE COMPLETED BY: 9.083
-----------------------------------------

ACTIVITY 9 (NODE 7 TO NODE 8)
IS A CRITICAL EVENT.
EXPECTED DURATION: 3.833
STANDARD DEVIATION: .5
START NO LATER THEN: 9.083
MUST BE COMPLETED BY: 12.916

THE CRITICAL PATH LENGTH IS 12.916
PLUS OF MINUS .886159128
ENTER DESIRED COMPLETION TIME
(0 TO END) ?11.916
PROBABILITY OF COMPLETION WITH
DURATION OF 11.916 IS .129551983

ENTER DESIRED COMPLETION TIME
(0 TO END) ?0
```

Practice Problems

1. A project is charted on the precedence table below:

Activity	Optimistic Time	Most Likely Time	Pessimistic Time
1-2	5	1	2
2-3	1	2	3
2-4	1	3	5
3-5	3	4	5
4-5	2	3	4
4-6	3	5	7
5-7	4	5	6
6-7	6	7	8
7-8	2	4	6
7-9	5	6	8
8-10	1	2	3
9-10	3	5	7

What is the critical path length? What is the probability of completing it within 30 weeks?

Answer: Critical path length is 27.25 weeks. The probability of completing the project within 30 weeks is 0.980952281.

2. Here is another precedence table:

Activity	Optimistic Time	Most Likely Time	Pessimistic Time
1-2	1	4	7
1-3	1	6	11
2-4	3	5	13
3-4	2	7	12
3-5	2	5	8
4-5	6	8	16
4-6	2	5	14
5-7	3	4	5
6-7	1	2	3

What are the slack times for the non-critical activities in this network? How many days will the project take if we want to be at least 90% sure of completing it on time?

Answer: Slack times: activity 1, 3 days; activity 3, 3 days; activity 5, 11 days; activity 7, 5 days; activity 9, 5 days. The project will take 29.725 days at the 90.0022732% confidence level.

Program Listing

```
10  REM  PROGRAM EVALUATION AND REVIEW TECHNIQUE (PERT)
20  REM   A()=START AND END NODES FOR EACH ACTIVITY
30  REM   S()=EARLY START TIMES FOR EACH ACTIVITY
40  REM   F()=LATE FINISH TIMES FOR EACH ACTIVITY
50  REM  E()=EXPECTED DURATIONS AND VARIANCES OF ACTIVITIES
60  DIM A(100,2),S(100),F(100),E(100,2)
65  DEF  FN R(Z1) =  INT ((Z1 * 1000 + .5)) / 1000
70  PRINT "  PROGRAM EVALUATION"
80  PRINT " AND REVIEW TECHNIQUE"
90  PRINT
```

```
100  PRINT "   ENTER THE NUMBER OF"
110  PRINT "ACTIVITIES IN THIS NETWORK ";
120  INPUT N
130  PRINT
140  FOR I = 1 TO N
150  PRINT
160  PRINT "-----ACTIVITY ";I;"-----"
170  PRINT "ENTER START NODE, END NODE ";
180  INPUT A(I,1),A(I,2)
190  IF A(I,2) <  = A(I,1) THEN 220
200  IF A(I,2) < N THEN 280
220  PRINT " START NODE MUST BE NUMBERED LOWER"
230  PRINT "  THEN END NODE, AND END NODE MUST"
240  PRINT "BE LESS THAN THE NUMBER OF ACTIVITIES."
250  PRINT "     *** TRY ENTRY AGAIN ***"F
260  PRINT
270  GOTO 150
280  PRINT "ENTER THREE TIME ESTIMATES"
290  PRINT "FOR THIS ACTIVITY (A,M,B) ";
300  INPUT A1,M,B
310  REM  E(I,1) IS THE EXPECTED DURATION
320 E(I,1) =  FN R((A1 + M * 4 + B) / 6)
330  REM  E(I,2) IS THE ACTIVITY VARIANCE
340 E(I,2) =  FN R((B - A1) / 6) ^ 2
350 S(I) = 0
360 F(I) = 0
370  NEXT I
380  REM  LOOP TO FIND EARLY START TIMES FOR NETWORK
390  FOR I = 1 TO N
400  IF S(A(I,2)) >  = S(A(I,1)) + E(I,1) THEN 420
410 S(A(I,2)) = S(A(I,1)) + E(I,1)
420  NEXT I
430 F(A(N,2)) = S(A(N,2))
440  REM  LOOP TO CALCULATE LATE FINISH TIMES FOR NETWORK
450  FOR I = N TO 1 STEP  - 1
460  IF F(A(I,1)) = 0 THEN 490
470  IF F(A(I,1)) > F(A(I,2)) - E(I,1) THEN 490
480  GOTO 500
490 F(A(I,1)) = F(A(I,2)) - E(I,1)
500  NEXT I
510 V = 0
520 C = 0
530 L = 0
540  FOR I = 1 TO N
550  REM  CALCULATE SLACK TIME IN S1
560 S1 = F(A(I,2)) - S(A(I,1)) - E(I,1)
565 S1 =  INT (S1 * 1E6 + .5) / 1E6
570  PRINT "-----------------------------------------"
580  PRINT
590  PRINT "ACTIVITY ";I;" (NODE ";A(I,1);" TO NODE ";A(I,2);")"
600  PRINT "IS A ";
610  IF S1 <  = 0 THEN 630
620  PRINT "NON-";
630  PRINT "CRITICAL EVENT."
640  PRINT "EXPECTED DURATION: ";E(I,1)
```

```
650  PRINT "STANDARD DEVIATION: "; SQR (E(I,2))
660  IF S1 > 0 THEN 740
670  PRINT "START NO LATER THEN: ";S(A(I,1))
680  PRINT "MUST BE COMPLETED BY: ";F(A(I,2))
690  REM  ACCUMULATE PATH LENGTH IN L, VARIANCE IN V
700  IF L >  = F(A(I,2)) THEN 720
710 L = F(A(I,2))
720 V = V + E(I,2)
730  GOTO 790
740  PRINT "EARLY START: ";S(A(I,1))
750  PRINT "LATE START: ";F(A(I,2)) - E(I,1)
760  PRINT "EARLY FINISH: ";S(A(I,1)) + E(I,1)
770  PRINT "LATE FINISH: ";F(A(I,2))
780  PRINT "SLACK TIME: ";S1
790  NEXT I
800  PRINT
810  PRINT "THE CRITICAL PATH LENGTH IS ";L
820 P =  SQR (V)
830  PRINT "PLUS OF MINUS ";P
840  PRINT "ENTER DESIRED COMPLETION TIME"
845  PRINT "(0 TO END) ";
850  INPUT D
860  IF D <  = 0 THEN 1010
870  REM  CALCULATE Z-SCORE FOR DESIRED DURATION
880 Y = (D - L) / P
890  REM  CALCULATE CUMULATIVE AREA UNDER NORMAL DISTRIBUTION
900  REM  REF: SOME COMMON BASIC PROGRAMS, 3RD ED. P.128
910 R =  EXP ( - (Y ^ 2) / 2) / 2.5066282746
920 Z = Y
930 Y = 1 / (1 + .33267 *  ABS (Y))
940 T = 1 - R * (.4361836 * Y - .1201676 * Y ^ 2 + .937298 * Y ^ 3)
950  IF Z >  = 0 THEN 970
960 T = 1 - T
970  PRINT "PROBABILITY OF COMPLETION WITH"
980  PRINT "DURATION OF ";D;" IS ";T
990  PRINT
1000  GOTO 840
1010  END
```

References

Brown, Kenneth S., and ReVelle, Jack B. *Quantitative Methods for Managerial Decisions.* Reading, Mass.: Addison-Wesley, 1979.

MacCrimmon, K.R., and Ryavec, C.A. *An Analytical Study of the PERT Assumptions.* Santa Monica, Calif.: Rand Corporation, Memo RM-3408-PR, 1962.

Moore, Franklin G., and Hendrick, Thomas E. *Production/Operations Management* (3rd ed.). Homewood, Ill.: Richard D. Irwin, 1977.

Transportation Algorithm

This program allows you to allocate a resource from multiple sources of supply to multiple destinations in the most cost-efficient way. The resource can be anything such as manufactured goods, personnel, and so forth. Linear programming can be used to solve this type of problem, but here you do not need to convert costs into an objective function, nor do you need to express data as coefficients in a series of linear equations.

To use this program, you will need to know how many sources of supply are available, as well as the supply capacity for each source. The number of demand destinations, as well as their exact demand for the resource, are also needed. Finally, you need to know the cost of transporting the resource from each source to each destination. The program will ask you for all of this information when you run it, so be sure to have it organized before entering it into the computer.

If available supply does not equal prevailing demand, the program automatically assigns the difference to a dummy source (supply less than demand) or dummy destination (supply greater than demand). Each assignment of the resource, its transportation cost per unit and its total assignment cost, print out at the end of the program. If dummy variables exist in a given problem, these assignments are printed out for your information.

Program Notes

This program allows for ten sources and ten destinations. If you want to change this to another maximum, modify lines 20 and 30 as follows:

```
20 DIM S(I,2), D(J,2), S1(I + J,2), C(I,J), A(I,J), Y(X,2), M(3)
30 DIM R1(I), K1(J)
```

Replace the expression I with the maximum number of sources, and replace J with the maximum number of destinations. Replace X with the maximum number of sources plus the maximum number of destinations minus one.

You may want to change the program to receive data through DATA statements, rather than INPUT statements. If so, modify the program as shown in the "Option" section.

Example

Smiling Jack owns an organic crop dusting operation. He has three planes which have capacities for 65, 150, and 80 gallons of insecticide each. Tomorrow, four farms need dusting. Jack calculates that, based on the sizes of the fields, they will need 100, 45, 90, and 60 gallons for the fields, respectively. Since each plane has a different capacity, and since the fields are in four different counties, Jack estimates the costs as follows for each gallon of insecticide: For plane 1 to field 1, 0.05; to field 2, 0.12; to field 3, 0.08; to field 4, 0.11. For plane 2 to field 1, 0.04; to field 2, 0.03; to field 3, 0.06; to field 4, 0.04. For plane 3 to field 1, 0.09; to field 2, 0.14; to field 3, 0.13; to field 4, 0.18. How does Jack enter this information, what are the assignments for tomorrow, and what is the total transportation cost?

Answer: The optimal assignments are: Plane 1 to field 1, where it will spray 20 gallons, and on to field 3 where it will spray 45 gallons. Plane 2 goes to field 2 first, spraying 45 gallons, then proceeds to field 3, where it uses 45 gallons of insecticide. Finally, Plane 2 goes on to field 4, where it uses the last 60 gallons of spray. Plane 3 goes to field 1 to complete the job which Plane 1 did partially. The total cost, based on those entered, is estimated at $18.25.

```
TRANSPORTATION ALGORITHM

NUMBER OF SOURCES ?3
NUMBER OF DESTINATIONS ?4
CAPACITY FOR SOURCE 1 ?65
CAPACITY FOR SOURCE 2 ?150
CAPACITY FOR SOURCE 3 ?80
DEMAND FROM DESTINATION 1 ?100
DEMAND FROM DESTINATION 2 ?45
DEMAND FROM DESTINATION 3 ?90
DEMAND FROM DESTINATION 4 ?60
TRANSPORTATION COSTS:
FROM SOURCE 1 TO DESTINATION 1 ?.05
FROM SOURCE 1 TO DESTINATION 2 ?.12
FROM SOURCE 1 TO DESTINATION 3 ?.08
FROM SOURCE 1 TO DESTINATION 4 ?.11
FROM SOURCE 2 TO DESTINATION 1 ?.04
FROM SOURCE 2 TO DESTINATION 2 ?.03
FROM SOURCE 2 TO DESTINATION 3 ?.06
FROM SOURCE 2 TO DESTINATION 4 ?.04
FROM SOURCE 3 TO DESTINATION 1 ?.09
FROM SOURCE 3 TO DESTINATION 2 ?.14
FROM SOURCE 3 TO DESTINATION 3 ?.13
FROM SOURCE 3 TO DESTINATION 4 ?.18

SOURCE DEST # UNITS COST TOTAL COST
1       1    20      .05  1
SOURCE DEST # UNITS COST TOTAL COST
1       3    45      .08  3.6
SOURCE DEST # UNITS COST TOTAL COST
2       2    45      .03  1.35
SOURCE DEST # UNITS COST TOTAL COST
2       3    45      .06  2.7
SOURCE DEST # UNITS COST TOTAL COST
2       4    60      .04  2.4
SOURCE DEST # UNITS COST TOTAL COST
3       1    80      .09  7.2

TOTAL COST OF SOLUTION: 18.25


DO YOU WANT TO RE-RUN THIS
PROGRAM WITH NEW DATA (Y/N) ?N
```

Practice Problems

1. The Skinheads Motorcycle Enthusiasts Society has three chapters in the state, and three imminent social engagements with competing clubs. Based on intelligence reports, the Skinheads know that they will encounter 75, 19, and 22 people respectively. Their three chapters have 35, 20, and 61 members. The mileage from chapter 1 to location 1 is 35 miles; to location 2, 80 miles; and to location 3, 60 miles. From chapter 2 to location 1, the distance is 90 miles; to location 2, 40 miles; and to location 3, 55 miles. From chapter 3 to location 1, the distance is 50 miles; to location 2, 28 miles; and to location 3, 65 miles.

How should people be assigned? How far, in miles, will everyone in the club have traveled to reach the destinations?

Answer: 35 persons from chapter 1 to location 1; 20 people from chapter 2 to location 3; 40 people from chapter 3 to location 1; 19 people from chapter 3 to location 2, and two from chapter 3 to location 3. The total miles traveled (assuming one person per bike): 4,987.

2. Given the following table, what is the optimal transportation mix? How much does it cost?

Project	Weekly Demand	Plant	Weekly Capacity
A	170	J	130
B	250	K	200
C	100	L	190

Costs:

From	To A	To B	To C
J	$2	$ 5	$5
K	9	13	9
L	2	4	6

Answer: 70 units from Plant J to Project Site A; 60 units from Plant J to Project B; 100 units from Plant K to Project A; 100 units from Plant K to Project C; and 190 units from Plant L to Project B.

Program Listing

```
10  REM  TRANSPORTATION ALGORITHM
20  DIM S(10,2),D(10,2),S1(20,2),C(10,10)
30  DIM A(10,10),Y(19,2),M(3),R1(10),K1(10)
40  PRINT "TRANSPORTATION ALGORITHM"
50  PRINT
60  PRINT "NUMBER OF SOURCES ";
70  INPUT S2
80  IF S2 < 1 THEN 60
90  PRINT "NUMBER OF DESTINATIONS ";
100  INPUT D1
110  IF D1 < 1 THEN 90
120  REM  ENTER SUPPLY CAPACITY FOR EACH SOURCE
130 T1 = 0
140  FOR R = 1 TO S2
150  PRINT "CAPACITY FOR SOURCE ";R;" ";
160  INPUT S(R,1)
170 S(R,2) = S(R,1)
180 T1 = T1 + S(R,1)
190  NEXT R
200 T2 = 0
210  REM   READ DATA LIST OF DEMAND FROM
215  REM   EACH DESTINATION
220  FOR R = 1 TO D1
230  PRINT "DEMAND FROM DESTINATION ";R;" ";
240  INPUT D(R,1)
250 D(R,2) = D(R,1)
260 T2 = T2 + D(R,1)
270  NEXT R
280  REM   LOOP TO READ TRANSPORTATION COSTS
290  PRINT "TRANSPORTATION COSTS: "
300  FOR R = 1 TO S2
310  REM   INITIALIZE ELEMENTS F S1() ARRAY
320 S1(R,1) = 0
```

```
330 S1(R,2) = 0
340  FOR K = 1 TO D1
345 A(R,K) = 0
350  PRINT "FROM SOURCE ";R;" TO DESTINATION ";K;" ";
360  INPUT C(R,K)
370  IF C(R,K) < 0 THEN 350
380  NEXT K
390  NEXT R
400  REM  THE MATRIX HAS BEEN ENTERED--START FIRST SOLUTION PHASE
410 S0 = 0
420 D0 = 0
430  IF T1 >  = T2 THEN 480
440  REM  SUPPLY MUST EQUAL DEMAND; SET UP DUMMY ROWS & COLUMNS
450 S(S2 + 1,1) = T2 - T1
460 S(S2 + 1,2) = T2 - T1
470 S0 = 1
475  GOTO 510
480  IF T2 = T1 THEN 510
490 D(D1 + 1,1) = T1 - T2
500 D(D1 + 1,2) = T1 - T2
505 D0 = 1
510 D2 = 0
520 T3 = 0
530  REM   START SOLUTION WITH NORTHWEST CORNER RULE
540  FOR R = 1 TO S2 + S0
550  REM  IF SUPPLY AT ROW R EXHAUSTED, MOVE TO NEXT SOURCE
560  IF S(R,2) = 0 THEN 770
570  REM  ALLOCATE SUPLY TO DEMAND
580  FOR K = 1 TO D1 + D0
590  REM  IF DESTINATION K FILLED, INCREMENT COLUMN INDEX
600  IF D(K,2) = 0 THEN 760
610  IF S(R,2) = 0 THEN 760
620  IF S(R,2) < D(K,2) THEN 690
630  REM  SET UP STONE SQUARE IF DEMAND<=SUPPLY
640 A(R,K) = D(K,2)
650 S(R,2) = S(R,2) - D(K,2)
660 D(K,2) = 0
670  GOTO 720
680  REM  SET UP STONE SQUARE IF DEMAND > SUPPLY
690 A(R,K) = S(R,2)
700 D(K,2) = D(K,2) - S(R,2)
710 S(R,2) = 0
720 D2 = D2 + 1
730 T3 = T3 + (A(R,K) * C(R,K))
740 S1(D2,1) = R
750 S1(D2,2) = K
760  NEXT K
770  NEXT R
780  REM  CHECK SOLUTION FOR FIRST-STAGE DEGENERACY
790  IF D2 = S2 + S0 + D1 + D0 - 1 THEN 1140
800  REM  SOLVE DEGENERATE SOLUTION
810 R = 0
820 K = 0
830 I = 0
840 I = I + 1
```

```
850  IF A(S1(I,1),S1(I,2)) = D(S1(I,2),1) THEN 870
860  IF A(S1(I,1),S1(I,2)) <  > S(S1(I,1),1) THEN 900
870 R = S1(I,1) + 1
880 K = S1(I,2)
890  GOTO 1030
900  IF I < D2 + D0 THEN 840
910  REM  IF R & K ARE ZERO, THE MATRIX IS NOT DEGENERATE
920  IF R + K = 0 THEN 1140
930  IF S1(I - 1,2) = K THEN 960
940 K = S1(I - 1,2)
950  GOTO 1000
960  IF K = D2 + D0 THEN 990
970 K = K + 1
980  GOTO 1000
990 K = K - 1
1000  REM  INSERT A NEW STONE SQUARE IN THE SOLUTION
1010  IF K > S1(I,2) THEN 1030
1020 I = I - 1
1030  FOR J = D2 + 1 TO I + 1 STEP  - 1
1040 S1(J,1) = S1(J - 1,1)
1050 S1(J,2) = S1(J - 1,2)
1055 M0 = J
1060  NEXT J
1070 S1(M0,1) = R
1080 S1(M0,2) = K
1090 Y(I,1) = 0
1100 Y(I,2) = 0
1110 D2 = D2 + 1
1120  GOTO 790
1130  REM  CALCULATE REM VALUES
1140  FOR I = 1 TO D1 + D0
1150 K1(I) =  - 9E4
1160  NEXT I
1170  FOR I = 1 TO S2 + S0
1180 R1(I) =  - 9E4
1190  NEXT I
1200 R1(S1(1,1)) = 0
1210 K1(S1(1,2)) = C(S1(1,1),S1(1,2))
1220 R = 1
1230 K = 1
1240 I = 1
1250 I = I + 1
1260  IF K1(S1(I,2)) <  >  - 9E4 THEN 1300
1270  IF R1(S1(I,1)) =  - 9E4 THEN 1330
1280 K1(S1(I,2)) = C(S1(I,1),S1(I,2)) - R1(S1(I,1))
1290 K = K + 1
1300  IF R1(S1(I,1)) <  >  - 9E4 THEN 1330
1310 R1(S1(I,1)) = C(S1(I,1),S1(I,2)) - K1(S1(I,2))
1320 R = R + 1
1330  IF I < D2 THEN 1250
1340  IF K < D1 + D0 THEN 1240
1350  IF R < S2 + S0 THEN 1240
1360 I = 1
1370 M(1) = 0
1380  REM  FIND AN ELEMENT WITH THE LOWEST INDEX
```

```
1390  FOR R = 1 TO S2 + SO
1400  FOR K = 1 TO D1 + DO
1410  IF R <  > S1(I,1) THEN 1450
1420  IF K <  > S1(I,2) THEN 1450
1430 I = I + 1
1440  GOTO 1490
1450  IF M(1) < C(R,K) - R1(R) - K1(K) THEN 1490
1460 M(1) = C(R,K) - R1(R) - K1(K)
1470 M(2) = R
1480 M(3) = K
1490  NEXT K
1500  NEXT R
1510  IF M(1) >  = 0 THEN 2790
1520  REM  FIND A CLOSED PATH FROM SQUARE AT ROW R, COL. K
1530 Y(1,1) = M(2)
1540 Y(1,2) = M(3)
1550 Q = 1
1560  IF M(2) = S2 + SO THEN 1960
1570  REM  MO=CURRENT ROW TO SEARCH ON;
1575  REM  M1=START COLUMN TO SEARCH ON
1580 MO = Y(Q,1)
1590 M1 = 1
1600  REM  START ROW SEARCH
1610 I = 0
1620 I = I + 1
1630  IF S1(I,1) > MO THEN 1670
1640  IF S1(I,1) < MO THEN 1660
1650  IF S1(I,2) >  = M1 THEN 1720
1660  IF I < D2 THEN 1620
1670  IF Q <  > 1 THEN 1700
1680  PRINT "MATRIX IS DEGENERATE"
1690  GOTO 2410
1700  REM  AT THIS POINT, NO ROW NEIGHBORS EXIST
1710  GOTO 1830
1720  REM  MAKE SURE V(I) IS NOT ALREADY ON THE CLOSED PATH
1730 XO = 0
1740  FOR J = 1 TO Q
1750  IF S1(I,1) <  > Y(J,1) THEN 1780
1760  IF S1(I,2) <  > Y(J,2) THEN 1780
1770 XO = 1
1780  NEXT J
1790  IF XO = 0 THEN 1890
1800 M1 = S1(I,1) + 1
1810  IF M1 <  = D1 + DO THEN 1660
1820  REM  ROW SEARCH FAILED;
1825  REM  SET NEXT COORDINATES FOR COLUMN SEARCH
1830 P = Y(Q,2)
1840 P1 = Y(Q,1) + 1
1850 Y(Q,1) = 0
1860 Y(Q,2) = 0
1870 Q = Q - 1
1880  GOTO 2000
1890 Q = Q + 1
1900 Y(Q,1) = S1(I,1)
1910 Y(Q,2) = S1(I,2)
```

```
1920  IF Q <  = 2 THEN 1950
1930  REM  IF PATH CLOSES ON A ROW SEARCH,
1935  REM  EXIT SEARCH ROUTINE
1940  IF Y(Q,2) = M(3) THEN 2340
1950 M1 = Y(Q,2) + 1
1960  REM  COLUMN SEARCH AREA
1970  REM  P=COLUMN NUMBER TO SEARCH ON
1975  REM  P1=STARTING ROW FOR SEARCH
1980 P = Y(Q,2)
1990 P1 = 1
2000 K = 0
2010 K = K + 1
2020  IF S1(K,1) < P1 THEN 2040
2030  IF S1(K,2) = P THEN 2120
2040  IF K < D2 THEN 2010
2050  REM  COLUMN SEARCH FAILURE;
2055  REM  SET NEW COORDINATES FOR ROW SEARCH
2060 M0 = Y(Q,1)
2070 M1 = Y(Q,2) + 1
2080 Y(Q,1) = 0
2090 Y(Q,2) = 0
2100 Q = Q - 1
2110  GOTO 1610
2120 X0 = 0
2130  REM  LOOKUP ROUTINE:
2135  REM  CHECK FOR ALREADY-USED STONE SQUARE
2140  FOR J = 1 TO Q
2150  IF S1(K,1) <  > Y(J,1) THEN 2180
2160  IF S1(K,2) <  > Y(J,2) THEN 2180
2170 X0 = 1
2180  NEXT J
2190  IF X0 = 0 THEN 2250
2200 P1 = S1(K,1) + 1
2210  IF P1 <  = S2 + S0 THEN 2040
2220  GOTO 2050
2230  REM  A UNIQUE STONE SQUARE WAS FOUND---
2240  REM  ADD IT TO THE CLOSED PATH ARRAY.
2250 Q = Q + 1
2260 Y(Q,1) = S1(K,1)
2270 Y(Q,2) = S1(K,2)
2280  REM   IF PATH CLOSES ON COLUMN SEARCH,
2285  REM  EXIT SEARCH ROUTINE
2290  IF Y(Q,1) = M(2) THEN 2340
2300 P1 = Y(Q,1) + 1
2310 M0 = Y(Q,1)
2320 M1 = Y(Q,2) + 1
2330  GOTO 1610
2340  REM  FIND LOWEST-ALLOCATION STONE
2345  REM  SQUARE ON CLOSED PATH
2350 X0 = A(Y(2,1),Y(2,2))
2360  FOR K = 4 TO Q STEP 2
2370  IF X0 <  = A(Y(K,1),Y(K,2)) THEN 2390
2380 X0 = A(Y(K,1),Y(K,2))
2390  NEXT K
2400  REM  ALTERNATELY ADD & SUBTRACT X0
```

```
2410 P = 0
2420  FOR K = 1 TO Q
2430 KO = K / 2
2435  IF KO =  INT (KO) THEN 2460
2440 A(Y(K,1),Y(K,2)) = A(Y(K,1),Y(K,2)) + XO
2450  GOTO 2630
2460 A(Y(K,1),Y(K,2)) = A(Y(K,1),Y(K,2)) - XO
2470  IF A(Y(K,1),Y(K,2)) > 0 THEN 2630
2480  REM  DELETE ANY SQUARES WITH A ZERO ALLOCATION
2490 I = 0
2500 P = P + 1
2510  REM  IF P>1, MATRIX WILL BE DEGENERATE
2515  REM  IF SQUARE IS DELETED; SKIP IT
2520  IF P > 1 THEN 2630
2530 I = I + 1
2540  IF S1(I,1) <  > Y(K,1) THEN 2530
2550  IF S1(I,2) <  > Y(K,2) THEN 2530
2560  FOR J = I TO D2
2570 S1(J,1) = S1(J + 1,1)
2580 S1(J,2) = S1(J + 1,2)
2590  NEXT J
2600 S1(D2,1) = 0
2610 S1(D2,2) = 0
2620 D2 = D2 - 1
2630  NEXT K
2640  REM  INSERT NEW STONE SQUARE FROM
2645  REM   FIRST ELEMENT OF Y()
2650 I = 0
2660 I = I + 1
2665  IF I > D2 THEN 2700
2670  IF Y(1,1) > S1(I,1) THEN 2660
2680  IF Y(1,1) < S1(I,1) THEN 2700
2690  IF Y(1,2) > S1(I,2) THEN 2660
2700  FOR J = D2 TO I STEP  - 1
2710 S1(J + 1,1) = S1(J,1)
2720 S1(J + 1,2) = S1(J,2)
2730  NEXT J
2740 S1(I,1) = Y(1,1)
2750 S1(I,2) = Y(1,2)
2760 D2 = D2 + 1
2770  REM  END OF RE-ALLOCATION;
2775  REM  REITERATE MODI CHECK
2780  GOTO 1140
2790  REM  DISPLAY RESULTS AND COST OF SOLUTION
2800  PRINT
2810  IF DO + SO = 0 THEN 2870
2820  PRINT "*** UNBALANCED SOLUTION ***"
2830  IF DO = 0 THEN 2850
2840  PRINT "EXCESS SUPPLY (";D(D1 + DO,1);")"
2845  PRINT "ASSIGNED TO DESTINATION ";D1 + DO
2850  IF SO = 0 THEN 2870
2860  PRINT "EXCESS DEMAND (";S(S2 + SO,1);")"
2865  PRINT "ASSIGNED TO SOURCE ";S2 + SO
2870 XO = 0
2880  FOR I = 1 TO D2
```

```
2890  PRINT "SOURCE DEST # UNITS COST TOTAL COST"
2900  PRINT S1(I,1); TAB( 8);S1(I,2); TAB( 13);A(S1(I,1),S1(I,2));
2905  PRINT  TAB( 21);C(S1(I,1),S1(I,2)); TAB( 26)
2910 J = C(S1(I,1),S1(I,2)) * A(S1(I,1),S1(I,2))
2920  IF J > 0 THEN 2950
2930  PRINT "DUMMY"
2940  GOTO 2970
2950 XO = XO + J
2960  PRINT J
2970  NEXT I
2980  PRINT
2990  PRINT "TOTAL COST OF SOLUTION: ";XO
3000  PRINT
3010  PRINT
3020  PRINT "DO YOU WANT TO RE-RUN THIS"
3030  PRINT "PROGRAM WITH NEW DATA (Y/N) ";
3040  INPUT XO$
3050  IF XO$ = "Y" THEN 50
3060  END
```

Option

If you want to avoid using INPUT statements for data entry, you can change the program to read input from DATA statements. This is especially useful if you intend to enter a large transportation problem, or if you want to run the program repeatedly with slightly different data without reentering the supply, demand and cost figures. Modify the statements below to allow for this feature.

```
151  REM   THIS DATA SHOWN TO SOLVE PROBLEM #1.
152  REM   PUT YOUR SUPPLY DATA HERE.
153  DATA   65,150,80
160  READ S(R,1)
165  PRINT S(R,1)
170 S(R,1) = S(R,1)
180 T1 = T1 + S(R,1)
190  NEXT R
200 T2 = 0
210  REM   READ DATA LIST OF DEMAND FROM
215  REM   EACH DESTINATION
220  FOR R = 1 TO D1
230  PRINT "DEMAND FROM DESTINATION ";R;" ";
231  REM   PUT DEMAND DATA HERE.
232  DATA   100,45,90,60
240  READ D(R,1)
245  PRINT D(R,1)
250 D(R,2) = D(R,1)
260 T2 = T2 + D(R,1)
270  NEXT R
280  REM   LOOP TO READ TRANSPORTION COSTS
290  PRINT "TRANSPORTATION COSTS: "
300  FOR R = 1 TO S2
310  REM   INITIALIZE ELEMENTS F S1() ARRAY
320 S1(R,1) = 0
330 S1(R,2) = 0
```

```
340   FOR K = 1 TO D1
350   PRINT "FROM SOURCE ";R;" TO DESTINATION ";K;" ";
351   REM   PUT TRANSPORTATION COST DATA HERE.
352   DATA   .05,.12,.08,.11,.04,.03,.06,.04,.09,.14,.13,.18
360   READ C(R,K)
365   PRINT C(R,K)
380   NEXT K
390   NEXT R
```

Also delete lines 2980 through 3060.

References

Chase, Richard B., and Aquilano, Nicholas J. *Production and Operations Management.* Homewood, Ill.: Richard D. Irwin, Inc., 1977.

Levin, Richard I., and Kirkpatrick, Charles A. *Quantitative Approaches to Management* (3rd ed.). New York: McGraw-Hill, 1975.

Swedish Machine (Queuing Theory)

This is the classic problem where you have X repairmen servicing Y machines. The machines are statistically identical. Their times-to-failure follow the exponential law, characterized by the mean time-to-failure. The repairmen are also statistically identical; their repair completion times follow the exponential law and are characterized by its expected value. All elements are mutually independent.

This program is especially useful in that it can provide a cash flow analysis that can help project the feasibility of a particular machines-to-repairmen ratio, given the repairmen's wages, machine revenue, and overhead costs.

To use the program, enter the number of machines, the mean time-to-failure for a machine, the number of repairmen, and the mean time to repair a machine. You may use any time unit base you wish, as long as you use the same time units throughout the program. A variety of system characteristics are output. If you choose to obtain a cash flow analysis, you must also enter the cost for one repairman per unit of time, the cost of possessing a machine (overhead) per unit of time, and the amount of revenue produced by a machine per unit of time. You may use any monetary unit you wish (pennies, dollars, thousands of dollars, pesos, whatever).

Example

Ace Laundromat has a total of 50 machines operating throughout the city. The machines have a mean time-to-failure of 300 hours, and there are currently three repairmen. Each repairman requires 24 hours to repair a machine. At any time, how many machines can be expected to be operating? How many machines are being repaired? How many are waiting to be repaired? What is the mean down time per machine? How many repairmen are idle? Repairmen cost $5.25 per hour (including fringe benefits, and so forth). Cost of possessing a machine is the overhead involved, such as lease or purchase payments, insurance payments, pro-rated administrative expenses, depreciation expense, and so forth. In this example the cost of possessing a machine is $36.00 per month, or $0.05 per hour. What cash flow do the machines generate if they each produce revenues of $0.50 per hour?

Answer: 37 machines can be expected to be operational at any time, three are being repaired, and ten are waiting to be repaired. The mean down time per machine is about 105 hours. The 50 machines will produce an average revenue of $0.268843 per hour.

```
SWEDISH MACHINE

INPUT THE NUMBER OF MACHINES;
COUNT MUST EXCEED ONE. ?50
INPUT MEAN TIME-TO-FAILURE
FOR A MACHINE ?300
INPUT NUMBER OF REPAIRMEN ?3
INPUT MEAN REPAIR TIME (PER MACHINE)
FOR A REPAIRMAN ?24


THE SYSTEM IF SAID TO BE 'IN STATE J'
IF J MACHINES ARE IN A FAILED
CONDITION.  THE STATIONARY PROBABILITY
DISTRIBUTION OVER THE POSSIBLE STATES,
0 THRU 50, AND OTHER CHARACTERISTICS
OF INTEREST, FOLLOW.
```

STATE	PROBABILITY	NO. MACHINES OPERATING	NO. MACHINES WAITING	NO. REPAIRMEN IDLE
0	1.963E-03	50	0	3
1	7.852E-03	49	0	2
2	.015391	48	0	1
3	.0197	47	0	0
4	.024691	46	1	0
5	.030288	45	2	0
6	.036345	44	3	0
7	.042645	43	4	0
8	.0489	42	5	0
9	.054768	41	6	0
10	.059879	40	7	0
11	.063871	39	8	0
12	.066426	38	9	0
13	.067312	37	10	0
14	.066414	36	11	0
15	.063758	35	12	0
16	.059507	34	13	0
17	.053953	33	14	0
18	.047479	32	15	0
19	.040515	31	16	0
20	.033493	30	17	0
21	.026794	29	18	0
22	.020721	28	19	0
23	.015471	27	20	0
24	.011139	26	21	0
25	7.723E-03	25	22	0
26	5.149E-03	24	23	0
27	3.295E-03	23	24	0
28	2.021E-03	22	25	0
29	1.186E-03	21	26	0
30	6.64E-04	20	27	0
31	3.54E-04	19	28	0
32	1.79E-04	18	29	0
33	8.6E-05	17	30	0
34	3.9E-05	16	31	0
35	1.7E-05	15	32	0
36	7E-06	14	33	0
37	2E-06	13	34	0
38	1E-06	12	35	0
39	0	11	36	0
40	0	10	37	0
41	0	9	38	0
42	0	8	39	0
43	0	7	40	0
44	0	6	41	0
45	0	5	42	0
46	0	4	43	0
47	0	3	44	0
48	0	2	45	0
49	0	1	46	0
50	0	0	47	0

```
TO CONTINUE, PRESS 'RETURN'?

         SYSTEM CHARACTERISTICS
         ----------------------
NO. OF MACHINES = 50
MEAN TIME-TO FAILURE PER
MACHINE = 300 TIME UNITS
NO. OF REPAIRMENT = 3
MEAN REPAIR-TIME PER
REPAIRMAN = 24 TIME UNITS
NO. OF MACHINES PER REPAIRMAN = 16.666667

PROBABILITY (SERVICE SYSTEM
IS EMPTY) = 1.963E-03
PROBABILITY (NO MACHINES
ARE WAITING FOR SERVICE) = .0449067487

EXP. NO. OF MACHINES OPERATING
                      = 37.037685
EXP. NO. OF INACTIVE MACHINES
                      = 12.962315
EXP. NO. OF MACHINES IN WAITING LINE
                      = 9.9993
EXP. NO. OF MACHINES IN A NON-EMPTY
'WAITING LINE'        = 10.469449
MEAN DOWN-TIME PER
MACHINE = 104.99291 TIME UNITS
MEAN WAITING TIME PER
MACHINE = 80.99291 TIME UNITS
EXP. NO. OF REPAIRMEN IDLE = .036985

TO CONTINUE, PRESS 'RETURN'?
'COEFFICIENT OF LOSS' FOR MACHINES =
  FRACTION OF TIME A MACHINE IS 'DOWN'
  AS A CONSEQUENCE OF THE SYSTEM
  CHARACTERISTICS = .199986

'COEFFICIENT OF LOSS' FOR REPAIRMEN =
  FRACTION OF TIME A REPAIRMAN IS IDLE
  AS A CONSEQUENCE OF THE SYSTEM
  CHARACTERISTICS = .012328

TYPE 1 FOR CASH FLOW ANALYSIS
     2 TO HALT
?1
THIS ANALYSIS ASSUMES THAT REPAIRMEN
ARE PAID 'A' MONETARY UNITS PER UNIT
TIME, THAT THE FIXED COST OF POSSESSING
EACH MACHINE IS 'B' MONETARY UNITS PER
UNIT TIME, AND THAT A MACHINE, WHEN
OPERATING, IS CAPABLE OF PRODUCING 'C'
UNITS OF REVENUE PER UNIT TIME.

INPUT THE REPAIRMAN-COST PER UNIT TIME,
'A' = ?5.25
```

```
INPUT THE FIXED COST PER UNIT TIME
'B', OF POSSESSING A MACHINE
'B' = ?.05

INPUT THE AMOUNT OF REVENUE A WORKING
MACHINE PRODUCES, PER UNIT OF
(OPERATING) TIME
'C' = ?0.5

THE AVERAGE CASH FLOW GENERATED BY THE
COMBINATION OF 50 MACHINE(S)
MAINTAINED BY 3 REPAIRMEN
IS .268843 MONETARY UNITS,
PER UNIT TIME.
```

Practice Problem

In the above example, suppose Ace invested $65.00 per machine to retrofit them with heavy duty motors, raising their mean times-to-failure to 305 hours. What cash flow will the machines produce? How much time must pass before Ace has recovered their $3,250.00 investment?

Answer: If retrofit, the machines will produce an average revenue of $0.525136 per hour. The investment will be recovered within 18 months.

Program Listing

```
10  PRINT "SWEDISH MACHINE"
15  DEF  FN R(X) =  INT (X * 1E6 + .5) / 1E6
20  REM   -- CHANGE DIMENSION OF Q() TO
25  REM  -- MAXIMUM NUMBER OF MACHINES + 1
30  DIM Q(100)
40  PRINT
50  PRINT "INPUT THE NUMBER OF MACHINES;"
55  PRINT "COUNT MUST EXCEED ONE. ";
60  INPUT N
70  PRINT "INPUT MEAN TIME-TO-FAILURE"
75  PRINT "FOR A MACHINE ";
80  INPUT F1
90 F = 1 / F1
100  PRINT "INPUT NUMBER OF REPAIRMEN ";
110  INPUT M
120  PRINT "INPUT MEAN REPAIR TIME (PER MACHINE)"
125  PRINT "FOR A REPAIRMAN ";
130  INPUT R1
140 R = 1 / R1
150  PRINT
160  REM  -- INITIALIZE VARIABLES
170  FOR I = 1 TO N + 1
180 Q(I) = 0
190  NEXT I
200 Q(1) = 1
210 E1 = 0
220 E2 = 0
```

```
230 E3 = 0
240 P0 = 0
250  REM  -- LOOP TO CALCULATE PROBABILITIES
255  REM  -- FOR EACH MACHINE
260 S = Q(1)
270  FOR J = 0 TO N - 1
280  REM  -- K=MIN(J+1,M)
290 K = M
300  IF J + 1 > M THEN 320
310 K = J + 1
320 Q(J + 2) = (N - J) * F * Q(J + 1) / K / R
330 S = S + Q(J + 2)
340  NEXT J
350  IF Q(1) <  > 1 THEN 380
360 Q(1) = 1 / S
370  GOTO 260
380  PRINT
390  PRINT "THE SYSTEM IF SAID TO BE 'IN STATE J'"
395  PRINT "IF J MACHINES ARE IN A FAILED"
400  PRINT "CONDITION.  THE STATIONARY PROBABILITY"
405  PRINT "DISTRIBUTION OVER THE POSSIBLE STATES,"
410  PRINT "0 THRU ";N;", AND OTHER CHARACTERISTICS"
420  PRINT "OF INTEREST, FOLLOW."
430  PRINT
440  PRINT "STATE PROBA- NO.        NO.       NO."
445  PRINT "    BILITY MACHINES  MACHINES REPAIRMEN"
450  PRINT "           OPERATING WAITING  IDLE"
460  FOR J = 1 TO N + 1
470 O = N - J + 1
480 W = J - M - 1
490  IF W > 0 THEN 520
500 W = 0
510 P0 = P0 + Q(J)
520 I = M - J + 1
530  IF I > 0 THEN 550
540 I = 0
550  IF I < M THEN 570
560 I = M
570  PRINT J - 1; TAB( 5); FN R(Q(J)); TAB( 15);O; TAB( 24);W;
     TAB( 33);I
580 E1 = E1 + W * Q(J)
590 E2 = E2 + I * Q(J)
600 E3 = E3 + O * Q(J)
610  NEXT J
620  PRINT
630  PRINT "TO CONTINUE, PRESS 'RETURN'";
640  INPUT Z$
650  PRINT
660  PRINT  TAB( 8);"SYSTEM CHARACTERISTICS"
670  PRINT  TAB( 8);"----------------------"
680  PRINT "NO. OF MACHINES = ";N
690  PRINT "MEAN TIME-TO FAILURE PER "
695  PRINT "MACHINE = ";F1;" TIME UNITS"
700  PRINT "NO. OF REPAIRMENT = "; FN R(M)
710  PRINT "MEAN REPAIR-TIME PER"
```

```
715  PRINT "REPAIRMAN = ";R1;" TIME UNITS"
720  PRINT "NO. OF MACHINES PER REPAIRMAN = "; FN R(N / M)
730  PRINT
740  PRINT "PROBABILITY (SERVICE SYSTEM"
745  PRINT "IS EMPTY) = "; FN R(Q(1))
750  PRINT "PROBABILITY (NO MACHINES"
755  PRINT "ARE WAITING FOR SERVICE) = ";P0
760  PRINT
770  PRINT "EXP. NO. OF MACHINES OPERATING"
775  PRINT  TAB( 20);"= "; FN R(E3)
780  PRINT "EXP. NO. OF INACTIVE MACHINES"
785  PRINT  TAB( 20);"= "; FN R(N - E3)
790  PRINT "EXP. NO. OF MACHINES IN WAITING LINE"
795  PRINT  TAB( 20);"= "; FN R(E1)
800  PRINT "EXP. NO. OF MACHINES IN A NON-EMPTY"
805  PRINT "'WAITING LINE'"; TAB( 20); FN R(E1 / (1 - P0))
810  PRINT "MEAN DOWN-TIME PER"
815  PRINT "MACHINE = "; FN R((N - E3) * F1 / E3);" TIME UNITS"
820  PRINT "MEAN WAITING TIME PER"
825  PRINT "MACHINE = "; FN R(E1 * F1 / E3);" TIME UNITS"
830  PRINT "EXP. NO. OF REPAIRMEN IDLE = "; FN R(E2)
840  PRINT
850  PRINT "TO CONTINUE, PRESS 'RETURN'";
860  INPUT Z$
870  PRINT "'COEFFICIENT OF LOSS' FOR MACHINES = "
880  PRINT "  FRACTION OF TIME A MACHINE IS 'DOWN'"
885  PRINT "  AS A CONSEQUENCE OF THE SYSTEM"
890  PRINT "  CHARACTERISTICS = "; FN R(E1 / N)
900  PRINT
910  PRINT "'COEFFICIENT OF LOSS' FOR REPAIRMEN = "
915  PRINT "  FRACTION OF TIME A REPAIRMAN IS IDLE"
920  PRINT "  AS A CONSEQUENCE OF THE SYSTEM"
930  PRINT "  CHARACTERISTICS = "; FN R(E2 / M)
940  PRINT
950  PRINT "TYPE 1 FOR CASH FLOW ANALYSIS"
960  PRINT "     2 TO HALT"
970  INPUT Q1
980  IF Q1 = 2 THEN 1250
990  PRINT "THIS ANALYSIS ASSUMES THAT REPAIRMEN "
995  PRINT "ARE PAID 'A' MONETARY UNITS PER UNIT"
1000  PRINT "TIME, THAT THE FIXED COST OF POSSESSING"
1010  PRINT "EACH MACHINE IS 'B' MONETARY UNITS PER"
1015  PRINT "UNIT TIME, AND THAT A MACHINE, WHEN"
1020  PRINT "OPERATING, IS CAPABLE OF PRODUCING 'C'"
1030  PRINT "UNITS OF REVENUE PER UNIT TIME."
1040  PRINT
1050  PRINT "INPUT THE REPAIRMAN-COST PER UNIT TIME,"
1055  PRINT "'A' = ";
1060  INPUT A
1070  PRINT
1080  PRINT "INPUT THE FIXED COST PER UNIT TIME"
1090  PRINT "'B', OF POSSESSING A MACHINE"
1095  PRINT "'B' = ";
1100  INPUT B
1110  PRINT
```

```
1120  PRINT "INPUT THE AMOUNT OF REVENUE A WORKING"
1130  PRINT "MACHINE PRODUCES, PER UNIT OF"
1135  PRINT "(OPERATING) TIME"
1137  PRINT "'C' = ";
1140  INPUT C
1150  PRINT
1160 D = C * E3 - A * M - B * N
1170  PRINT "THE AVERAGE CASH FLOW GENERATED BY THE"
1175  PRINT "COMBINATION OF ";N;" MACHINE(S) "
1180  PRINT "MAINTAINED BY ";M;" REPAIR";
1190  IF M > 1 THEN 1220
1200  PRINT "MAN "
1210  GOTO 1230
1220  PRINT "MEN "
1230  PRINT "IS "; FN R(D);" MONETARY UNITS,"
1240  PRINT "PER UNIT TIME."
1250  END
```

Markov Analysis

This program calculates the future changes, over time, in a given variable based on its current movement. Management scientists adopted this analysis, using it mostly as a simulation technique for analyzing competitors in the marketplace. Markov analysis has many other applications, however, as illustrated by the examples below.

To use the program, first enter how many states of nature are under consideration. The second entry is optional. If you want to see changes occur over time from stage to stage, you must enter the current population proportion vector. If you are only interested in long-run steady-state equilibrium, the program will seed the vector with equal probabilities. The number of elements in this vector equals the states of nature.

The program then asks you to enter each cell of the transition probabilities matrix (N*N, where N = states of nature). For each cell, enter a transition probability, ranging 0<=p<=1. The sum of the probabilities entered for each row should always add up to 1. Once you have entered the entire matrix, you have the option of looking at each future period or letting the computer calculate the transition matrix at equilibrium.

The program displays the equilibrium vector, the period at which equilibrium was reached, and the first passage times for each state of nature. First passage times will not print for recurrent or null-recurrent states.

Program Notes

This program allows for a maximum of 12 states of nature. You can change this by modifying line 20 as follows:

```
20 DIM V1(I), T(I,I), V2(I)
```

Replace I with your maximum (for example, 15, 20, or 25).

If you have large matrices to enter, or if you want to repeatedly run this program with mostly the same data, you can modify the program to accept data through DATA statements, as shown in the "Option" section.

Example

Caffrey's Hardware wants to analyze its accounts receivable in order to estimate its cash flow from credit customers. The company has three aging categories: current, 45-89 days, and 90-plus days past due. Customers in this last category are eventually written off as uncollectable accounts.

The latest aging analysis shows that, for each dollar of accounts receivable outstanding, $0.60 is current, $0.33 is 45-89 days old, and $0.07 is 90-plus days old. Further analysis shows that accounts in the "current" category have a 38% chance of being paid in the next month, 45% of all current accounts will remain current, and 17% will be 45-89 days old. Accounts in the 45-89 days category stand a 65% chance of paying all back payments, a 25% chance of paying only the late installment, and a 5% chance of becoming 90-plus days overdue. Of the accounts in the 90-plus category, there is a 25% chance they will be paid and a 75% chance they will become bad debts.

The paid and bad debt categories are "absorbing" states, in that the probability of a paid item remaining paid is assumed to be 100%. The same is true for bad debts. These are called absorbing states because all accounts outstanding now will eventually be paid up or written off. How much of accounts receivable will be collected? How much will be written off?

On the printout below, the paid category and bad debt category have absorbed all outstanding debts. Caffrey can expect about 91% of his accounts to be paid, and 9% to be written off.

```
MARKOV ANALYSIS

HOW MANY STATES OF NATURE ?5

IS THE POPULATION PROPORTION
VECTOR KNOWN (Y/N) ?Y

          ENTER VECTOR ELEMENT 1 ?0
          ENTER VECTOR ELEMENT 2 ?.6
          ENTER VECTOR ELEMENT 3 ?.33
          ENTER VECTOR ELEMENT 4 ?.07
          ENTER VECTOR ELEMENT 5 ?0

ENTER ELEMENT IN ROW 1 COLUMN 1 ?1
ENTER ELEMENT IN ROW 1 COLUMN 2 ?0
ENTER ELEMENT IN ROW 1 COLUMN 3 ?0
ENTER ELEMENT IN ROW 1 COLUMN 4 ?0
ENTER ELEMENT IN ROW 1 COLUMN 5 ?0

ENTER ELEMENT IN ROW 2 COLUMN 1 ?.38
ENTER ELEMENT IN ROW 2 COLUMN 2 ?.45
ENTER ELEMENT IN ROW 2 COLUMN 3 ?.17
ENTER ELEMENT IN ROW 2 COLUMN 4 ?0
ENTER ELEMENT IN ROW 2 COLUMN 5 ?0

ENTER ELEMENT IN ROW 3 COLUMN 1 ?.65
ENTER ELEMENT IN ROW 3 COLUMN 2 ?.25
ENTER ELEMENT IN ROW 3 COLUMN 3 ?0
ENTER ELEMENT IN ROW 3 COLUMN 4 ?.05
ENTER ELEMENT IN ROW 3 COLUMN 5 ?0
-PROBABILITIES DO NOT ADD UP TO 1.0-
    TRY ENTERING THE ROW AGAIN.
ENTER ELEMENT IN ROW 3 COLUMN 1 ?.65
ENTER ELEMENT IN ROW 3 COLUMN 2 ?.25
ENTER ELEMENT IN ROW 3 COLUMN 3 ?0
ENTER ELEMENT IN ROW 3 COLUMN 4 ?.1
ENTER ELEMENT IN ROW 3 COLUMN 5 ?0

ENTER ELEMENT IN ROW 4 COLUMN 1 ?.25
ENTER ELEMENT IN ROW 4 COLUMN 2 ?0
ENTER ELEMENT IN ROW 4 COLUMN 3 ?0
ENTER ELEMENT IN ROW 4 COLUMN 4 ?0
ENTER ELEMENT IN ROW 4 COLUMN 5 ?.75

ENTER ELEMENT IN ROW 5 COLUMN 1 ?0
ENTER ELEMENT IN ROW 5 COLUMN 2 ?0
ENTER ELEMENT IN ROW 5 COLUMN 3 ?0
ENTER ELEMENT IN ROW 5 COLUMN 4 ?0
ENTER ELEMENT IN ROW 5 COLUMN 5 ?1

DO YOU WANT TO OBSERVE EACH
PERIOD UNDER ANALYSIS (Y/N) ?Y

POPULATION PROPORTION
VECTOR AT PERIOD 2 IS:
.46
```

```
.3525
.102
.033
.0525

POPULATION PROPORTION
VECTOR AT PERIOD 3 IS:
.6686
.1841
.0599
.0102
.0772

POPULATION PROPORTION
VECTOR AT PERIOD 4 IS:
.7801
.0978
.0313
6E-03
.0849

POPULATION PROPORTION
VECTOR AT PERIOD 5 IS:
.8391
.0518
.0166
3.1E-03
.0894

POPULATION PROPORTION
VECTOR AT PERIOD 6 IS:
.8704
.0275
8.8E-03
1.7E-03
.0917

POPULATION PROPORTION
VECTOR AT PERIOD 7 IS:
.887
.0146
4.7E-03
9E-04
.093

POPULATION PROPORTION
VECTOR AT PERIOD 8 IS:
.8958
7.8E-03
2.5E-03
5E-04
.0937

POPULATION PROPORTION
VECTOR AT PERIOD 9 IS:
```

```
.9005
4.1E-03
1.3E-03
2E-04
.0941

POPULATION PROPORTION
VECTOR AT PERIOD 10 IS:
.903
2.1E-03
7E-04
1E-04
.0943

POPULATION PROPORTION
VECTOR AT PERIOD 11 IS:
.9043
1.1E-03
4E-04
1E-04
.0944

POPULATION PROPORTION
VECTOR AT PERIOD 12 IS:
.905
6E-04
2E-04
0
.0945

POPULATION PROPORTION
VECTOR AT PERIOD 13 IS:
.9053
4E-04
1E-04
0
.0945

POPULATION PROPORTION
VECTOR AT PERIOD 14 IS:
.9056
2E-04
1E-04
0
.0945

POPULATION PROPORTION
VECTOR AT PERIOD 15 IS:
.9058
1E-04
0
0
.0945

POPULATION PROPORTION
```

```
VECTOR AT PERIOD 16 IS:
.9058
0
0
0
.0945

POPULATION PROPORTION
VECTOR AT PERIOD 17 IS:
.9058
0
0
0
.0945

EQUILIBRIUM REACHED AT PERIOD 17
VECTOR AT EQUILIBRIUM:
.9058
0
0
0
.0945

DO YOU WANT TO RE-RUN THIS PROGRAM
WITH DIFFERENT DATA (Y/N) ? N
```

Practice Problems

1. A survey by Hanley, Ohio, city planners shows recent commuting trends. Citizens were polled to find out if they carpool, take the bus, or drive alone to and from work. Presently, 43% of commuters drive their cars alone, 30% carpool and 27% take the bus to work. The city wants to know how these patterns will change over the coming months in order to increase or decrease their bus fleet. The survey shows that 65% of those who drive alone will continue to do so. Twenty percent of this group said they would carpool, and 15% would take the bus if gas prices continue to rise. Twenty-five percent of carpoolers say that they find driving alone is preferable, and that they will switch back to it. Fifty-five percent of carpoolers say that they will continue to carpool, and the remaining 20% will switch to the bus.

Twelve percent of bus riders will switch back to driving alone. Thirteen percent of bus riders say they will switch to carpooling, and 75% say they will continue to ride the bus. What will the commuting mix be six months from now? What will it look like at equilibrium?

Answer: In the sixth month, 33.5% will be driving alone, 26.66% will be carpooling and 39.86% will be taking the bus. At equilibrium, 32.86% will be driving alone, 26.4% will be carpooling and 40.83% will be riding the bus.

2. Rita's Rent-A-Car competes with two other rental agencies at Manteca Airport. In the past month, Rita's kept 85% of its customers from the previous month, lost 5% of its business to Competitor A, and lost 10% to Competitor B. Competitor A retained 90% of its customers while losing 10% to Competitor B. Competitor B retained 75% of its customers, while losing 15% to Competitor A, and 10% to Rita's. What are the equilibrium market shares, assuming no known proportion vector? How long, in months, does it take for a customer to return to Rita's to rent a car after having taken his/her business elsewhere?

Answer: Equilibrium shares: Rita's, 19.1%; Competitor A, 52.45%; Competitor B, 28.63%. On the average, it takes about 5.2356 months for a patron of either competitor to switch to Rita's.

Program Listing

```
10  REM  MARKOV ANALYSIS
20  DIM V1(12),T(12,12),V2(12)
30  REM  V1()=POPULATION PROPORTION VECTOR
40  REM  T() =TRANSITION PROBABILITIES MATRIX
50  REM  V2()=SCRATCH FOR VECTOR ARRAY
60  DEF  FN R(Z) =  INT ((Z * 10000 + 0.5)) / 10000
70  PRINT "MARKOV ANALYSIS"
80  PRINT
90 N = 1
100  PRINT "HOW MANY STATES OF NATURE ";
110  INPUT S
120  PRINT
130  PRINT "IS THE POPULATION PROPORTION"
135  PRINT "VECTOR KNOWN (Y/N) ";
140  INPUT A$
150  IF A$ = "Y" THEN 220
160  IF A$ <  > "N" THEN 130
170  REM  IF VECTOR UNKNOWN, ASSIGN EQUAL
175  REM  PROBABILITIES TO EACH STATE
180  FOR I = 1 TO S
190 V1(I) =  FN R(1 / S)
200  NEXT I
210  GOTO 280
220  REM  LOOP TO ENTER POPULATION PROPORTIONS
230  PRINT
240  FOR I = 1 TO S
250  PRINT "          ENTER VECTOR ELEMENT ";I;" ";
260  INPUT V1(I)
270  NEXT I
280  REM  ENTER TRANSITION MATRIX (I BY J ARRAY)
290  PRINT
300  FOR I = 1 TO S
310 K = 0
320  FOR J = 1 TO S
330  PRINT "ENTER ELEMENT IN ROW ";I;" COLUMN ";J;" ";
340  INPUT T(I,J)
350 K = K + T(I,J)
360  NEXT J
370  IF K = 1 THEN 410
380  PRINT "-PROBABILITIES DO NOT ADD UP TO 1.0-"
390  PRINT "    TRY ENTERING THE ROW AGAIN."
400  GOTO 310
410  PRINT
420  NEXT I
430  PRINT "DO YOU WANT TO OBSERVE EACH"
440  PRINT "PERIOD UNDER ANALYSIS (Y/N) ";
450  INPUT A$
460  IF A$ = "Y" THEN 480
470  IF A$ <  > "N" THEN 430
480  REM  LOOP TO MULTIPLY VECTOR (V1) BY
485  REM  TRANSITION MATRIX (T)
490 N = N + 1
500  FOR I = 1 TO S
```

```
510 V2(I) = 0
520  FOR J = 1 TO S
530  REM  ADD MULTIPLIED COLUMNS TO V2 ARRAY
540 V2(I) = V2(I) +  FN R(V1(J) * T(J,I))
550  NEXT J
560  NEXT I
570  REM  SKIP PRINTING VECTOR IF NOT REQUESTED
580  IF A$ <  > "Y" THEN 620
590  PRINT
600  PRINT "POPULATION PROPORTION"
610  PRINT "VECTOR AT PERIOD ";N;" IS:"
620 N1 = 0
630  FOR I = 1 TO S
640  IF A$ <  > "Y" THEN 660
650  PRINT V2(I)
660  IF V2(I) <  > V1(I) THEN 680
670 N1 = N1 + 1
680 V1(I) = V2(I)
690  NEXT I
700  IF N1 <  > S THEN 480
710  REM  PRINT EQUILIBRIUM VECTOR VALUES
720  PRINT
730  PRINT "EQUILIBRIUM REACHED AT PERIOD ";N
740  PRINT "VECTOR AT EQUILIBRIUM:"
750  FOR I = 1 TO S
760  PRINT  FN R(V1(I))
770  NEXT I
780  PRINT
790  REM  PRINT TRANSITIONS NEEDED FOR
800  REM  EACH STATE TO BE REOCCUPIED
810  FOR I = 1 TO S
820  IF T(I,I) = 1 OR V1(I) <  = 0 THEN 860
840  PRINT "FIRST PASSAGE--STATE ";I;": ";
850  PRINT  FN R(1 / V1(I))
860  NEXT I
870  PRINT "DO YOU WANT TO RE-RUN THIS PROGRAM"
880  PRINT "WITH DIFFERENT DATA (Y/N) ";
890  INPUT A$
900  IF A$ = "Y" THEN 80
910  IF A$ <  > "N" THEN 870
920  END
```

Option

If you plan on entering large matrices, or if you want to run this program repeatedly with the same data, you should use this option. The program will read input from DATA statements, rather than asking you to enter the population proportion vector and the transition probabilities matrix. Replace lines 170 through 350 with the lines shown below. Also delete lines 870 through 910, and leave line 920 where it is. If you plan to re-run the program without entering the population proportion vector, you must delete lines 242 through 249 if they contain DATA statements for a population proportion vector from a previous run.

```
170  REM    IF VECTOR UNKNOWN, ASSIGN EQUAL
173  REM    PROBABILITIES TO EACH STATE
175  REM    IF UNKNOWN, YOU MUST DELETE
177  REM    LINES 242-249 OR THE DATA WILL
178  REM    BE OUT OF SEQUENCE.
180  FOR I = 1 TO S
190 V1(I) =  FN R(1 / S)
200  NEXT I
210  GOTO 280
220  REM    LOOP TO READ POPULATION PROPORTIONS
230  PRINT
240  FOR I = 1 TO S
241  REM    PUT PROPORTION VECTOR ELEMENTS HERE
242  DATA    0,.6,.33,.07,0
250  PRINT "                    VECTOR ELEMENT ";I;": ";
260  READ V1(I)
265  PRINT V1(I)
270  NEXT I
280  REM    READ TRANSITION MATRIX (I BY J ARRAY)
290  PRINT
300  FOR I = 1 TO S
310 K = 0
320  FOR J = 1 TO S
321  REM    PUT TRANSITION PROBABILITIES MATRIX HERE
322  DATA     1,0,0,0,0,.38,.45,.17,0,0,.65,.25,0,.05,0
323  DATA    .65,.25,0,.1,0,.25,0,0,0,.75,0,0,0,0,1
330  PRINT "        ELEMENT IN ROW ";I;" COLUMN";" ";J;" ";
340  READ T(I,J)
345  PRINT T(I,J)
350 K = K + T(I,J)
```

References

Cabot, A., Victor, and Harnett, Donald L. *An Introduction to Management Science.* Reading, Mass.: Addison-Wesley, 1977.

Levin, Richard I., and Kirkpatrick, Charles A. *Quantative Approaches to Management* (3rd ed.). New York: McGraw-Hill, 1975.

Nonlinear Break-even Analysis

This program computes the break-even point of a product using a nonlinear method which more closely reflects actual production situations than a linear method. It incorporates a "learning curve" for both costs and prices. This curve means that each time production or sales double, cumulative average costs or revenue per unit will increase or decrease by the amount of the curves. Zero curve values means no change occurs. When you enter different curve values for costs and prices, the program indicates the point of maximum gross profit.

To use the program, enter the unit selling price, the selling price learning curve, the variable costs, the variable costs learning curve, and the fixed costs. Variable costs are those which can be directly ascribed to the production of each unit, such as raw material. Fixed costs, like rent and wages, generally do not vary with each unit produced.

Example

Acme Widget Supply is considering producing and marketing a new widget. New machines, employee training, and all other overhead costs associated with production of this widget total $10,000. Each unit produced requires $5.00 of raw materials, labor, machine depreciation, and so forth, but they will need proportionally more machines and personnel to produce more widgets, and will therefore use a 5% cost increase learning curve. The marketing department expects the selling price of $25.00 to decrease on a 5% curve. What is the break-even point on the new widget? What is the maximum gross profit margin that Acme may realize? What are total costs and total revenue at maximum gross profit?

Answer: Break-even will occur at 1,663 units. The maximum gross profit margin is 17.182%. Total costs and revenue at maximum gross profit are $74,134.00 and $89,514.00, respectively.

```
BREAKEVEN ANALYSIS

ENTER THE UNIT PRICE ?25
ENTER THE UNIT PRICE EROSION RATE
(NEGATIVE VALUE MEANS REVENUE
DECREASES AS SALES INCREASE)?-5

ENTER THE AMOUNT OF VARIABLE COSTS PER
UNIT ?5
ENTER VARIABLE COSTS LEARNING RATE
(NEGATIVE VALUE MEANS COSTS DECREASE AS
PRODUCTION DOUBLES) ?5

ENTER THE TOTAL AMOUNT OF FIXED
COSTS ?10000

BREAKEVEN POINT = 1663 UNITS
TOTAL REVENUE AT BREAKEVEN = $24015

MAXIMUM GROSS PROFIT MARGIN AT 6886
UNITS = 17.182%

TOTAL REVENUE = $89514
TOTAL COSTS = $74134
```

```
TOTAL PROFIT = $15380

WOULD YOU LIKE TO RE-RUN THIS PROGRAM
WITH NEW DATA (Y/N) ?N
```

Practice Problems

1. The selling price is $30.00, and revenue will decrease by 2.5% each time production doubles. Variable costs are $1.20 per unit, but cumulative average costs will increase by 8% when production quantities double. Fixed costs are $180,000.00. What is the break-even point? What is the maximum gross profit margin?

Answer: Break-even at 9,945 units, maximum gross profit margin of 71.185% occurs at 246,752 units.

2. With a unit price of $19.95, variable costs of $4.75, and fixed costs of $6800, how many units must be sold to break even? (No price or cost changes will occur. Use curve values of zero for both revenue and costs.)

Answer: Break-even at 447 units.

Program Listing

```
10  PRINT "BREAKEVEN ANALYSIS"
20  REM  -- THESE FUNCTIONS COMPUTE THE CURVATURE
30  DEF  FN A(X) =  -  LOG (1 + (X / 100)) /  LOG (2)
40  DEF  FN B(X) =  LOG (1 + (X / 100)) /  LOG (2) + 1
50  DEF  FN C(X) =  INT (((T1 - T2) / T1) * 1E5 + 0.5) / 1000
60  PRINT
70  PRINT "ENTER THE UNIT PRICE ";
80  INPUT U
90  PRINT "ENTER THE UNIT PRICE EROSION RATE "
100  PRINT "(NEGATIVE VALUE MEANS REVENUE"
105  PRINT "DECREASES AS SALES INCREASE)";
110  INPUT L1
120 A1 =  FN A(L1)
130 B1 =  FN B(L1)
140  PRINT
150  PRINT "ENTER THE AMOUNT OF VARIABLE COSTS PER"
155  PRINT "UNIT ";
160  INPUT V
170  PRINT "ENTER VARIABLE COSTS LEARNING RATE"
180  PRINT "(NEGATIVE VALUE MEANS COSTS DECREASE AS"
185  PRINT "PRODUCTION DOUBLES) ";
190  INPUT L2
200 A2 =  FN A(L2)
210 B2 =  FN B(L2)
220  PRINT
230  PRINT "ENTER THE TOTAL AMOUNT OF FIXED"
235  PRINT "COSTS ";
240  INPUT F
250  PRINT
260  REM  INITIALIZE LAST GUESS, LOW GUESS, HIGH GUESS
270 C = 0
280 L = 1
290 H = 1E4
```

```
300  REM  CALCULATE POINT USING BINARY SEARCH
310 B =  INT ((L + H) / 2)
320  REM  IF NEW POINT = LAST QUESS, EXIT
330  IF B = C THEN 480
340  REM  SET LAST GUESS TO NEW POINT
350 C = B
360  REM  CALCULATE TOTAL REVENUE AND
365  REM  TOTAL COSTS AT QUANTITY B
370 T1 =  INT ((U * B ^ B1) + 0.5)
380 T2 =  INT ((V * B ^ B2 + F) + 0.5)
390  REM  BREAKEVEN POINT FOUND IF TOTAL
395  REM  REVENUE = TOTAL COSTS
400  IF T1 = T2 THEN 480
410  REM  ADJUST GUESS HIGH OR LOW POINTS, TRY AGAIN
420  IF T1 > T2 THEN 450
430 L = B
440  GOTO 310
450 H = B
460  GOTO 310
470  REM  BREAKEVEN POINT FOUND, OUTPUT RESULT
480  PRINT "BREAKEVEN POINT = ";B;" UNITS"
490  PRINT "TOTAL REVENUE AT BREAKEVEN = $";T1
500  REM  USE THIS SECTION IF FIGURES ARE LINEAR
510  IF L1 <  > L2 THEN 570
520  PRINT "COSTS AND REVENUE ARE LINEAR."
530  PRINT "NO MAXIMUM GROSS PROFIT MARGEN POSSIBLE"
540  GOTO 680
550  REM  OUTPUT MAXIMUM GROSS PROFIT
555  REM  MARGIN DATA FOR NON-LINEAR VALUES
560  REM  (SKIP THIS SECTION IF FIGURES ARE LINEAR)
570 B =  INT ( EXP ( LOG ((F * (A1 - 1)) / (V * (A2 - A1))) /
    (1 - A2)) + 0.5)
580 T1 =  INT (U * B ^ B1)
590 T2 =  INT (V * B ^ B2 + F)
600  PRINT
610  PRINT "MAXIMUM GROSS PROFIT MARGIN AT ";B
620  PRINT "UNITS = "; FN C((T1 - T2) / T1);"%"
630  PRINT
640  PRINT "TOTAL REVENUE = $";T1
650  PRINT "TOTAL COSTS = $";T2
660  PRINT
670  PRINT "TOTAL PROFIT = $";T1 - T2
680  PRINT
690  PRINT "WOULD YOU LIKE TO RE-RUN THIS PROGRAM"
695  PRINT "WITH NEW DATA (Y/N) ";
700  INPUT Z$
710  IF Z$ = "Y" THEN 60
720  IF Z$ <  > "N" THEN 680
730  END
```

References

Solomon and Pringle. *An Introduction to Financial Management.* Santa Monica, Calif.: Goodyear Publishing Company, 1977.

Texas Instruments. *Programmable 58/59 Calculator Business Decisions Library*, Part number 1014984-9.

Payoff Matrix Analysis

This program evaluates a set of alternatives, each of which has some measurable benefit, or "payoff," subject to varying states of nature. Under different conditions, payoff amounts could be large or they could become losses. To analyze payoffs in conditions of uncertainty, this program employs three criteria: "maximax" (find the alternative with the highest possible payoff), "maximin" (the best alternative under the worst case), and "minimax regret" (the alternative which minimizes opportunity cost).

To use this program, you should carefully consider your alternatives. They must relate to one another (for example, you have $20,000 and you want to know which of four types of investments is optimal to make, given varying states of the economy). You must be able to "guesstimate" what the payoffs will be (positive, negative or zero) for each alternative under each state of nature, as well as the probability of each state of nature's occurrence.

The computer will ask you how many states of nature to consider and how many alternatives exist. Then you will enter the payoff matrix row by row, starting with action 1 under state 1, action 2 under state 2, and so on. After you enter the matrix, you will input the probabilities of each state of nature. These probabilities are mutually exclusive, and they must add up to 1.0. The computer will ask you to reenter them if they do not add up to 1.0.

The program shows you what choices are best under the maximax and maximin rules. The computer will optionally display the regret matrix. The optimal maximin regret choice displays, followed by the expected payoff values of each alternative.

Program Notes

The program allows for ten states of nature and ten alternatives. You can change this by modifying line 20 of this program as follows:

```
20 DIM S(N,A), M(A), R(N), X(A)
```

Replace the expression N with the maximum states of nature, and A with the maximum number of alternatives.

Example

Fred wants to invest capital in the market. He sees his choices as stocks, Baa bonds or options. These three choices will pay off relative to how the economy behaves:

	State of Economy		
Investment	**Recession**	**Stable**	**Inflation**
Stocks	−20	65	200
Baa Bonds	0	80	80
Options	−300	0	300
Probability	0.3	0.2	0.5

How does Fred run the program?

Answer:

```
PAYOFF MATRIX ANALYSIS

HOW MANY STATES OF NATURE ?3
HOW MANY POSSIBLE ACTIONS ?3
```

```
PAYOFF OF ACTION 1 IN STATE 1 ?-20
PAYOFF OF ACTION 1 IN STATE 2 ?65
PAYOFF OF ACTION 1 IN STATE 3 ?200

PAYOFF OF ACTION 2 IN STATE 1 ?0
PAYOFF OF ACTION 2 IN STATE 2 ?80
PAYOFF OF ACTION 2 IN STATE 3 ?80

PAYOFF OF ACTION 3 IN STATE 1 ?-300
PAYOFF OF ACTION 3 IN STATE 2 ?0
PAYOFF OF ACTION 3 IN STATE 3 ?300

ENTER PROBABILITY FOR STATE 1 ?.3
ENTER PROBABILITY FOR STATE 2 ?.2
ENTER PROBABILITY FOR STATE 3 ?.5

MAXIMAX PAYOFF OF 300 FROM ACTION 3

MAXIMIN PAYOFF OF 0 FROM ACTION 2

DO YOU WANT TO SEE THE REGRET
TABLE (Y/N) ?Y

STATE      1    2    3

ACTION 1   20   15   100   MAX REGRET=100
ACTION 2   0    0    220   MAX REGRET=220
ACTION 3   300  80   0     MAX REGRET=300

MINIMAX REGRET PAYOFF OF 100
FROM ACTION 1

EXPECTED VALUES ARE:
FOR ACTION 1: 107
FOR ACTION 2: 56
FOR ACTION 3: 60

DO YOU WANT TO RUN THIS PROGRAM
AGAIN WITH DIFFERENT DATA (Y/N) ?N
```

Practice Problems

1. A business is considering a service agreement for its computer system. The service agreement costs $100 per month, and covers all repairs. Because the system is five years old, it may be necessary to repair it more often than in the past. Downtime for this system can be for minor or major repairs; the minor repairs averaging $140, and major repairs averaging $900. The probability of downtime requiring minor repair is 0.07; for major repairs, 0.08. What are the payoffs?

Answer: maximax payoff (cost, in this problem): $0. Maximin payoff: $100. Minimax regret: $100. Expected value (cost) of service agreement: $100. Expected cost of no service agreement: $81.80.

2. A market researcher is interested in gathering responses to an opinion poll in one day. The researcher is paid for each completed survey. The number of responses depends on the weather,

as shown below:

	Prevailing Weather		
Location	**Sunny**	**Cloudy**	**Rainy**
Beach	150	30	0
Door-to-Door	40	70	90
Flea Market	80	50	5
Probability of weather:	0.5	0.3	0.2

What are the optimal alternatives under each criterion?

Answer: Under Maximax, option one with a payoff of 150; under maximin, option two with a payoff of 40; under minimax regret, option three with a maximum payoff of 85. Expected values: alternative 1,84; alternative 2, 59; alternative 3, 56.

Program Listing

```
10  REM  ANALYSIS OF A PAYOFF MATRIX
20  DIM S(10,10),M(10),R(10),X(10)
30  PRINT "PAYOFF MATRIX ANALYSIS"
40  PRINT
50  PRINT "HOW MANY STATES OF NATURE ";
60  INPUT N
70  PRINT "HOW MANY POSSIBLE ACTIONS ";
80  INPUT A
90  PRINT
100  FOR Q = 1 TO A
110 M(Q) =  - 9E9
120  PRINT
130  FOR P = 1 TO N
140  PRINT "PAYOFF OF ACTION ";Q;" IN STATE ";P;" ";
150  INPUT S(Q,P)
160  NEXT P
170  NEXT Q
180  REM  ENTER PROBABILITIES FOR EACH
185  REM  STATE OF NATURE
190 A1 = 0
200  PRINT
210  FOR Q = 1 TO N
220  PRINT "ENTER PROBABILITY FOR STATE ";Q;" ";
230  INPUT P1(Q)
240 A1 = A1 + P1(Q)
250 R(Q) = 0
260  NEXT Q
270  IF A1 = 1 THEN 330
280  PRINT
290  PRINT "-PROBABILITIES DO NOT ADD TO 1.0-"
300  PRINT "  CHECK YOUR ENTRIES AND RE-TRY."
310  PRINT
320  GOTO 190
330  REM  CALCULATE MAXIMAX & MAXIMIN VALUES
340 A1 =  - 9E9
350  FOR Q = 1 TO A
360 A3 = 0
370  REM  REPLACE A3 WITH THE HIGHEST PAYOFF
380  FOR P = 1 TO N
```

```
390  IF A1 =  - 9E9 THEN 410
400  IF S(Q,P) <  = A1 THEN 430
410 A1 = S(Q,P)
420 A2 = Q
430  REM   PUT MINIMUM PAYOFF OF EACH ACTION IN M()
440  IF M(Q) =  - 9E9 THEN 460
450  IF S(Q,P) >  = M(Q) THEN 470
460 M(Q) = S(Q,P)
470  REM  SAVE HIGHEST PAYOFF FOR REGRET TABLE
480  IF S(Q,P) <  = R(P) THEN 500
490 R(P) = S(Q,P)
500  NEXT P
510  NEXT Q
520  PRINT
530  PRINT "MAXIMAX PAYOFF OF ";A1;" FROM ACTION ";A2
540  PRINT
550 A1 =  - 9E9
560  FOR Q = 1 TO A
570  IF M(Q) < A1 THEN 600
580 A1 = M(Q)
590 A2 = Q
600  NEXT Q
610  PRINT "MAXIMIN PAYOFF OF ";A1;" FROM ACTION ";A2
620  PRINT
630  PRINT "DO YOU WANT TO SEE THE REGRET"
635  PRINT "TABLE (Y/N) ";
640  INPUT A$
650  IF A$ = "N" THEN 870
660  IF A$ <  > "Y" THEN 630
670  PRINT
680  PRINT "STATE "; TAB( 10);
690 A1 = 0
700  REM  PRINT HEADINGS FOR TABLE
710  FOR P = 1 TO N
720  PRINT P;"   ";
730  NEXT P
740  PRINT
750  PRINT
760  FOR Q = 1 TO A
770  PRINT "ACTION ";Q; TAB( 10)
780  REM  PRINT REGRET VALUES
790 A1 = 0
800  FOR P = 1 TO N
810  PRINT R(P) - S(Q,P);"  ";
820  IF R(P) - S(Q,P) <  = A1 THEN 840
830 A1 = R(P) - S(Q,P)
840  NEXT P
850  PRINT "MAX REGRET=";A1
860  NEXT Q
870  FOR Q = 1 TO A
880 A1 = 0
890  FOR P = 1 TO N
900  IF R(P) - S(Q,P) <  = A1 THEN 930
910 A1 = R(P) - S(Q,P)
920 X(Q) = R(P) - S(Q,P)
```

```
930  NEXT P
940  NEXT Q
950 A1 = 0
960  FOR P = 1 TO A
970  IF P = 1 THEN 990
980  IF X(P) > A1 THEN 1010
990 A1 = X(P)
1000 A2 = P
1010  NEXT P
1020  PRINT
1030  PRINT "MINIMAX REGRET PAYOFF OF ";A1
1035  PRINT "FROM ACTION ";A2
1040  PRINT
1050  PRINT "EXPECTED VALUES ARE:"
1060  FOR P = 1 TO A
1070 A1 = 0
1080  FOR Q = 1 TO N
1090 A1 = A1 + (S(P,Q) * P1(Q))
1100  NEXT Q
1110  PRINT "FOR ACTION ";P;": ";A1
1120  NEXT P
1130  PRINT
1140  PRINT "DO YOU WANT TO RUN THIS PROGRAM"
1150  PRINT "AGAIN WITH DIFFERENT DATA (Y/N) ";
1160  INPUT A$
1170  IF A$ = "Y" THEN 40
1180  IF A$ <  > "N" THEN 1130
1190  END
```

Reference

Cabot, A. Victor, and Harnett, Donald L. *An Introduction to Management Science.* Reading, Mass.: Addison-Wesley, 1977.

Bayesian Decision Analysis

This program revises probabilities (given multiple states of nature) according to Bayes's Theorem for conditional events, and further evaluates possible actions by use of a payoff matrix. This technique applies to sampling for quality based on subjective probabilities you enter.

To use this program, first enter how many possible states of nature there are; for example, an outgoing lot of products can have three possible outcomes: 99% good, 90% good, or 85% good. Then enter the number of conditional actions (for example, send out the lot, send out the lot and retool machines to correct defects, or rework the lot and retool the machines). The next set of entries is the payoff matrix. You enter payoffs (or costs as negative numbers) for each action, within each state of nature. Next, enter two probabilities for each state of nature; first, the "prior" probability that each state of nature occurs, and then the "conditional" probability based on the occurrence of that state.

To illustrate, consider the three possibilities above: 99%, 90, and 85% good. These are conditional probabilities; in other words, "99% good" is a possible outcome of a production run. Therefore, if "99% good" is the present state of nature, then the probability of 99% is conditional based on being in that state of nature. The "prior" probability is the likelihood of that state of nature's occurrence in the first place. Prior probabilities are often "guesstimates" made by production personnel, based on experience.

The last two entries are the size of the sample in question and the actual number of "successes" in the sample taken. In the example above, you may have looked at 50 pieces out of an outgoing lot of 1,000, and you find that five of them are defective. Enter 50 as the sample size, and five as the actual number of successes. The program then prints the expected values of each action, based on revised probabilities. You choose the optimal action from these values, which is usually that action which minimizes costs or maximizes payoff.

After the expected values, the prior probabilities, likelihoods, joint and posterior probabilities print for each action. A final figure, the marginal probability, prints. This is the "unconditional" or expected success rate. You can go back and re-enter a new sample size (or enter zero to end the program).

Example

The quality control department at Fergis Bolt International estimates that bolts produced fall into three categories; 99% acceptable, 90% acceptable, and 80% acceptable. These three levels of quality occur 70, 20, and 10% of the time, respectively. Roland Fergis II wants to impress his father with a comprehensive study which documents how much the company may lose by not making the right quality control decision. He puts together a payoff matrix which looks like this:

	Payoffs		
Actions	**If 99% good**	**If 90% good**	**If 80% good**
Send lot out	−1200	−1800	−2400
Retool machines without rework	−1400	−1600	−2200
Retool machines and rework	−2000	−2000	−2000

The cost of producing the lot itself is $1,200. If the lot is sent out and the quality is less than 99%, Fergis will incur costs of returned merchandise. If they decide to retool the machines only, they will incur downtime, but the rate of returned merchandise will be lower for future lots. If the machines are retooled and the bolts are reworked, the lot will be 99% good no matter what. Therefore, the cost

remains constant. How would Roland Jr. run this program? What will be the optimal strategy-based payoffs if 46 of 50 bolts sampled are acceptable?

Answer: The optimal strategy is to retool the machines, at an expected cost of $1,616.75. This sample has a 94.8% probability of being 90% free of defects.

```
 BAYESIAN DECISION ANALYSIS

HOW MANY STATES OF NATURE ?3
HOW MANY CONDITIONAL ACTIONS ?3

ENTER PAYOFFS FOR:
ACTION 1 UNDER STATE 1 ?-1200
ACTION 1 UNDER STATE 2 ?-1800
ACTION 1 UNDER STATE 3 ?-2400
ACTION 2 UNDER STATE 1 ?-1400
ACTION 2 UNDER STATE 2 ?-1600
ACTION 2 UNDER STATE 3 ?-2200
ACTION 3 UNDER STATE 1 ?-2000
ACTION 3 UNDER STATE 2 ?-2000
ACTION 3 UNDER STATE 3 ?-2000

ENTER PRIOR AND CONDITIONAL PROB.:
FOR STATE 1 ?.7,.99
FOR STATE 2 ?.2,.9
FOR STATE 3 ?.1,.8

ENTER SAMPLE SIZE (0 TO END) ?50

ENTER ACTUAL NUMBER OF SUCCESSES ?46

GIVEN 46 SUCCESSES IN A SAMPLE
OF 50,THE EXPECTED VALUES ARE:
ACTION 1: -1809.42408
ACTION 2: -1616.75393
ACTION 3: -2000

PROBABILITY REVISIONS:
STATE PRIOR LIKELIHOOD JOINT POSTERIOR
1       .7    1E-03     7E-04    .018
2       .2    .181      .0362    .948
3       .1    .013      1.3E-03  .034
ENTER SAMPLE SIZE (0 TO END) ?0
```

Practice Problems

1. In the example above, is the minimum number of acceptable bolts allowable in order to send the lot out without retooling machines? At this point, what is the probability that this lot is actually 99% free of defects? (Hint: Find the answer by trial-and-error. Enter a successively smaller number of successes until you get the answer.)

Answer: The minimum is 48 out of 50, with an expected cost of $1,337.59. At this rate, it is 77.2% likely that the bolts are 99% free of defects.

2. In the example above, does action 3 — rework the lot and retool the machines — become optimal?

Answer: At 41 acceptable items from a sample of 50, the cost of $2,000 is less than the other two

alternatives (send out lot: $2,203.96, send out and retool: $2,003.96). At this point, it is 67.3% probable that the lot is 80% good.

Program Listing

```
1  PRINT " BAYESIAN DECISION ANALYSIS"
2  PRINT
10  DIM P1(4),P2(4),P3(4),P5(4),A(4,4),M(3)
15  DEF  FN R(Z1) =  INT (Z1 * 1000 + 0.5) / 1000
20  PRINT "HOW MANY STATES OF NATURE ";
30  INPUT N1
40  PRINT "HOW MANY CONDITIONAL ACTIONS ";
50  INPUT A1
60  PRINT
70  PRINT "ENTER PAYOFFS FOR:"
79  REM  ENTER PAYOFF MATRIX
80  FOR I = 1 TO A1
90  FOR J = 1 TO N1
100  PRINT "ACTION ";I;" UNDER STATE ";J;" ";
110  INPUT A(I,J)
120  NEXT J
130  NEXT I
140  PRINT
149 XO = 0
150  PRINT "ENTER PRIOR AND CONDITIONAL PROB.:"
160  FOR I = 1 TO N1
165  PRINT "FOR STATE ";I;" ";
170  INPUT P1(I),P2(I)
180 XO = XO + P1(I)
185 P3(I) = 0
190  NEXT I
200  IF XO = 1 THEN 230
210  PRINT "PRIOR PROBABILITIES DO NOT EQUAL 1.0"
220  GOTO 140
230  PRINT
240  PRINT "ENTER SAMPLE SIZE (0 TO END) ";
250  INPUT S
255  IF S = 0 THEN 670
260  PRINT
290  PRINT "ENTER ACTUAL NUMBER OF SUCCESSES ";
300  INPUT I1
301  REM  CALCULATE EXPECTED COST FOR SAMPLE SIZE
320 M(1) = S
330 M(2) = I1
340 M(3) = S - I1
350  FOR J = 1 TO 3
360  IF M(J) = 0 THEN 420
370 Z = 0
380  FOR K = 1 TO M(J)
390 Z = Z +  LOG (K)
400  NEXT K
410 M(J) = Z
420  NEXT J
430 P4 = 0
```

```
450  FOR H = 1 TO N1
459  REM   STORE LIKELIHOOD IN P5()
460 Y = I1 *  LOG (P2(H)) + (S - I1) *  LOG (1 - P2(H))
465 P5(H) =  FN R( EXP (M(1) - M(2) - M(3) + Y))
469  REM  STORE JOINT PROBABILITY IN P3()
470 P3(H) = P5(H) * P1(H)
474  REM  SUM POINT PROBABILITIES IN P3()
475 P4 = P4 + P3(H)
480  NEXT H
489  REM  CALCULATE EXPECTED MONETARY VALUES
490  FOR I = 1 TO A1
500 E(I) = 0
510  FOR J = 1 TO N1
520 E(I) = E(I) + (A(I,J) * (P3(J) / P4))
530  NEXT J
535  NEXT I
540  PRINT
550  PRINT "GIVEN ";I1;" SUCCESSES IN A SAMPLE"
560  PRINT "OF ";S;",";"THE EXPECTED VALUES ARE:"
570  FOR I = 1 TO A1
580  PRINT "ACTION ";I;": ";E(I)
590  NEXT I
600  PRINT
610  PRINT "PROBABILITY REVISIONS: "
620  PRINT "STATE PRIOR LIKELIHOOD JOINT POSTERIOR"
630  FOR I = 1 TO N1
640  PRINT I; TAB( 7);P1(I); TAB( 13);P5(I);
645  PRINT  TAB( 22);P3(I); TAB( 30); FN R(P3(I) / P4)
650  NEXT I
660  GOTO 240
670  END
```

References

Cabot and Harnett. *An Introduction to Management Science.* Reading, Mass.: Addison-Wesley, 1977.

Economic Order Quantity

The purpose of this program is to determine the economic order quantity of an item. You must enter the number of available price breaks, minimum and maximum quantities and unit price for each level, the inventory holding cost as a percentage of each unit's cost, cost of placing an order (in dollars), and the annual demand quantity. The program will compute the EOQ of each price break and indicate if the quantity is within the minimum and maximum quantities for that level.

Program Notes

It may be more convenient for you to enter holding costs as a fixed dollar amount per unit. Make these changes:

```
150  PRINT "ENTER THE UNIT HOLDING COST"
155  PRINT "($) ";
  .
  .
200 H = H / 100                (DELETE THIS LINE)
  .
  .
310 E =  INT ( SQR ((2 * D * S) / H))
```

Your price breaks may be computed as a percentage discount from a fixed price. Make these changes:

```
60   PRINT "ENTER THE NUMBER OF AVAILABLE PRICE"
65   PRINT "BREAKS ";
70   INPUT B
72   PRINT "ENTER THE BASE UNIT PRICE ";
74   INPUT U1
80   PRINT
90   PRINT "ENTER MINIMUM QUANTITY, MAXIMUM"
95   PRINT "QUANTITY, DISCOUNT"
100   FOR I = 1 TO B
110   PRINT "AT PRICE BREAK ";I;" ";
120   INPUT Q(1,I),Q(2,I),D1
130   NEXT I
140   PRINT
```

Example

Joe Blow, purchasing agent for a small manufacturer, needs to order motor armatures from a machine shop. The machine shop offers three price breaks to Joe's company: 0 to 499 units, $5.00 per unit; 500 to 999, $4.50 per unit; 1,000 and up, $3.90 per unit. Joe's company requires 10,000 units each year. $20.00 in clerks' time and forms is needed to place an order. About 20% of each unit's cost is spent on warehousing, shipping, breakage, and so forth. How many orders of how many units should be placed this year in order to minimize costs?

Answer: Joe should place 15 orders of 666 units each.

```
ECONOMIC ORDER QUANTITY

ENTER THE NUMBER OF AVAILABLE PRICE
BREAKS ?3

ENTER MINIMUM QUANTITY, MAXIMUM
QUANTITY, PRICE
AT PRICE BREAK 1 ?0,499,5
AT PRICE BREAK 2 ?500,999,4.5
AT PRICE BREAK 3 ?1000,99999,3.9

ENTER THE UNIT HOLDING COST
(% PER UNIT) ?20
ENTER THE COST OF PLACING AN
ORDER ($)?20
ENTER THE DEMAND QUANTITY PER YEAR
(0=END)?10000

EOQ # OF    QUANTITIES UNIT PRICE
    ORDERS
632   16      0-499       5--NOT POSSIBLE
666   15      500-999     4.5
716   14      1000-99999  3.9--NOT POSSIBLE

ENTER THE DEMAND QUANTITY PER YEAR
(0=END)?0

WOULD YOU LIKE TO RE-RUN THIS PROGRAM
WITH NEW DATA? (Y/N) ?N
```

Practice Problems

1. Three price breaks: $2.50 per unit for 0-999 units; $2.25 each for 1,000-1,999 units; 2,000-9,999 units cost $2.00 each. Cost of placing an order is $50.00, and holding costs represent 10% of an item's cost. What is the EOQ if annual demand is 5,065 units?

Answer: EOQ is four orders of 1,500 units each.

2. Four price breaks: $89.00 each for 0-9 units; $82.50 per unit for 10 to 19 units; 20 to 29 units are $78.00 each; 30 and up are $75.00 apiece. Cost of placing an order is $75.00. Holding costs are 15%. What is the EOQ if annual use is 50 units?

Answer: The EOQ is two orders of 25 units each.

Program Listing

```
10  PRINT "ECONOMIC ORDER QUANTITY"
20  REM  -- CHANGE SIZE OF ARRAYS Q(2,N)
25  REM  -- AND U(N) AS NECESSARY WHERE N
30  REM  -- = MAXIMUM NUMBER OF PRICE
35  REM  -- BREAKS YOU WILL USE
40  DIM Q(2,10),U(10)
50  PRINT
60  PRINT "ENTER THE NUMBER OF AVAILABLE PRICE"
65  PRINT "BREAKS ";
```

```
70  INPUT B
80  PRINT
90  PRINT "ENTER MINIMUM QUANTITY, MAXIMUM"
95  PRINT "QUANTITY, PRICE"
100  FOR I = 1 TO B
110  PRINT "AT PRICE BREAK ";I;" ";
120  INPUT Q(1,I),Q(2,I),U(I)
130  NEXT I
140  PRINT
150  PRINT "ENTER THE UNIT HOLDING COST"
155  PRINT "(% PER UNIT) ";
160  INPUT H
170  IF H > 0 THEN 200
180  PRINT "HOLDING COST MUST BE GREATER THAN ZERO."
190  GOTO 150
200 H = H / 100
210  PRINT "ENTER THE COST OF PLACING AN"
215  PRINT "ORDER ($)";
220  INPUT S
230  PRINT "ENTER THE DEMAND QUANTITY PER YEAR"
235  PRINT "(0=END)";
240  INPUT D
250  IF D = 0 THEN 440
260  PRINT
270  REM  OUTPUT THE RESULTS
280  PRINT "EOQ # OF   QUANTITIES UNIT PRICE"
285  PRINT "    ORDERS"
290  REM  CALCULATE EOQ BY FORMULA FOR
295  REM  EACH PRICE BREAK
300  FOR I = 1 TO B
310 E =  INT ( SQR ((2 * D * S) / (U(I) * H)))
320  PRINT E; TAB( 5); INT (D / E + 0.9); TAB( 12);Q(1,I);"-";Q(2,I);
325  PRINT  TAB( 23);U(I);
330  REM  TEST TO SEE IF EOQ FALLS WITHIN
335  REM  ORDER QUANTITY FOR THIS PRICE
340  IF Q(1,I) > E THEN 390
350  IF Q(2,I) < E THEN 390
360  PRINT
370  GOTO 400
380  REM  PRICE BREAK IS NOT AVAILABLE
385  REM  AT THIS EOQ
390  PRINT "--NOT POSSIBLE"
400  NEXT I
410  PRINT
420  GOTO 230
430  REM  RESTART OF END PROGRAM?
440  PRINT
450  PRINT "WOULD YOU LIKE TO RE-RUN THIS PROGRAM"
455  PRINT "WITH NEW DATA? (Y/N) ";
460  INPUT Z$
470  IF Z$ = "Y" THEN 50
480  IF Z$ <  > "N" THEN 450
490  END
```

References

Chase and Aquilano. *Production and Operations Management.* Homewood, Ill.: Richard D. Irwin, Inc., 1977.

McLaughlin and Pickhardt. *Quantitative Techniques for Management Decisions.* New York: McGraw-Hill, 1975.

Economic Production Quantity

It is often useful to know the optimal quantity of an item to produce in order to minimize expenses. This program computes that quantity for a given item, and incorporates simultaneous sales calculations (where units are being sold while more are being produced).

To use the program, enter the rate of production, the sales or use rate (the average number of units removed from inventory each day), the total number of units sold in a year, the holding cost (in dollars per unit), and the set-up cost. The program will output the optimal number of setups per year, and the optimum quantity to produce in each lot. The optimum quantity is that which minimizes set-up and carrying costs.

Example

Waldo's Paint Factory produces several different paint colors using a single mixing and filling machine. The machine will produce 300 gallons each day, and currently Waldo ships 125 gallons of each color every day, and 35,000 gallons per year. Holding costs are $0.15 per gallon. For each lot produced, the machine must be completely cleaned, at a cost of $150. How many lots of each color per year should Waldo produce? How many gallons in each lot?

Answer: Each year, Waldo should run three lots of 11,666 gallons each.

```
ECONOMIC PRODUCTION QUANTITY

ENTER THE RATE OF PRODUCTION
(UNITS/DAY) ?300
ENTER THE SALES OR USE RATE
(UNITS/DAY) ?125
ENTER ANNUAL SALES OF USE ?35000
ENTER THE UNIT HOLDING COST
($ PER UNIT) ?.15
ENTER THE SETUP COST ($) ?150

OPTIMAL NUMBER OF SETUPS = 3
PER YEAR
EPQ= 11666 UNITS

WOULD YOU LIKE TO RE-RUN THIS PROGRAM
WITH NEW DATA? (Y/N) ?N
```

Practice Problems

1. Daily production of 45 units, daily sales of 20 units. Annual sales total 4,000 units. Holding costs are $0.67 per unit. Set-up costs are $25.00. What is the EPQ?

Answer: Five lots of 800 units each.

2. 50 units per day are produced, 35 are sold. Annually, 6,500 units are sold. Holding costs are $0.45 per unit. Set-up costs are $60.00 per lot. How many lots are optimum? What size lots?

Answer: Three lots of 2,166 units each.

Program Listing

```
10  PRINT "ECONOMIC PRODUCTION QUANTITY
20  PRINT
30  PRINT "ENTER THE RATE OF PRODUCTION"
35  PRINT "(UNITS/DAY) ";
40  INPUT R
50  IF R > 0 THEN 100
60  PRINT
70  PRINT "PRODUCTION RATE MUST BE GREATER"
75  PRINT "THAN ZERO."
80  PRINT
90  GOTO 30
100  PRINT "ENTER THE SALES OR USE RATE "
105  PRINT "(UNITS/DAY) ";
110  INPUT U
120  IF U >  = 0 THEN 170
130  PRINT
140  PRINT "SALES (USE) RATE MUST BE NON-ZERO."
150  PRINT
160  GOTO 100
170  PRINT "ENTER ANNUAL SALES OF USE ";
180  INPUT H
190  IF H >  = U THEN 240
200  PRINT
210  PRINT "ANNUAL RATE MUST BE HIGHER THAN"
215  PRINT "DAILY RATE."
220  PRINT
230  GOTO 170
240  PRINT "ENTER THE UNIT HOLDING COST"
245  PRINT "($ PER UNIT) ";
250  INPUT J
260  IF J > 0 THEN 310
270  PRINT
280  PRINT "HOLDING COST MUST BE GREATER THAN ZERO."
290  PRINT
300  GOTO 240
310  PRINT "ENTER THE SETUP COST ($) ";
320  INPUT S
330  PRINT
340  IF S > 0 THEN 380
350  PRINT "SETUP COST MUST BE GREATER THAN ZERO."
360  PRINT
370  GOTO 310
380  REM  OUTPUT THE RESULTS
390 N =  INT ( SQR (((J * H) / (2 * S)) * (1 - (U / R))) + 0.5)
400  PRINT "OPTIMAL NUMBER OF SETUPS = ";N
405  PRINT "PER YEAR"
410  PRINT "EPQ= "; INT (H / N);" UNITS"
420  REM  RESTART OF END PROGRAM?
430  PRINT
440  PRINT "WOULD YOU LIKE TO RE-RUN THIS PROGRAM"
445  PRINT "WITH NEW DATA? (Y/N) ";
450  INPUT Z$
```

```
460   IF Z$ = "Y" THEN 20
470   IF Z$ <  > "N" THEN 440
480   END
```

Reference

McLaughlin and Pickhardt. *Quantitative Techniques for Management Decisions.* New York: McGraw-Hill, 1975.

Statistical Estimation Theory

Statistical estimation theory is the science of determining unbiased estimates for various statistics from sample figures, establishing confidence interval estimates for those statistics, and determining the number of samples that must be taken to reduce the probability of error in these estimates to stated maxima. This program performs these calculations.

At the start of the program you must enter the size of the sample, the mean of the sample, and the sample variance. The program then prints the unbiased estimate of the population variance and, for both the mean and the standard deviation, each of seven different confidence levels, the confidence interval estimate, and the maximum and minimum values produced thereby. You may then have the program calculate how large a sample you would have to take to reduce the error of your estimate to a given maximum. You enter the desired confidence level, the maximum desired error, and whether you are testing the mean or the standard deviation. The program then calculates the sample size needed.

Example

A government researcher did a study to determine how long people had to wait in line at the post office. He took 100 samples. The mean of the sample was 15 minutes, and the sample variance was 2.02. At each of the seven confidence levels, what is the maximum and minimum for the mean and standard deviation? How many samples would have to be taken to be 99% confident that the error in the mean was no greater than 0.2?

Answer:

```
STATISTICAL ESTIMATION THEORY

ENTER NUMBER OF SAMPLES TAKEN ?100
ENTER MEAN OF SAMPLE ?15
ENTER SAMPLE VARIANCE?2.02
UNBIASED ESTIMATE OF SIGMA SQUARED
POPULATION VARIANCE = 2.04040404
CONFIDENCE INTERVAL ESTIMATES FOR MEAN:

CONFIDENCE PLUS OR
              MINUS     MAXIMUM     MINIMUM
50        .096346016   15.096346   14.903654
60        .120219488   15.1202195  14.8797805
70        .148046977   15.148047   14.851953
80        .183060302   15.1830603  14.8169397
90        .234955361   15.2349554  14.7650446
95        .279966588   15.2799666  14.7200334
99        .367938199   15.3679382  14.6320618

CONFIDENCE INTERVAL ESTIMATES
FOR STANDARD DEVIATION:

50        .0681269213  1.49655404  1.3603002
60        .0850080148  1.51343514  1.34341911
70        .104685021   1.53311214  1.3237421
80        .129443181   1.5578703   1.29898394
```

```
90        .166138529   1.59456565 1.26228859
95        .197966273   1.62639339 1.23046085
99        .260171595   1.68859872 1.16825553

DO YOU WANT A CALCULATION OF HOW LARGE
A SAMPLE YOU MUST TAKE TO REDUCE
THE ERROR OF YOUR ESTIMATE TO A
MAXIMUM QUANTITY? (Y/N)
?Y
ENTER YOUR CHOSEN CONFIDENCE LEVEL
(FROM ABOVE CHOICES ONLY),1 FOR 50,
2 FOR 60, 3 FOR 70, 4 FOR 80, 5 FOR 90,
6 FOR 95, AND 7 FOR 99
?7
ENTER MAXIMUM DESIRED ERROR OF ESTIMATE
?0.2
ARE YOU TESTING THE MEAN (M) OR THE
STANDARD DEVIATION (S)?
?M
AT THE 99 PERCENT CONFIDENCE LEVEL
IT WOULD BE NECESSARY TO TAKE 339
SAMPLES TO BE SURE THAT YOUR ESTIMATE
OF THE ERROR IN THE MEAN
DID NOT EXCEED .2
DO YOU HAVE NO MORE CALCULATIONS (0),
MORE WITH THE SAME SAMPLES (1), OR
BRAND-NEW SAMPLING (2)?
?0
```

Practice Problems

1. Using the data from the above example, how many samples would have to be taken to reduce the error in the standard deviation to 0.0746353654 at the 99% confidence level?
Answer: 1,204

2. If all the data is the same as in the above example, how many samples must be taken to reduce the error in the mean to 0.0995503798 at the 95% confidence level?
Answer: 784

Program Listing

```
10  PRINT "STATISTICAL ESTIMATION THEORY"
20  DIM C(7),F(7)
25  PRINT
29  REM  READ CONFIDENCE LEVELS AND COEFFICIENTS
30  FOR I = 1 TO 7
40  READ C(I),F(I)
50  NEXT I
60  PRINT "ENTER NUMBER OF SAMPLES TAKEN ";
70  INPUT N
80  PRINT "ENTER MEAN OF SAMPLE ";
90  INPUT X
```

```
100  PRINT "ENTER SAMPLE VARIANCE";
110  INPUT S2
120 S1 = S2 * N / (N - 1)
130  PRINT "UNBIASED ESTIMATE OF SIGMA SQUARED"
140  PRINT "POPULATION VARIANCE = ";S1
150 S =  SQR (S1)
280 S3 = S /  SQR (N)
290  PRINT "CONFIDENCE INTERVAL ESTIMATES FOR MEAN:"
300  PRINT
310  PRINT "CONFIDENCE PLUS OR"
320  PRINT "              MINUS     MAXIMUM     MINIMUM"
330  FOR I = 1 TO 7
340  PRINT C(I); TAB( 8);F(I) * S3; TAB( 20);X + F(I) * S3; TAB( 31);
     X - F(I) * S3
350  NEXT I
360  PRINT
370  PRINT "CONFIDENCE INTERVAL ESTIMATES"
380  PRINT "FOR STANDARD DEVIATION:"
390  PRINT
400  FOR I = 1 TO 7
410 J = F(I) * S /  SQR (2 * N)
420  PRINT C(I); TAB( 8);J; TAB( 20);S + J; TAB( 31);S - J
430  NEXT I
440  PRINT
450  PRINT "DO YOU WANT A CALCULATION OF HOW LARGE"
460  PRINT "A SAMPLE YOU MUST TAKE TO REDUCE"
470  PRINT "THE ERROR OF YOUR ESTIMATE TO A"
480  PRINT "MAXIMUM QUANTITY? (Y/N)"
490  INPUT B$
500  IF B$ = "N" THEN 780
510  IF B$ <  > "Y" THEN 450
520  PRINT "ENTER YOUR CHOSEN CONFIDENCE LEVEL"
530  PRINT "(FROM ABOVE CHOICES ONLY),1 FOR 50,"
540  PRINT "2 FOR 60, 3 FOR 70, 4 FOR 80, 5 FOR 90,"
545  PRINT "6 FOR 95, AND 7 FOR 99"
550  INPUT J
560  PRINT "ENTER MAXIMUM DESIRED ERROR OF ESTIMATE"
570  INPUT M
580  PRINT "ARE YOU TESTING THE MEAN (M) OR THE "
590  PRINT "STANDARD DEVIATION (S)?"
600  INPUT C$
610  IF C$ = "S" THEN 680
620  IF C$ <  > "M" THEN 580
640 N3 =  INT ((S * F(J) / M) ^ 2) + 1
650  GOTO 690
680 N3 =  INT (((F(J) * S / M) ^ 2 / 2) + 1
690  PRINT "AT THE ";C(J);" PERCENT CONFIDENCE LEVEL"
700  PRINT "IT WOULD BE NECESSARY TO TAKE ";N3
710  PRINT "SAMPLES TO BE SURE THAT YOUR ESTIMATE"
720  PRINT "OF THE ERROR IN THE ";
730  IF C$ = "S" THEN 760
740  PRINT "MEAN"
750  GOTO 770
760  PRINT "STANDARD DEVIATION"
770  PRINT "DID NOT EXCEED ";M
```

```
780  PRINT "DO YOU HAVE NO MORE CALCULATIONS (0),"
790  PRINT "MORE WITH THE SAME SAMPLES (1), OR"
800  PRINT "BRAND-NEW SAMPLING (2)?"
810  INPUT Y
820  IF Y = 1 THEN 440
830  IF Y = 2 THEN 60
900  DATA  50,0.6744902454373
910  DATA  60,0.8416214285714
920  DATA  70,1.0364335334476
930  DATA  80,1.2815515669516
940  DATA  90,1.6448536821705
950  DATA  95,1.9599641025641
960  DATA   99,2.575827586207
999  END
```

References

Harnett. *Introduction to Statistical Methods.* 2nd ed. Reading, Mass.: Addison-Wesley, 1975.

Spiegal. *Statistics.* New York: McGraw-Hill, 1961.

Statistics

This program analyzes grouped and ungrouped data which you enter, and prints as many as 26 statistics: measures of central tendency, variance, skewness, kurtosis, and correlation.

When you run the program, enter the total population (if known), or 0 (if unknown). If the data are grouped, enter G; if ungrouped, enter U. The next step is to enter the frequency, followed by the value observed at that frequency. After the last item, enter a frequency and value of 0. If you are entering ungrouped data, just enter the observations; enter 9E9 after the last one. The program then calculates and prints the statistics, indicating which are not available based on the data entered.

Program Notes

This program accepts a maximum of 250 grouped or ungrouped observations. To change this, modify lines 10 and 15 of the program as follows:

```
10 DIM S(40),X(I),Y(I),Z(I)
15 N1=I
```

Replace the expression I with a constant equal to the maximum number of observations.

Example

Randy Flashpan is a local disk jockey. His weekly show has a segment during which listeners phone in their evaluations of certain songs by rating them on a scale of one to ten. One hundred listeners called in their scores on one record, and their scores are listed below:

Score	Number of Listeners
1	13
2	6
3	2
4	4
5	10
6	13
7	22
8	18
9	10
10	2

In Randy's lexicon, a song with a median score of seven or more is "boss hit-bound." If the median is between five and seven, the song is classified as "lukewarm." If the median falls below four, the record is dropped from the radio station's playlist.

Based on the sample data shown, how should Randy classify the record? Furthermore, how does someone with the intelligence of a disk jockey run this program?

Answer: This song resides in the lukewarm category, with a median of 6.59.

```
STATISTICS

ENTER TOTAL POPULATION (0=UNKNOWN) ?100

ARE DATA (G) GROUPED OR (U) UNGROUPED ?G
```

```
ENTER FREQUENCY, THEN VALUE
(0,0 TO END)
PAIR NO. 1 ?13,1
PAIR NO. 2 ?6,2
PAIR NO. 3 ?2,3
PAIR NO. 4 ?4,4
PAIR NO. 5 ?10,5
PAIR NO. 6 ?13,6
PAIR NO. 7 ?22,7
PAIR NO. 8 ?18,8
PAIR NO. 9 ?10,9
PAIR NO. 10 ?2,10
PAIR NO. 11 ?0,0
RESULTS TABULATED AS FOLLOWS:
TOTAL POPULATION: 100
DATA ARE: GROUPED
NO. OF SAMPLES: 100
SUM OF SAMPLES: 583
MEAN: 5.83
SUM OF SQUARES: 4077
MEAN DEVIATION: 2.141
MEDIAN: 6.59090909
VARIANCE: 6.78109996
STANDARD DEVIATION: 2.60405452
UNBIASED ESTIMATE OF VARIANCE:
6.84959592
STANDARD DEVIATION USING THAT VARIANCE:
2.61717327
PROBABLE ERROR: 1.75640874
STANDARD ERROR OF MEAN: .261717327
COEFF. OF VARIATION: 44.6664584%
3RD MOMENT ABOUT MEAN: -11.946726
4TH MOMENT ABOUT MEAN: 105.989549
MOMENT COEFF. SKEWNESS: -.676548108
MOMENT COEFF. KURTOSIS: 2.30495658
UNBIASED ESTIMATE 3RD CENT. MOMENT:
-12.3136735
STANDARD ERROR MEAN WITH FINITE POPULA-
TION CORRECTION FACTOR: 0
PEARSON'S 2ND COEFF. SKEWNESS:
-.876605023
RANGE: 9
INDEX OF MEAN DEVIATION TO PRODUCT OF
M.A.E. AND STANDARD DEVIATION:
1.03044907
```

Practice Problems

1. Meter readings from a holding tank at a fuel processing plant are: 12.98, 13.001, 18.25, 4.4, 9.8, 11, 14.5, 12.7, 7.2, and 6.1. What are the mean and median meter readings? What is the standard deviation?
Answer: The mean reading is 10.9931; the median is 11.85. The standard deviation is 3.98843859.

2. An actuarial clerk wants statistics on the population of Casper County relative to the occurrence of heart disease. The table below shows age brackets and the number of diagnosed heart disease cases for those ages:

Age	Diagnosed Cases (per 1000 people)
0-5	6
6-10	5
11-20	3
21-25	8
26-30	7
31-35	12
36-40	17
41-45	19
46-50	30
51-55	35
56-60	43
61-65	50
66-70	61

What is the median age of the onset of heart disease in Casper County? Twelve hundred cases were evaluated. What is the measure of skewness for this population, since it appears to be skewed to the right of the mean? What is the standard error of the mean? (Hint: You must increase array sizes on line 10 to 300.)

Answer: The median age is 58.1976744 for the onset of heart disease. Skewness −1.26117836. The standard error of the mean is 0.903236727.

Program Listing

```
1  PRINT "STATISTICS"
2  PRINT
10  DIM S(40),X(250),Y(250),Z(250)
14  REM  N1=DIMENSION OF X, Y & Z
15 N1 = 250
20  FOR I = 1 TO 40
29  REM  READ CONFIDENCE LEVELS AND COEFFICIENTS
30 S(I) = 0
40  NEXT I
45 S1 = 0
50  PRINT "ENTER TOTAL POPULATION ";
55  PRINT "(0=UNKNOWN) ";
70  INPUT T9
75  PRINT
80  PRINT "ARE DATA (G) GROUPED OR (U) UNGROUPED ";
90  INPUT U$
95  PRINT
100  IF U$ = "G" THEN 440
104  REM  --- UNGROUPED DATA
105 J = 1
110  PRINT "ITEM NO. ";J;" "
119  REM  - ENTER 9E9 AFTER LAST ITEM
120  INPUT X(J)
130  IF X(J) <  > 9E9 THEN 150
140 J = J - 1
145  GOTO 190
149  REM  - CALCULATES NO. OF ITEMS
150 S(1) = S(1) + 1
```

```
159  REM  - CALCULATE SUM OF ITEMS
160 S(2) = S(2) + X(J)
169  REM  - CALCULATES THE SUM OF SQUARES
170 S(4) = S(4) + X(J) * X(J)
175 J = J + 1
180  IF J < N1 THEN 110
189  REM  - CALCULATES MEAN
190 S(3) = S(2) / S(1)
209  REM  - CALCULATES DEVIATION FROM MEAN
210 S(5) =  ABS (S(3) - X(J))
219  REM  - CALCULATES SUM OF DEVIATIONS
220 S(6) = S(6) + S(5)
229  REM  - CALCULATES 3RD POWER OF DEVIATION
230 S(8) = (X(J) - S(3)) ^ 3
239  REM  - CALCULATES SUM OF 3RD POWERS
240 S(9) = S(9) + S(8)
249  REM  - CALCULATES 4TH POWER OF DEVIATION
250 S(10) = (X(J) - S(3)) ^ 4
259  REM  - CALCULATES SUM OF 4TH POWERS
260 S(11) = S(11) + S(10)
279  REM  - CALCULATES MEAN DEVIATION
280 S(7) = S(6) / S(1)
288  REM  - USE SHELL-METZNER SORT TO
289  REM  - ARRANGE DATA IN ASCENDING ORDER
290 M1 = S(1)
295 M1 =  INT (M1 / 2)
300  IF M1 = 0 THEN 370
305 K = S(1) - M1
310 J = 1
315 I = J
320 L = I + M1
325  IF X(I) <  = X(L) THEN 355
330 W = X(I)
335 X(I) = X(L)
340 X(L) = W
345 I = I - M1
350  IF I >  = M1 THEN 320
355 J = J + 1
360  IF J > K THEN 295
365  GOTO 315
369  REM  - CALCULATE MEDIAN
370  IF S(1) / 2 =  INT (S(1) / 2) THEN 410
379  REM  - ODD NO. OF ITEMS
380 M = S(1) / 2 + 0.5
390 S(12) = X(M)
400  GOTO 840
409  REM  - EVEN NO. OF ITEMS
410 M = S(1) / 2
420 S(12) = (X(M) + X(M + 1)) / 2
430  GOTO 840
439  REM  ---- GROUPED DATA ----
440  PRINT "ENTER FREQUENCY, THEN VALUE"
442  PRINT "(0,0 TO END) "
445 J = 1
450  PRINT "PAIR NO. ";J;" ";
```

```
459  REM  - CALCULATE ABSOLUTE DEVIATION
460  INPUT Y(J),Z(J)
470  IF Y(J) = 0 THEN 529
489  REM  - CALCULATE NO. OF SAMPLES
490 S(1) = S(1) + Y(J)
495 S1 = S1 + 1
499  REM  - CALCULATE TOTAL OF VALUES
500 S(2) = S(2) + Y(J) * Z(J)
509  REM  - CALCULATE SUM OF SQUARES
510 S(4) = S(4) + Y(J) * Z(J) * Z(J)
520 J = J + 1
525  IF J <  = N1 THEN 450
529  REM  - CALCULATE MEAN
530 S(3) = S(2) / S(1)
540  FOR J = 1 TO S(1)
550 S(5) = Y(J) *  ABS (S(3) - Z(J))
559  REM  - CALCULATE SUM OF ABS. DEVIATIONS
560 S(6) = S(6) + S(5)
569  REM  - CALCULATE 3RD POWER OF DEVIATIONS
570 S(8) = Y(J) * (Z(J) - S(3)) ^ 3
579  REM  - CALCULATE SUM OF 3RD POWERS
580 S(9) = S(9) + S(8)
589  REM  - CALCULATE 4TH POWERS OF DEVIATIONS
590 S(10) = Y(J) * (Z(J) - S(3)) ^ 4
599  REM  - CALCULATE SUM OF 4TH POWERS
600 S(11) = S(11) + S(10)
610  NEXT J
619  REM  - CALCULATE MEAN DEVIATION
620 S(7) = S(6) / S(1)
628  REM  - USE SHELL- METZNER SORT TO
629  REM  - ARRANGE DATA IN ASCENDING ORDER
630 M1 = S1
635 M1 =  INT (M1 / 2)
640  IF M1 = 0 THEN 740
645 K = S1 - M1
650 J = 1
655 I = J
660 L = I + M1
665  IF Z(I) <  = Z(L) THEN 710
670 V = Y(I)
675 W = Z(I)
680 Y(I) = Y(L)
685 Z(I) = Z(L)
690 Y(L) = V
695 Z(L) = W
700 I = I - M1
705  IF I >  = 1 THEN 660
710 J = J + 1
715  IF J > K THEN 635
720  GOTO 655
730  IF C$ = "S" THEN 760
739  REM  - CALCULATES MEDIAN
740 T = 0
750 K = 1
760  IF T + Y(K) > S(1) / 2 THEN 800
```

```
765 T = T + Y(K)
770 K = K + 1
780  GOTO 760
785  IF K <  = S(1) THEN 750
790  PRINT "MORE WITH THE SAME SAMPLES (1), OR"
800 P = ((Z(K) - Z(K - 1)) / Y(K)) * (S(1) / 2 - T)
810 S(12) = (Z(K) + Z(K - 1)) / 2 + P
840 N = S(1)
850  PRINT "RESULTS TABULATED AS FOLLOWS:"
860  PRINT "TOTAL POPULATION: ";
870  IF T9 = 0 THEN 900
880  PRINT T9
890  GOTO 910
900  PRINT "UNKNOWN/NOT INDICATED"
905  PRINT
910  PRINT "DATA ARE: ";
920  IF U$ = "G" THEN 950
930  PRINT "UNGROUPED"
940  GOTO 960
950  PRINT "GROUPED"
960  PRINT "NO. OF SAMPLES: ";S(1)
970  PRINT "SUM OF SAMPLES: ";S(2)
980  PRINT "MEAN: ";S(3)
990  PRINT "SUM OF SQUARES: ";S(4)
1000  PRINT "MEAN DEVIATION: ";S(7)
1010  PRINT "MEDIAN: ";S(12)
1020 S(13) = S(4) / N - S(3) ^ 2
1030  PRINT "VARIANCE: ";S(13)
1040  IF U$ = "G" THEN 1070
1050 S(14) = S(13) - (1 / 12) * (Z(2) - Z(1)) ^ 2
1060  PRINT "VARIANCE WITH SHEP. CORR.: ";S(14)
1070 S(15) =  SQR (S(13))
1080  PRINT "STANDARD DEVIATION: ";S(15)
1090  IF U$ = "G" THEN 1120
1100 S(16) =  SQR (S(14))
1110  PRINT "STANDARD DEVIATION WITH SHEP. CORR.:"
1115  PRINT S(16)
1120 S(17) = S(13) * N / (N - 1)
1130  PRINT "UBIASED ESTIMATE OF VARIANCE:"
1135  PRINT S(17)
1140 S(18) =  SQR (S(17))
1150  PRINT "STANDARD DEVIATION USING THAT VARIANCE:"
1155  PRINT S(18)
1160 S(19) = .67449 * S(15)
1170  PRINT "PROBABLE ERROR: ";S(19)
1180 S(20) =  SQR (S(17) / N)
1190  PRINT "STANDARD ERROR OF MEAN: ";S(20)
1200 S(21) = S(15) / S(3)
1210  PRINT "COEFF. OF VARIATION: ";100 * S(21);"%"
1220 S(22) = S(9) / N
1230  PRINT "3RD MOMENT ABOUT MEAN: ";S(22)
1240 S(23) = S(11) / N
1250  PRINT "4TH MOMENT ABOUT MEAN: ";S(23)
1260  IF U$ = "G" THEN 1300
1270 R = Z(2) - Z(1)
```

```
1280 S(24) = S(23) - 0.5 * (R ^ 2) * S(17) + (7 / 240) * R ^
1290  PRINT "4TH MOMENT WITH SHEP. CORR.:"
1295  PRINT S(24)
1300 S(25) = S(22) / (S(15) ^ 3)
1310  PRINT "MOMENT COEFF. SKEWNESS: ";S(25)
1320 S(26) = S(23) / (S(13) ^ 2)
1330  PRINT "MOMENT COEFF. KURTOSIS: ";S(26)
1340 S(27) = (S(22) * N ^ 2) / ((N - 1) * (N - 2))
1350  PRINT "UNBIASED ESTIMATE 3RD CENT. MOMENT:"
1355  PRINT S(27)
1360  IF T9 = 0 THEN 1420
1370  IF N <  = 0.05 * T9 THEN 1420
1380 S(28) = S(20) *  SQR ((T9 - N) / (T9 - 1))
1390  PRINT "STANDARD ERROR MEAN WITH FINITE POPULA-"
1400  PRINT "TION CORRECTION FACTOR: ";S(28)
1410  GOTO 1430
1420  PRINT "FINITE POPULATION CORRECTION FACTOR N/A"
1430 S(29) = 3 * (S(3) - S(12)) / S(15)
1440  PRINT "PEARSON'S 2ND COEFF. SKEWNESS:"
1445  PRINT S(29)
1450  IF U$ = "G" THEN 1480
1460 S(30) = X(N) - X(1)
1470  GOTO 1490
1480 S(30) = Z(S1) - Z(1)
1490  PRINT "RANGE: ";S(30)
1500 S(31) = S(7) / (.7978845608 * S(15))
1510  PRINT "INDEX OF MEAN DEVIATION TO PRODUCT OF"
1520  PRINT "M.A.E. AND STANDARD DEVIATION:"
1525  PRINT S(31)
1530  END
```

References

Mendenhall, William, et al. *Statistics: A Tool for the Social Sciences.* Belmont, Calif.: Duxbury Press, 1974.

Spiegal. *Statistics* (Schaum's Series). New York: McGraw-Hill, 1961.

Unbiased Estimator of Standard Deviation

The concept of an unbiased estimator of the standard deviation is not common among American statisticians. However, according to the Russian mathematician A. A. Sveshnikov, the unbiased estimator of the standard deviation is given by the following formula:

$$\tilde{\sigma} = K_N \sqrt{\frac{1}{N-1} \sum_{J=1}^{N} (x_j - \tilde{x})^2}$$

where

$$K_N = \sqrt{\frac{N-1}{2}} \left(\frac{\Gamma\left(\frac{N-1}{2}\right)}{\Gamma\left(\frac{N}{2}\right)} \right)$$

Using this symbolism N = sample size, it is easily shown that:

for N = 2M (even sample size),

$$K_N = \sqrt{\frac{N-1}{2}} \left(\frac{\frac{2M-3}{2} \cdot \frac{2M-5}{2} \cdots \frac{3}{2} \cdot \frac{1}{2} \sqrt{\pi}}{(M-1)\,(M-2) \cdots 2 \cdot 1} \right)$$

while for N = 2M+1 (odd sample size),

$$K_N = \sqrt{\frac{N-1}{2}} \left(\frac{(M-1)\,(M-2) \cdots 2 \cdot 1}{\frac{2M-1}{2} \cdot \frac{2M-3}{2} \cdots \frac{3}{2} \cdot \frac{1}{2} \sqrt{\pi}} \right)$$

To use the program, you must enter the number of samples, and the sum of the squares of the deviations. The program prints out the unbiased estimator of the standard deviation, and asks if you want another calculation.

Example

In a class of 35 seventh grade students, the sum of the squares of the deviations for their ages is 3.156. What is the unbiased estimator of the standard deviation?

Answer: 0.30691769

```
UNBIASED ESTIMATOR OF
STANDARD DEVIATION

THIS PROGRAM CALCULATES THE UNBIASED
ESTIMATOR OF THE STANDARD DEVIATION
WHEN VARIABLE IS NORMALLY DISTRIBUTED

ENTER THE SUM OF THE SQUARES
OF THE DEVIATIONS ?3.156
ENTER THE NUMBER OF SAMPLES ?35
UNBIASED ESTIMATOR OF STANDARD
DEVIATION = .30691769
ANOTHER CALCULATION? (Y/N) ?N
```

Practice Problems

1. If 40 samples are randomly distributed and the sum of the squares of their deviations is 9.63, what is the unbiased estimator of the standard deviation?

Answer: 0.500108775

2. In a group of 26 randomly distributed samples, the sum of the squares of the deviations is 34.953. What is the unbiased estimator of the standard deviation?

Answer: 1.1943016

Program Listing

```
5  PRINT "UNBIASED ESTIMATOR OF"
7  PRINT "STANDARD DEVIATION"
8  PRINT
10  PRINT "THIS PROGRAM CALCULATES THE UNBIASED"
20  PRINT "ESTIMATOR OF THE STANDARD DEVIATION"
30  PRINT "WHEN VARIABLE IS NORMALLY DISTRIBUTED"
40  PRINT
50  PRINT "ENTER THE SUM OF THE SQUARES
60  PRINT "OF THE DEVIATIONS ";
70  INPUT S
80  PRINT "ENTER THE NUMBER OF SAMPLES ";
90  INPUT N
99  REM  COMPUTE K-SUB-N TERM
100 A =  SQR ((N - 1) / 2)
110  FOR M = (((N - 1) / 2) - 1) TO 1 STEP  - 1
120 A = A * M / (M + 0.5)
130  NEXT M
139  REM  SQR(PI)/2=.8862269255
140 P = .8862269255
150  IF N / 2 =  INT (N / 2) THEN 170
159  REM  ODD SAMPLE SIZE
160 P = 1 / P
170  PRINT "UNBIASED ESTIMATOR OF STANDARD"
180  PRINT "DEVIATION = ";A * P *  SQR (S / (N - 1))
190  PRINT "ANOTHER CALCULATION? (Y/N) ";
200  INPUT Y$
210  IF Y$ = "Y" THEN 50
220  END
```

References

National Bureau of Standards. *Handbook of Mathematical Functions.* Washington, D.C., 1966.

Sveshnikov, A. A. *Problems in Probability Theory, Mathematical Statistics and Theory of Random Functions.* New York: Dover, 1968.

Chi-Square

The chi-square test in statistics tests the compatibility of observed frequencies with the expected or theoretical frequencies. For example, suppose we are testing whether a die is fair or biased. We throw the die 60 times, recording the result each time. If the die is fair, we would expect that each of the six sides would come up close to ten times during the test. But we know that events do not always correspond to theoretical expectations. The chi-square test provides the means of determining whether the observed and theoretical results are so divergent that the die cannot be considered fair.

Chi-square is defined as follows:

$$x^2 = \sum_{I=1}^{K} \frac{(O_I - E_I)^2}{E_I}$$

where O represented the observed frequencies and E the expected frequencies. Statisticians have determined what value (the "5% critical value") the chi-square must be below in order that we be 95% positive that two results are compatible. This program tests whether the actual results fall within that level of confidence. It also employs Yates's correction (which some statisticians prefer and some dislike) to test the results. The chi-square formula with Yates's correction is

$$x^2 = \sum_{I=1}^{K} \left(\frac{|O_I - E_I| - 0.5}{E_I} \right)^2$$

The program also tests whether the results are too good (below the 95% critical value), which makes clinical workers suspicious of the results.

The program first asks if the expected frequency is a constant. In the above example, each face of the die is expected to appear ten times, so the answer is "Yes" and you would enter ten as the constant. You then enter the observed frequencies one by one; enter 99999 after the last one. If the expected frequencies are not constant, the program will ask for each set of observed and expected frequencies. After the last entry, enter 99999,1 to end the sequence.

The program will then calculate the chi-square statistics, both with and without Yates's correction, and print them out, indicating the degrees of freedom. It then tests each statistic against the 5% and 95% critical values, and prints out the results.

Example

Suppose the results of the 60 throws of the die in the above example are as follows:

Face	Expected	Actual
1	10	9
2	10	8
3	10	12
4	10	10
5	10	13
6	10	8

What are the results of the chi-square test for this data? Can the die be considered fair?

Answer: The die can be considered fair.

```
CHI-SQUARE

IS THE AMOUNT OF EXPECTED FREQUENCY
CONSTANT? (Y/N) ?Y
ENTER CONSTANT EXPECTED FREQUENCY ?10
ENTER OBSERVED FREQUENCIES ONE BY ONE
AS REQUESTED BELOW
ENTER 99999 TO END
?9
?8
?12
?10
?13
?8
?99999
CHI SQUARE FOR THESE
OBSERVATIONS = 2.2
FOR 5 DEGREES OF FREEDOM
SQUARE = 1.35
FIVE PERCENT CRITICAL VALUE OF
CHI SQUARE IS 11.071
THEREFORE THE HYPOTHESIS IS NOT
REJECTED AT THE 5% CRITICAL VALUE
```

Practice Problems

1. A student in a genetics class is performing an experiment to test classical Mendelian theory. That theory predicts that certain biological characteristics should appear in the species under review in the ratios 900:300:300:100. In the 1,600 samples which the student takes, they appear 904, 297, 302, and 97 times, respectively. Are these results compatible with orthodox Mendelian theory?

Answer: The unadjusted chi-square result is 0.151111111, and with Yates's correction that result is 0.104444444. The 5% critical value for three degrees of freedom is 7.8147, so the results are compatible. However, the 95% critical value is 0.35185, so either with or without Yates's correction, the results are "too good," and the instructor must view the student's experiment with suspicion.

2. A Las Vegas pit boss noticed that a particular roulette wheel seemed to be coming up red more often than black. He kept track of the next 1,000 spins; red came up 546 times, and black 454 times. Is the wheel biased?

Answer: The chi-square without Yates's correction is 8.46400001, and with it is 8.28100001. The 5% critical value is 3.8415, and the hypothesis is therefore rejected. The pit boss should junk that roulette wheel immediately.

Program Listing

```
10  PRINT "CHI-SQUARE"
20  PRINT
100  PRINT "IS THE AMOUNT OF EXPECTED FREQUENCY"
110  PRINT "CONSTANT? (Y/N) ";
120  INPUT A$
130  IF A$ = "N" THEN 500
135  IF A$ <  > "Y" THEN 100
140  PRINT "ENTER CONSTANT EXPECTED FREQUENCY ";
150  INPUT Y
```

```
299  REM  EXPECTED FREQUENCY IS A CONSTANT
300  PRINT "ENTER OBSERVED FREQUENCIES ONE BY ONE"
310  PRINT "AS REQUESTED BELOW"
315  PRINT "ENTER 99999 TO END"
320  INPUT X
330  IF X = 99999 THEN 1000
350 N = N + 1
370 S = S + ( ABS (X - Y) ^ 2) / Y
390 T = T + (( ABS (X - Y) - 0.5) ^ 2) / Y
400  IF A$ = "N" THEN 520
410  GOTO 320
499  REM  EXPECTED FREQUENCY IS NOT A CONSTANT
500  PRINT "ENTER, PAIR BY PAIR AS REQUESTED, THE"
510  PRINT "OBSERVED, THEN THE EXPECTED,"
515  PRINT "FREQUENCIES"
517  PRINT "ENTER 99999,1 TO END"
520  INPUT X,Y
530  GOTO 330
1000  PRINT "CHI-SQUARE FOR THESE"
1010  PRINT "OBSERVATIONS = ";S
1020  PRINT "FOR ";N - 1;" DEGREES OF FREEDOM"
1030  PRINT "WITH YATES'S CORRECTION, CHI-"
1040  PRINT "SQUARE = ";T
1099  REM  BRANCH FOR CALCULATION OF CRITICAL VALUES
1100  IF N > 101 THEN 1600
1110  IF N = 101 THEN 1500
1120  IF N > 31 THEN 1400
1200  FOR I = 1 TO N - 1
1210  READ C
1220  NEXT I
1230  FOR I = N TO N + 29
1240  READ D
1250  NEXT I
1260  GOTO 2500
1400 W = 1.6449 *  SQR (2 / (9 * (N - 1))) ^ 3
1405 C = (N - 1) * (1 - 2 / (9 * (N - 1)) + W
1410 D = (N - 1) * (1 - 2 / (9 * (N - 1)) - W
1420  GOTO 2500
1500 C = 124.342
1510 D = 77.9295
1520  GOTO 2500
1600 C = 0.5 * (1.6449 +  SQR (2 * (N - 1) - 1))) ^ 2
1610 D = 0.5 * ( SQR (2 / (9 * (N - 1)) - 1.6449) ^ 2
2500  PRINT "FIVE PERCENT CRITICAL VALUE OF"
2510  PRINT "CHI-SQUARE IS ";C
2520  IF T > C THEN 2700
2530  IF S > C THEN 2800
2540  IF S < D OR T < D THEN 2900
2600  PRINT "THEREFORE THE HYPOTHESIS IS NOT"
2610  PRINT "REJECTED AT THE 5% CRITICAL VALUE"
2620  GOTO 9999
2700  PRINT "THEREFORE THE HYPOTHESIS IS"
2710  PRINT "REJECTED AT THE 5% CRITICAL VALUE"
2720  GOTO 9999
2800  PRINT "WHILE THE UNADJUSTED CHI-SQUARE"
```

```
2810  PRINT "VALUES ARE UNACCEPTABLE, THOSE WITH"
2820  PRINT "YATES'S CORRECTION ARE NOT; THEREFORE"
2830  PRINT "SAMPLE SIZES SHOULD BE INCREASED OR"
2840  PRINT "SUBSTITUTE MULTINOMIAL DISTRIBUTION"
2850  PRINT "METHODS"
2860  GOTO 9999
2900  PRINT "AGREEMENT IS TOO GOOD AND SHOULD BE"
2910  PRINT "EXAMINED CRITICALLY, BECAUSE EITHER"
2920  PRINT "WITH OR WITHOUT YATES'S CORRECTION, THE"
2930  PRINT "CHI SQUARE VALUE IS BELOW THE 95%"
2940  PRINT "CRITICAL VALUE"
5000  DATA  3.8415,5.9915,7.8147,9.4877,11.071, 12.592
5010  DATA  14.067,15.507,16.919,18.307,19.675,21.026
5020  DATA  22.362,23.685,24.996,26.296,27.587,28.869
5030  DATA  30.140,31.410,32.671,33.924,35.173,36.415
5040  DATA  37.653,38.885,40.113,41.337,42.557,43.773
5050  DATA  .003932,.10259,.35185,.71072,1.1455
5060  DATA  1.635,2.167,2.733,3.325,3.940
5070  DATA  4.575,5.226,5.892,6.571,7.261
5080  DATA  7.962,8.672,9.390,10.117,10.851
5090  DATA  11.591,12.338,13.091,13.848,14.611
5100  DATA  15.379,16.151,16.928,17.708,18.493
9999  END
```

References

Hoel. *Introduction to Mathematical Statistics,* 2nd ed. New York: John Wiley, 1954.

Spiegel. *Statistics* (Schaum's series). New York: McGraw-Hill, 1961.

Data Forecasting Divergence

This program determines the degree to which a forecast diverges from actual data. You enter pairs of actual data and corresponding forecast. After the last data pair, enter 99999,1. The program will then print out the number of pairs of figures, the total error, the total absolute error, the total squared error, the mean error, the mean absolute error (MAE), the mean square error, and the root mean square error.

Example

A statistical forecaster determined the following data having made the following respective forecasts:

Data	Forecast
1	1.0
2	2.2
3	2.9
4	3.9
5	5.3
6	6.1
7	7.0
8	7.9

What are the error statistics for these figures?

Answer: Number of pairs = 8; total error = 0.300000001; total absolute error = 0.899999999; total squared error = 0.17; mean error = 0.0375000001; mean absolute error = 0.1125; mean square error = 0.02125; root mean square error = 0.145773797.

```
DATA FORECASTING DIVERGENCE

ENTER DATA AND FORECAST
(99999,1 TO END)
?1,1
?2,2.2
?3,2.9
?4,3.9
?5,5.3
?6,6.1
?7,7
?8,7.9
?99999,1
NO. OF PAIRS OF FIGURES = 8
TOTAL ERROR = -.300000001
TOTAL ABSOLUTE ERROR = .899999999
TOTAL SQUARED ERROR = .17
MEAN ERROR = -.0375000001
MEAN ABSOLUTE ERROR = .1125
MEAN SQUARE ERROR = .02125
ROOT MEAN SQUARE ERROR = .145773797
```

Practice Problems

1. The actual and predicted results in a city council race are as follows:

	Vote %	Poll %
Candidate A	40.3	42.7
Candidate B	22.5	21.4
Candidate C	16.3	18.2
Candidate D	10.5	6.0
Candidate E	7.2	7.4
Candidate F	3.2	4.3

How accurate were the polls?

Answer: Number of pairs = 6; total error ≈ 0; total absolute error = 11.2; total squared error = 32.0800001; mean error ≈ 0; mean absolute error = 1.86666667; mean square error = 5.34666668; root mean square error = 2.31228603.

2. A new television weatherman lasted only one week at the station. Following are the actual and predicted temperatures during that week:

	Actual Temperature	Predicted Temperature
Monday	74	49
Tuesday	70	62
Wednesday	58	75
Thursday	60	82
Friday	65	37
Saturday	73	58
Sunday	70	92

What statistics were on the dismissal notice?

Answer: Number of pairs = 7; total error = 15; total absolute error = 137; total squared error = 2955; mean error = 2.14285714; mean absolute error = 19.5714286; mean square error = 422.142858; root mean square error = 20.5461154.

Program Listing

```
10  PRINT "DATA FORECASTING DIVERGENCE"
15  PRINT
20  PRINT "ENTER DATA AND FORECAST"
30  PRINT "(99999,1 TO END)"
40  INPUT X,Y
50  IF X = 99999 THEN 110
60 T1 = T1 + 1
70 T2 = T2 + X - Y
80 T3 = T3 +  ABS (X - Y)
90 T4 = T4 + ( ABS (X - Y)) ^ 2
100  GOTO 40
110  PRINT "NO. OF PAIRS OF FIGURES = ";T1
120  PRINT "TOTAL ERROR = ";T2
130  PRINT "TOTAL ABSOLUTE ERROR = ";T3
140  PRINT "TOTAL SQUARED ERROR = ";T4
150  PRINT "MEAN ERROR = ";T2 / T1
160  PRINT "MEAN ABSOLUTE ERROR = ";T3 / T1
170  PRINT "MEAN SQUARE ERROR = ";T4 / T1
180  PRINT "ROOT MEAN SQUARE ERROR = "; SQR (T4 / T1)
190  END
```

Reference

Gilchrist. *Statistical Forecasting.* London: John Wiley, 1976.

Newtonian Interpolation

This program applies to Newton's forward difference formula for interpolation of a given function. Newton's formula is intended to work when the arguments you use in the interpolation commence just below the argument for which you are seeking the tabular value.

You first enter the independent variables on either side of the value for which you want the tabular value interpolated, followed by that value (your desired independent variable). The program then asks for the precision (in decimal places) you want in your answer. This should not exceed the accuracy of either your original data, or your computer's Basic. The program will cease calculating differences when they drop below this level of accuracy.

You then enter the tabular values immediately below and above the desired tabular value. The program prints out the difference between these values, called the first difference. The program asks for additional tabular values, printing out the new difference each time, until the new difference drops below the level of precision you entered earlier. To end the entry of tabular values before this, you enter 99999 as the new tabular value, and the program will branch to computation of the answer.

Example

Bill Miller is going to take out a five-year loan at 4¼%. He has a table that shows the factors by which he should multiply the principle of the loan to determine the amount of each monthly payment. Unfortunately, the table only gives figures at half-percent intervals. How should Bill use this program to determine the factor at 4¼%?

Interest Rate	Factor
4%	0.018416522
4½%	0.018643019
5%	0.018871233
5½%	0.019101162
6%	0.019332801
6½%	0.019566148
7%	0.019801198
7½%	0.020037949
8%	0.020276394

Answer:

```
          INTERPOLATION
NEWTON'S FORWARD DIFFERENCE FORMULA

  LOWER INDEPENDENT VARIABLE ?.04
  UPPER INDEPENDENT VARIABLE ?.045
DESIRED INDEPENDENT VARIABLE ?.0425
PRECISION (IN DECIMAL PLACES) ?9

ENTER TABULAR VALUE AT .04 ?.018416522
ENTER TABULAR VALUE AT .045 ?.018643019

1ST DIFFERENCE = 2.26496995E-04
ENTER TABULAR VALUE AT .05 ?.018871233

2ND DIFFERENCE = 1.71700231E-06
ENTER TABULAR VALUE AT .055 ?.019101162
```

```
3RD DIFFERENCE = -1.99361239E-09
INTERPOLATION IS TO THE ORDER OF
3RD DIFFERENCES ANSWER = .0185295558
```

Program Problems

1. Jeanne needs to know the sine of 0.63, using the following table. What is that figure?

X	0.6	0.7	0.8	0.9	1.0
SIN X	0.564642	0.0644218	0.717356	0.783327	0.841471

Answer: The sine of 0.63 is approximately 0.58919079.

2. Joe Statistics wants to determine the area under the normal curve at 0.095 standard deviation to the right of the mean. From the following table, what is that area?

Standard Deviations	0.08	0.09	0.1	0.11	0.12
Area	0.53188	0.53586	0.53983	0.54380	0.54776

Answer: The area is 0.53784625.

Program Listing

```
10  PRINT "           INTERPOLATION"
20  PRINT "NEWTON'S FORWARD DIFFERENCE FORMULA"
25  PRINT
30  PRINT "  LOWER INDEPENDENT VARIABLE ";
40  INPUT A(1)
50  PRINT "  UPPER INDEPENDENT VARIABLE ";
60  INPUT A(2)
70  PRINT "DESIRED INDEPENDENT VARIABLE ";
80  INPUT X
90 P = (X - A(1)) / (A(2) - A(1))
100  PRINT "PRECISION (IN DECIMAL PLACES) ";
110  INPUT E
120  IF E = 0 THEN 140
130 E = 1 / (10 ^ E)
140 J = 1
150  PRINT
160  GOSUB 470
170 J = 2
180  GOSUB 470
190  IF B(1,J) = 99999 THEN 300
200  FOR I = 2 TO J
210 B(I,J - I + 1) = B(I - 1,J - I + 2) - B(I - 1,J - I + 1)
220  NEXT I
230  PRINT
240  PRINT J - 1;
250  GOSUB 500
260  PRINT " DIFFERENCE = ";B(J,1)
270  IF B(J,1) < E THEN 300
280 J = J + 1
290  IF J <  = 9 THEN 180
300 Z = 0
```

```
310 P1 = 1
320 X = 1
330  FOR I = 1 TO 8
340 X = X * I
350 P1 = P1 * (P - I + 1)
360 Z = Z + P1 * B(I + 1,1) / X
370  NEXT I
380  IF A(2) > A(1) THEN 410
390 Z = B(1,1) - Z
400  GOTO 420
410 Z = B(1,1) + Z
420  PRINT "INTERPOLATION IS TO THE ORDER OF"
430  PRINT J - 1;
440  GOSUB 500
450  PRINT " DIFFERENCES ANSWER = ";Z
460  GOTO 590
469  REM  SUBROUTINE TO ENTER TABULAR VALUES
470  PRINT "ENTER TABULAR VALUE AT ";A(1) + (J - 1) * (A(2) - A(1));" ";
480  INPUT B(1,J)
490  RETURN
499  REM  ROUTINE TO PRINT "ST", "ND", ETC
500  IF J <  > 2 THEN 520
510  PRINT "ST";
520  IF J <  > 3 THEN 540
530  PRINT "ND";
540  IF J <  > 4 THEN 560
550  PRINT "RD";
560  IF J < 5 THEN 580
570  PRINT "TH";
580  RETURN
590  END
```

References

Hildebrand, F.B. *Introduction to Numerical Analysis,* 2nd. ed. New York: McGraw-Hill, 1974.

National Bureau of Standards. *Handbook of Mathematical Functions.* Washington, D.C., 1976.

Phillips, G. M., and Taylor, R.J. *Theory and Applications of Numerical Analysis.* New York: Academic Press, 1973.

Scheid. *Numerical Analysis.* New York: McGraw-Hill, 1968.

Lagrangian Interpolation

This program applies Lagrange's formula for interpolation to a given function. For each succeeding tabular value you enter, the program displays the corresponding difference. Starting with the second difference, you may either calculate the interpolated value or proceed to the next order of difference. If you go on, you have one more option at each succeeding order of difference, and that is to back up to calculate the interpolated value on the previous order of difference. This effectively lets you take an uncommitted look ahead to see whether the next order of difference is smaller than the present one. Thus, you need not choose the order of difference beforehand. The program permits three-point through ten-point Lagrangian interpolation.

The program first asks you for the central argument, which is the argument immediately *below* the one you want. It also requests the next higher argument listed in the table, and your desired argument. You must then enter tabular values for the central argument and the arguments on either side of the central argument. The program calls these values f_0, f_1, and f_{-1}, respectively.

At this point the program displays the first and second differences. You have the option of stopping here with three-point interpolation, or going on to the higher orders of difference. If you go on you must enter, one at a time, tabular values f_2, f_{-2}, f_3, ..., f_5. As you make each entry, the program displays the next higher difference. You must decide whether to stop and interpolate based on that difference, back up to interpolate on the previous difference, or proceed to enter another tabular value. You can only proceed as far as the ninth difference, since the program calculates at most a ten-point interpolation.

Program Notes

The program employs the algorithm set forth by Pearson for simplifying the Lagrangian coefficients, thus precluding the need for coefficient tables. The program also disregards the remainder term in Lagrange's formula. Finally, the program does not perform two-point interpolation, since it is of little use.

Example

Using the following table, determine the sine of 1.00006 radians.

Angle X in Radians	Tabular Value Sin X	Name of Tabular Value
0.996	0.83930 30496	f_{-4}
0.997	0.83984 62937	f_{-3}
0.998	0.84038 86980	f_{-2}
0.999	0.84093 02619	f_{-1}
1.000	0.84147 09848	f_0
1.001	0.84201 08663	f_1
1.002	0.84254 99058	f_2
1.003	0.84308 81027	f_3
1.004	0.84362 54565	f_4
1.005	0.84416 19667	f_5

Answer: 0.841795015

```
LAGRANGIAN INTERPOLATION

ENTER THE CENTRAL ARGUMENT, NEXT HIGHER
ARGUMENT, AND THE DESIRED ARGUMENT
?1,1.001,1.0006
ENTER F(0) ?.841470985
ENTER F(1) ?.842010866
ENTER F(-1) ?.840930262
DIFFERENCE # 1 = 5.3988141E-04
DIFFERENCE # 2 = 8.41217116E-07
DO YOU WANT FURTHER DIFFERENCES?
(Y/N) ?Y
ENTER F(2) ?.842549906
DIFFERENCE # 3 = 4.65661287E-10
WANT FURTHER DIFFERENCES?
YES(Y), NO(N), ONE LESS(L) ?Y
ENTER F(-2) ?.840388698
DIFFERENCE # 4 = 2.32830644E-10
WANT FURTHER DIFFERENCES?
YES(Y), NO(N), ONE LESS(L) ?Y
ENTER F(3) ?.843088103
DIFFERENCE # 5 = 2.32830644E-10
WANT FURTHER DIFFERENCES?
YES(Y), NO(N), ONE LESS(L) ?N
LAGRANGIAN 6-POINT INTERPOLATION
PRODUCES A VALUE OF .841795015
```

Practice Problems

1. What is the sine of 1.0001 radians?
   Answer: 0.841525014

2. To ten places, the mantissas of the common logarithms of certain arguments are shown below:

Argument	Mantissa
6.1242	0.787 0493 652
6.1243	0.787 0564 565
6.1244	0.787 0635 478
6.1245	0.787 0706 390
6.1246	0.787 0777 300
6.1247	0.787 0848 209

What is the common logarithm mantissa for 6.12449?
Answer: 0.787069729

Program Listing

```
5  PRINT "LAGRANGIAN INTERPOLATION"
7  PRINT
10  DIM D(10,10),E(10),F(10,10),N(4),G(10)
15 G(1) = 1
20 F(1,1) = 1
29  REM  SET UP INITIAL TABLES OF VALUES
30  FOR I = 2 TO 10
```

```
40 F(I,1) =  SGN (I / 2 -  INT (I / 2) - .1)
50  FOR J = 2 TO I
60 WP =  INT ((I + J) / 2) - (I + J) / 2 + 0.1
65 F(I,J) = ( ABS (F(I - 1,J - 1)) +  ABS (F(I - 1,J))) *  SGN (WP)
70  NEXT J
80 G(I) = G(I - 1) * (I - 1)
90  NEXT I
100  PRINT "ENTER THE CENTRAL ARGUMENT, NEXT HIGHER"
110  PRINT "ARGUMENT, AND THE DESIRED ARGUMENT"
120  INPUT X1,X2,X3
130 P = (X3 - X1) / (X2 - X1)
140  IF P < 0 OR P > 1 THEN 100
160  PRINT "ENTER F(0) ";
170  INPUT D(1,1)
180  PRINT "ENTER F(1) ";
190  INPUT D(2,1)
200  PRINT "ENTER F(-1) ";
210  INPUT D(3,1)
220 D(1,2) =  ABS (D(2,1) - D(1,1))
230  PRINT "DIFFERENCE # 1 = ";D(1,2)
240 D(2,2) =  ABS (D(3,1) - D(1,1))
250 D(1,3) =  ABS (D(2,2) - D(1,2))
260  PRINT "DIFFERENCE # 2 = ";D(1,3)
269  REM  GIVE OPERATOR OPTION OF STOPPING NOW OR CONTINUING
270  PRINT "DO YOU WANT FURTHER DIFFERENCES?"
275  PRINT "(Y/N) ";
280  INPUT Y$
290 I = 3
300  IF Y$ = "N" THEN 570
310  IF Y$ <  > "Y" THEN 270
320 I = I + 1
330  PRINT "ENTER F(";
340  IF I / 2 =  INT (I / 2) THEN 360
350  PRINT "-";
360  PRINT  INT (I / 2);") ";
370  INPUT D(I,1)
380  FOR J = 1 TO I - 2
390 D(I - J,J + 1) =  ABS (D(I - J + 1,J) - D(I - J - 1,J))
400  NEXT J
410 D(1,I) =  ABS (D(1,I - 1) - D(2,I - 1))
420  PRINT "DIFFERENCE # ";I - 1;" = ";D(1,I)
425  IF I = 10 THEN 510
430  PRINT "WANT FURTHER DIFFERENCES?"
440  PRINT "YES(Y), NO(N), ONE LESS(L) ";
450  INPUT Y$
458  REM  OPERATOR MAY STOP NOW, CONTINUE,
459  REM  OR GO BACK TO ONE LESS DIFFERENCE
460  IF Y$ = "N" THEN 570
470  IF Y$ = "Y" THEN 320
480  IF Y$ <  > "L" THEN 430
490 I = I - 1
500  GOTO 570
509  REM  NO MORE THAN NINE DIFFERENCES POSSIBLE
510  PRINT "WANT NINTH DIFFERENCE (N), OR"
520  PRINT "ONLY EIGHT DIFFERENCE (E) ";
```

```
530  INPUT Y$
540  IF Y$ = "N" THEN 570
550  IF Y$ <  > "E" THEN 510
560 I = I - 1
568  REM  LINES 570 TO 630 SET UP VARIABLES
569  REM  USED IN PEARSON'S ALGORITHM
570 N(1) = P ^ 3 - P
580 N(2) = N(1) * (P ^ 2 - 4)
590 N(3) = N(3) * (P ^ 2 - 16)
610  FOR J = 1 TO 10
620 E(J) = D( ABS (11 - (J * 2)) +  SGN ( INT (J / 6)),1) / (P + 5 - J)
630  NEXT J
640  FOR J = 1 TO I
650 T = T + E( INT ((10 - I) / 2) + J) * F(I,J)
660  NEXT J
670  IF I / 2 <  >  INT (I / 2) THEN 690
680 T = T * (P - I / 2)
690  PRINT "LAGRANGIAN ";I;"-POINT INTERPOLATION"
700  PRINT "PRODUCES A VALUE OF ";T * N( INT ((I - 1) / 2)) / G(I)
710  END
```

References

National Bureau of Standards. *Handbook of Mathematical Functions.* Washington, D.C., 1966.

Scheid. *Numerical Analysis* (Schaum's series). New York: McGraw-Hill, 1968.

Vega. *Vollständige Sammlung grösserer logarithmisch-trigonometrischer Tafeln.* 1794. Reprint. New York: Hafner, 1958.

Sums of Powers

This program calculates the sum of the Pth powers (up to the 10th powers) of the first N integers. It will also compute the sums of powers which are not the first N integers, but instead a series of higher integers. For example, if you want the sum of squares of numbers 101 to 1,000, subtract the total of the first 100 squares from the total of the first 1,000.

Program Notes

Clearly, a simple algorithm exists for computing the sums of powers: a loop with provision for adding the successive powers obtained. When you want the sum of very lengthy series of integers, the methods in this program are more efficient.

Example

What is the sum of the first ten 7th powers?
Answer: 18,080,425

```
SUM OF POWERS

THIS PROGRAM COMPUTES THE SUM OF THE
P-TH POWERS (LIMIT: 10) FOR THE FIRST
N INTEGERS. ENTER P AND N ?7,10
THE SUM OF THE 7TH POWERS OF
THE FIRST 10 INTEGERS IS 18080425
```

Practice Problems

1. What is the sum of the first 100 5th powers?
Answer: 1.717083335 times 10^{11}.

2. What is the sum of the first six 10th powers?
Answer: 71,340,451.1

3. What is the sum of the squares of the numbers from 101 to 1,000?
Answer: 333,495,150

Program Listing

```
5  PRINT "SUM OF POWERS"
7  PRINT
10  PRINT "THIS PROGRAM COMPUTES THE SUM OF THE"
15  PRINT "P-TH POWERS (LIMIT: 10) FOR THE FIRST"
20  PRINT "N INTEGERS. ENTER P AND N ";
30  INPUT P,N
```

```
32 P =  INT (P)
35  IF P < 1 OR P > 10 THEN 10
39  REM  BRANCH TO PROPER POWER
40  IF P = 1 THEN 50
41  IF P = 2 THEN 70
42  IF P = 3 THEN 90
43  IF P = 4 THEN 110
44  IF P = 5 THEN 130
45  IF P = 6 THEN 150
46  IF P = 7 THEN 180
47  IF P = 8 THEN 210
48  IF P = 9 THEN 240
49  IF P = 10 THEN 270
50 S = N * (N + 1) / 2
55  PRINT "THE SUM OF THE FIRST POWERS OF"
60  GOTO 380
70 S = N * (N + 1) * (2 * N + 1) / 6
75  PRINT "THE SUM OF THE SECOND POWERS OF"
80  GOTO 380
90 S = (N ^ 2) * ((N + 1) ^ 2) / 4
95  PRINT "THE SUM OF THE THIRD POWERS OF"
100  GOTO 380
110 S = N * (N + 1) * (2 * N + 1) * (3 * N ^ 2 + 3 * N - 1) / 30
120  GOTO 370
130 S = (N ^ 2) * ((N + 1) ^ 2) * (2 * N ^ 2 + 2 * N - 1) / 12
140  GOTO 370
150 S1 = (2 * N + 1) * (3 * N ^ 4 + 6 + N ^ 3 - 3 * N + 1)
160 S = N * (N + 1) * S1 / 24
170  GOTO 370
180 S1 = 3 * N ^ 4 + 6 * N ^ 3 - N ^ 2 - 4 * N + 2
190 S = (N ^ 2) * ((N + 1) ^ 2) * S1 / 24
200  GOTO 370
210 S1 = 5 * N ^ 6 + 15 * N ^ 5 + 5 * N ^ 4 - 15 * N ^ 3 - N ^ 2 + 9 *
    N - 3
220 S = N * (N + 1) * (2 * N + 1) * S1 / 90
230  GOTO 370
240 S1 = 2 * N ^ 6 + 6 * N ^ 5 + N ^ 4 - 8 * N ^ 3 + N ^ 2 + 6 * N - 3
250 S = (N ^ 2) * ((N + 1) ^ 2) * S1 / 20
260  GOTO 370
270 S2 = 3 * N ^ 8 + 12 * N ^ 7 + 8 * N ^ 6 - 18 * N ^ 5
280 S1 = S2 - 10 * N ^ 4 + 24 * N ^ 3 + 2 * N ^ 2 - 15 * N + 5
290 S = N * (N + 1) * (2 * N + 1) * S1 / 66
300  GOTO 370
370  PRINT "THE SUM OF THE ";P;"TH POWERS OF"
380  PRINT "THE FIRST ";N;" INTEGERS IS ";S
390  END
```

Reference

Chemical Rubber Co. *Handbook of Tables for Mathematicians,* 4th ed. Cleveland: 1970.

Factorials

This program calculates the factorial of an integer. For the factorial of a small number N we recursively multiply the integers from 1 through N. For larger numbers this becomes impractical, and we instead use Stirling's approximation:

$$N! \simeq e^{-N} N^N \sqrt{2 N \pi}$$

This has very high accuracy for large N.

Program Notes

Note that for any given computer there is a theoretical limit beyond which overflow cannot be avoided.

Example

How much is 8!?
Answer: 40320

```
FACTORIALS

ENTER THE NUMBER WHOSE
FACTORIAL YOU WANT?8
THE FACTORIAL OF 8 IS
40320
TIMES 10 TO THE POWER 0
COMPUTED RECURSIVELY
```

Practice Problems

1. How much is 100!?
Answer: $9.32484812 \times 10^{157}$

2. What is the factorial of 20?
Answer: $2.43290201 \times 10^{18}$

3. How much is 141!?
Answer: $1.89702238 \times 10^{243}$

Program Listing

```
5  PRINT "FACTORIALS"
7  PRINT
10  PRINT "ENTER THE NUMBER WHOSE"
15  PRINT "FACTORIAL YOU WANT";
20  INPUT N
```

```
30 F = 1
50  IF N > 69 THEN 150
59  REM  CALCULATE USING RECURSIVE ALGORITHM
60  FOR I = 2 TO N
70 F = F * I
80  IF F < 1E + 10 THEN 120
100 F = F / (1E + 10)
110 J = J + 10
120  NEXT I
130  GOTO 300
149  REM  CALCULATE USING STIRLING'S APPROXIMATION
150 K =  INT (N / 5)
160 I = I + 5
170  IF I > K * 5 THEN 280
180 F = (F * N ^ 5) /  EXP (5)
190  IF F > 1E + 30 THEN 220
200  IF F > 1E + 20 THEN 250
210  GOTO 160
220 F = F / (1E + 30)
230 J = J + 30
240  GOTO 190
250 F = F / (1E + 20)
260 J = J + 20
270  GOTO 190
280 X =  SQR (N * 6.28318530718)
290 F = (F * N ^ (N - K * 5)) /  EXP (N - K * 5) * X
300  PRINT "THE FACTORIAL OF ";N;" IS"
310  PRINT F
320  PRINT "TIMES 10 TO THE POWER ";J
330  IF K > 0 THEN 360
340  PRINT "COMPUTED RECURSIVELY"
350  GOTO 370
360  PRINT "COMPUTED BY STIRLING'S APPROXIMATION"
370  END
```

References

Korn & Korn. *Mathematical Handbook,* 2nd ed. New York: McGraw-Hill, 1968.

National Bureau of Standards. *Handbook of Mathematical Functions.* Washington, D.C., 1966.

Temperature Conversion

Chemists, physicists, and other scientists are constantly involved in taking temperatures in one scale and converting them to other scales. In science, temperatures are commonly recorded and manipulated in five scales: Fahrenheit, Celsius (formerly called centigrade), Réaumur, Kelvin, and Rankine. This program takes any temperature (above absolute zero) recorded in any scale and converts it into all four of the other scales.

Example

Convert 98.6° Fahrenheit into the other scales.

```
TEMPERATURE CONVERSION

WHAT IS THE TEMPERATURE WHICH
YOU WISH TO BE CONVERTED? ?98.6
IN WHAT SCALE WAS THAT RECORDED?
ENTER 1 FOR FAHRENHEIT, 2 FOR
CELSIUS, 3 FOR REAUMUR, 4 FOR
KELVIN, 5 FOR RANKINE ?1
98.6            DEGREES FAHRENHEIT =
37              DEGREES CELSIUS
29.6            DEGREES REAUMUR
310.1           DEGREES KELVIN
558.18          DEGREES RANKINE
```

Practice Problems

1. The boiling point of water is 100° Celsius. What is it on the other scales?
Answer: 212° Fahrenheit, 80° Réaumur, 373.1° Kelvin, 671.58° Rankine.

2. Lonna keeps her hot tub at 104° Fahrenheit. How hot is it on the other scales?
Answer: 40° Celsius, 32° Réaumur, 313.1° Kelvin, 563.58° Rankine.

Program Listing

```
5  PRINT "TEMPERATURE CONVERSION"
7  PRINT
10  PRINT "WHAT IS THE TEMPERATURE WHICH"
20  PRINT "YOU WISH TO BE CONVERTED? ";
30  INPUT T
40  PRINT "IN WHAT SCALE WAS THAT RECORDED? "
50  PRINT "ENTER 1 FOR FAHRENHEIT, 2 FOR"
60  PRINT "CELSIUS, 3 FOR REAUMUR, 4 FOR"
70  PRINT "KELVIN, 5 FOR RANKINE ";
80  INPUT S
90 S =  INT (S)
```

```
100  IF S < 1 THEN 40
110  IF S > 5 THEN 40
119  REM  BRANCH ON TYPE OF SCALE
120  IF S = 1 THEN 130
121  IF S = 2 THEN 170
122  IF S = 3 THEN 210
123  IF S = 4 THEN 250
124  IF S = 5 THEN 290
130  IF T <  - 459.58 THEN 420
140 T1 = T
150  PRINT T,"DEGREES FAHRENHEIT ="
160  GOTO 340
170  IF T <  - 273.1 THEN 420
180 T1 = 32 + T * 1.8
190  PRINT T,"DEGREES CELSIUS ="
200  GOTO 320
210  IF T <  - 218.48 THEN 420
220 T1 = 32 + T * 2.25
230  PRINT T,"DEGREES REAUMUR ="
240  GOTO 320
250  IF T < 0 THEN 420
260 T1 = 32 + 1.8 * (T - 273.1)
270  PRINT T,"DEGREES KELVIN ="
280  GOTO 320
290  IF T < 0 THEN 420
300 T1 = T - 459.58
310  PRINT T,"DEGREES RANKINE ="
320  PRINT T1,"DEGREES FAHRENHEIT"
330  IF S = 2 THEN 360
340  PRINT 5 * (T1 - 32) / 9,"DEGREES CELSIUS"
350  IF S = 3 THEN 380
360  PRINT 4 * (T1 - 32) / 9,"DEGREES REAUMUR"
370  IF S = 4 THEN 400
380  PRINT 5 * (T1 - 32) / 9 + 273.1,"DEGREES KELVIN"
390  IF S = 5 THEN 450
400  PRINT T1 + 459.58,"DEGREES RANKINE"
410  GOTO 450
420  PRINT "TEMPERATURE YOU ENTERED DOES NOT"
430  PRINT "EXIST.  PLEASE ENTER A NEW ONE"
440  GOTO 10
450  END
```

Reference

Lange. *Lange's Handbook of Chemistry,* 10th rev. ed. New York: McGraw-Hill, 1967.

Numeric Base Conversion

This program will convert numbers between any two bases 2 through 36. The program will continue to convert values from and to the same bases until you enter zero as the value to convert. Then you can enter a new base to convert to, still using the previously entered base to convert from. If you enter zero as the base to convert to, you must enter a new base to convert from. Enter zero at this point to end the program.

Program Notes

You may convert between a base greater than 36, as long as you define the characters to represent values greater than 35. To do this, add the character(s) you choose between the Z and the closing quotes in line 30. For example, to convert to base 37, we'll represent the number 36 with the character #. Change line 30 so that it reads:

```
30 N$="0123456789ABCDEFGHIJKLMNOPQRSTUVWXYZ#"
```

Signs, decimal points, and any other characters you enter as part of the value to be converted that are not included in the chapter representations for the FROM base you selected are interpreted as zeros wherever they appear.

Note that because the value you enter is converted to its base 10 value, which is stored in the numeric variable D, accuracy of the output value is limited by the accuracy of your computer. This is also true because of the repeated division used in the conversion process.

You may encounter problems using this program on your computer because of the use of string variables. See the Appendix of this book for information on conversion of programs which use string variables.

Example

What is the base 16 number ABCD in base 10? What is the base 8 value? What is the base 36 equivalent of the base 10 number 825,062?

Answer: ABCD base 16 is 43,981 base 10. The base 8 value is 125,715. 825,062 base 10 is HOME base 36.

```
NUMERIC BASE CONVERSION

FROM BASE (0 TO END) ?16
TO BASE ?10
VALUE ?ABCD
ABCD BASE 16 IS 43981 BASE 10
VALUE ?0
TO BASE ?8
VALUE ?ABCD
ABCD BASE 16 IS 125715 BASE 8
VALUE ?0
TO BASE ?0
FROM BASE (0 TO END) ?10
TO BASE ?36
VALUE ?825062
```

```
825062 BASE 10 IS HOME BASE 36
VALUE ?0
TO BASE ?0
FROM BASE (0 TO END) ?0
```

Practice Problems

1. What is the base 16 representation of the base 10 number 45? What is the base 8 representation?
Answer: 45 base 10 is 2D base 16. 45 base 10 is 55 base 8.

2. What is the base 32 representation of the base 18 number 1G6? What is the base 10 value?
Answer: 1G6 base 18 is JA base 32. 1G6 base 18 is base 10.

Program Listing

```
10  PRINT "NUMERIC BASE CONVERSION"
20  PRINT
30 N$ = "0123456789ABCDEFGHIJKLMNOPQRSTUVWXYZ"
40  REM  -- VARIABLE 'M' IS THE HIGHEST
45  REM  -- BASE YOU MAY CONVERT FROM / TO
50 M =  LEN (N$)
60  PRINT "FROM BASE (0 TO END) ";
70  INPUT B1
80  REM  -- END PROGRAM?
90  IF B1 = 0 THEN 450
100  REM  -- TEST FOR VALID INPUT BASE
110  IF B1 > 1 THEN 140
120  PRINT "BASES 2 THROUGH ";M;"ONLY. SELECT AGAIN. "
130  GOTO 60
140  IF B1 > M THEN 120
150  PRINT "TO BASE ";
160  INPUT B2
170  IF B2 = 0 THEN 60
180  REM  -- TEST FOR VALID OUTPUT BASE
190  IF B2 > 1 THEN 220
200  PRINT "BASES 2 THROUGH ";M;" ONLY. SELECT AGAIN."
210  GOTO 150
220  IF B2 > M THEN 200
230  PRINT "VALUE ";
240  INPUT V$
250  IF V$ = "0" THEN 150
260  REM  -- FIRST, CONVERT INPUT VALUE TO BASE 10
270 L =  LEN (V$)
280 D = 0
290  FOR I = 1 TO L
300  FOR J = 1 TO B1
310  IF  MID$ (N$,J,1) <  >  MID$ (V$,I,1) THEN 330
320 D = D +  INT ((J - 1) * (B1 ^ (L - I)) + 0.5)
330  NEXT J
340  NEXT I
350  REM  -- NOW CONVERT BASE 10 VALUE TO
355  REM  -- DESIRED OUTPUT BASE
360 O$ = ""
```

```
370 X =  INT (((D / B2) -  INT (D / B2)) * B2 + 1.5)
380 O$ =  MID$ (N$,X,1) + O$
390 D =  INT (D / B2)
400  IF D > 0 THEN 370
410  REM  -- OUTPUT THE RESULT
420  PRINT V$;" BASE ";B1;" IS ";O$;" BASE ";B2
430  REM  -- LOOP BACK TO ENTER ANOTHER VALUE
440  GOTO 230
450  END
```

Musical Transposition

In music, transposition is the art of playing music in a different key from that in which it was written. Some musicians can transpose by sight or by ear; others have to convert each note from one key into another, laboriously, one by one. This program is for those in the latter group. The notes transposed by this program can be used as the roots of harmonies for piano, guitar, and so forth, as easily as they can be used as single notes.

The program first displays all the keys and key signatures, comprising seven flats through seven sharps, with their identifying numbers. You enter the numbers for the keys from which and to which you are transposing. The program then displays each of the 12 possible notes, along with their transposed equivalents.

Note that the program will in all cases print out the correct pitch of the note it is transposing to, and in virtually all cases the correct name as well. However, in those rare cases of some minor keys with multiple accidentals, you may have to supply the alternate name where a double accidental (double sharp or double flat) is called for.

Example

What do notes in the key of B$^{\flat}$ become when you transpose to the key of G?

Answer:

```
MUSICAL TRANSPOSITION

IN THE FOLLOWING LIST OF KEYS
AND KEY SIGNATURES,
1.   A MAJOR/F-SHARP MINOR-3 SHARP
2.   B-FLAT MAJOR/G-MINOR-2 FLATS
3.   C-FLAT MAJOR/A-FLAT MINOR-7 FLATS
     B-MAJOR/G SHARP MINOR-5 SHARPS
4.   C MAJOR/A MINOR-NO SHARPS OR FLATS
5.   D-FLAT MAJOR/B-FLAT MINOR-5 FLATS
     C-SHARP MAJOR/A-SHARP MINOR-5
     SHARPS
6.   D MAJOR/B MINOR-2 SHARPS
7.   E-FLAT MAJOR/C MINOR-3 FLATS
8.   E MAJOR/C-SHARP MINOR-4 SHARPS
9.   F MAJOR/D MINOR-1 FLAT
10.  G-FLAT MAJOR/E-FLAT MINOR-6 FLATS
     F-SHARP MAJOR/D-SHARP MINOR-6
     SHARPS
11.  G MAJOR/E MINOR-1 SHARP
12.  A-FLAT MAJOR/F MINOR-4 FLATS
ENTER THE NO. OF THE KEYS FROM WHICH
YOU ARE TRANSPOSING, THEN THE NO. OF
THE KEY TO WHICH YOU ARE TRANSPOSING
?2,11
```

```
        TRANSPOSITION TABLE
 TRANSPOSED              TRANSPOSED
    FROM                     TO
A                       G-FLAT/F-SHARP
B-FLAT/A-SHARP          G
B (C-FLAT)              A-FLAT/G-SHARP

C (B-SHARP)             A
D-FLAT/C-SHARP          B-FLAT/A-SHARP
D                       B (C-FLAT)

E-FLAT/D-SHARP          C (B-SHARP)
E (F-FLAT)              D-FLAT/C-SHARP
F (E-SHARP)             D

G-FLAT/F-SHARP          E-FLAT/D-SHARP
G                       E (F-FLAT)
A-FLAT/G-SHARP          F (E-SHARP)


DO YOU WANT ANOTHER TRANSPOSITION? (Y/N)
?N
```

Practice Problems

1. In the key of G, the first chords of "My Country 'Tis of Thee" are: G, E^{m}, C, D, G, E^{m}, C, G, B^{7}, E^{m}. If it is transposed to E, what would these chords be?
Answer: E, $C^{\#m}$, A, B, E, $C^{\#m}$, A, E, $G^{\#7}$, $C^{\#m}$.

2. Bach's Fifth Brandenburg Concerto, written in D major, begins: D, D, $F^{\#}$, $F^{\#}$, A, A, D, D, $C^{\#}$, D, $C^{\#}$, B, A, G, $F^{\#}$, E. If he had written it in C major what would these notes have been?
Answer: C, C, E, E, G, G, C, C, B, C, B, A, G, F, E, D.

Program Listing

```
5  PRINT "MUSICAL TRANSPOSITION"
7  PRINT
10  DIM A$(12)
19  REM  READ TABLE OF NOTES
20  FOR I = 1 TO 12
30  READ A$(I)
40  NEXT I
50  DATA  "A","B-FLAT/A-SHARP","B (C-FLAT)","C (B-SHARP)"
60  DATA  "D-FLAT/C-SHARP","D","E-FLAT/D-SHARP","E (F-FLAT)"
70  DATA   "F (E-SHARP)","G-FLAT/F-SHARP","G","A-FLAT/G-SHARP"
280  PRINT "IN THE FOLLOWING LIST OF KEYS"
290  PRINT "AND KEY SIGNATURES,"
300  PRINT "1.  A MAJOR/F-SHARP MINOR-3 SHARP"
310  PRINT "2.  B-FLAT MAJOR/G-MINOR-2 FLATS"
320  PRINT "3.  C-FLAT MAJOR/A-FLAT MINOR-7 FLATS"
325  PRINT "    B-MAJOR/G SHARP MINOR-5 SHARPS"
330  PRINT "4.  C MAJOR/A MINOR-NO SHARPS OR FLATS"
```

```
340  PRINT "5.  D-FLAT MAJOR/B-FLAT MINOR-5 FLATS"
345  PRINT "    C-SHARP MAJOR/A-SHARP MINOR-5"
347  PRINT "    SHARPS"
350  PRINT "6.  D MAJOR/B MINOR-2 SHARPS"
360  PRINT "7.  E-FLAT MAJOR/C MINOR-3 FLATS"
370  PRINT "8.  E MAJOR/C-SHARP MINOR-4 SHARPS"
380  PRINT "9.  F MAJOR/D MINOR-1 FLAT"
390  PRINT "10. G-FLAT MAJOR/E-FLAT MINOR-6 FLATS"
395  PRINT "    F-SHARP MAJOR/D-SHARP MINOR-6"
397  PRINT "    SHARPS
400  PRINT "11. G MAJOR/E MINOR-1 SHARP"
410  PRINT "12. A-FLAT MAJOR/F MINOR-4 FLATS"
450  PRINT "ENTER THE NO. OF THE KEYS FROM WHICH"
460  PRINT "YOU ARE TRANSPOSING, THEN THE NO. OF"
470  PRINT "THE KEY TO WHICH YOU ARE TRANSPOSING"
480  INPUT A,B
500  PRINT
510  IF A > 12 OR B > 12 OR A < 1 OR B < 1 THEN 620
610  IF A <  > B THEN 710
620  PRINT "ERROR. PLEASE ENTER AGAIN"
630  GOTO 450
710  PRINT "        TRANSPOSITION TABLE"
720  PRINT " TRANSPOSED"; TAB( 20);"TRANSPOSED"
730  PRINT  TAB( 4);"FROM"; TAB( 24);"TO"
740 P = 0
749  REM  PRINT TABLE
750  FOR I = 1 TO 12
755 D = B - A + I -  SGN ( INT ((B - A + I) / 12)) * 12
757  IF D > 0 THEN 760
758 D = 12
760  PRINT A$(I); TAB( 20);A$(D)
770 P = P + 1
780  IF P / 3 <  >  INT (P / 3) THEN 810
790  PRINT
800 P = 0
810  NEXT I
820  PRINT
830  PRINT "DO YOU WANT ANOTHER TRANSPOSITION? (Y/N)"
840  INPUT Y$
850  IF Y$ = "Y" THEN 280
860  END
```

References

Pistan. *Harmony,* 3rd ed. New York: Norton, 1969.

Priesing and Tecklin. *Language of the Piano.* Boston: Carl Fischer, 1959.

Appendix

Here in the appendix you will find suggestions for changing the programs to accommodate different output devices.

We describe each of the specific changes listed below in a general way and illustrate wherever possible with an example taken from the book. You must decide how a suggested change would apply to any particular program, if at all. Therefore, you will need some understanding of Basic programming in order to implement these changes.

Pausing With Full Display Screen

Many programs have more lines of output than will fit on a typical screen. This means the first lines of output flash by quickly and scroll off the top of the screen, leaving you with no idea of what they contained. On the Apple II, you can press the CONTROL and S keys simultaneously to freeze the display temporarily. You can then review and record anything on the display. Subsequently pressing any key other than the CONTROL key sets the computer in motion. More program output appears. You may have to freeze the display several times in order to see all the output. The number of times you must freeze the display depends not only on which program you are running, but also on the nature of the problem you present it with.

Alternatively, you can modify a program so that it pauses at one or more points during its output, waiting for the user to cue it to continue. To do this, add the following subroutine to the program, and call the subroutine at suitable intervals during the output phrase of the program.

```
5799 REM WAIT FOR OPERATOR CUE
5800 PRINT "ENTER 'C' TO CONTINUE"
5810 INPUT W$
5820 RETURN
```

This technique is used in the Income Averaging program. In programs where some or all of the output occurs inside a loop (for example, between FOR and NEXT statements), you may not be able to merely place calls to this subroutine between appropriate PRINT statements, as we did in the Income Averaging program on lines 1890, 2010, and 2110. In this case, use the subroutine below, which counts the number of lines displayed since the last pause. Each time you call this subroutine, it increments a counter, and tests to see if the new count exceeds the size of the display. If so, it pauses for the operator cue. Otherwise, it simply returns to the calling point in the program. Therefore, you would insert a call to this subroutine immediately after every PRINT statement that causes a line of output (that is, a PRINT statement not ending with a comma or semicolon).

```
5797 REM SUBROUTINE CHECKS LINE COUNT
5798 REM WAITS FOR CUE IF DISPLAY IS FULL
5799 REM FIRST INCREMENT AND CHECK LINE COUNT
5800 L9 = L9 + 1
5810 IF L9 < 20 THEN 5850
5819 REM SCREEN IS FULL - -
5820 PRINT "ENTER 'C' TO CONTINUE";
5830 INPUT W$
5839 REM RESET LINE COUNT
5840 L9 = 0
5850 RETURN
```

Printer Output

Viewing program output on the display screen is perfectly acceptable when you are using a program as an experimental or investigative tool. But sooner or later, you will probably tire of continually copying program output from the display by hand. The solution, of course, is to direct program output to a printer. The procedure for doing this varies from one Apple to the next. You can cause output to appear only on the printer by entering PR #I where I is the port your printer card is in just before you run a program.

Changing the Precision of Rounded Values

Many of the programs employ user-defined functions to round numeric values to a certain number of decimal places. For example, the Net Present Value program has a function on line 20 which does this:

```
20 DEF FNA(X)=INT(X*100 + 0.5)/100
```

This function rounds to the nearest hundredth, thus calculating the net present value to the nearest cent. The value 100 which appears twice in the function definition statement shown above determines how many decimal digits there will be (two in this case). To change the number of decimal digits, change both occurrences of the value 100, or whatever value is specified in the program you are considering. For example, the following replacement for line 20 will calculate net present value to the nearest whole dollar:

```
20 DEF FNA(X) = INT(X*1 + 0.5)/1
```

Or more simply stated:

```
20 DEF FNA(X) = INT(X + 0.5)
```

Frequency of Compounding Interest

Several of these programs base their computations on interest compounded annually. This is acceptable in most cases. But you can have the calculations compound interest more frequently. Perhaps the easiest way to do this is to convert the annual interest rate to the effective interest rate, based on the number of compounding periods per year. Then enter this effective rate when the program asks for an interest rate. The general formula for this is

$$E = \left(1 + \frac{I}{N}\right)^{NY}$$

where E is the effective interest rate, I is the annual interest rate expessed as a decimal fraction, N is the number of compounding periods per year, and Y is the number of years. The formula for continuous compounding is:

$$E = e^{IY}$$

where E is the effective interest rate, e is 2.718281828... (the base of natural logarithms), I is the nominal interest rate, and Y is the number of years.

Of course, you can change a program to accept the nominal interest rate and convert it automatically to the effective interest rate. The program would have to ask for the number of compounding periods per year in order to make the conversion. Alternatively, you could restate the interest compounding calculation in the program so that it compounds at the desired frequency. For example, this calculation occurs in the Future Value of an Investment program on line 240. If you restate line 240 as shown below, the program will compute the future value of an investment at growth rate R, compounded continuously.

```
240 T=T +FNA(C(J)*EXP(R*N-J)))
```

Other OSBORNE/McGraw-Hill Publications

An Introduction to Microcomputers: Volume 0 — The Beginner's Book
An Introduction to Microcomputers: Volume 1 — Basic Concepts, 2nd Edition
An Introduction to Microcomputers: Volume 3 — Some Real Support Devices
Osborne 4 & 8-Bit Microprocessor Handbook
Osborne 16-Bit Microprocessor Handbook
8089 I/O Processor Handbook
CRT Controller Handbook
68000 Microprocessor Handbook
8080A/8085 Assembly Language Programming
6800 Assembly Language Programming
Z80 Assembly Language Programming
6502 Assembly Language Programming
Z8000 Assembly Language Programming
6809 Assembly Language Programming
Running Wild — The Next Industrial Revolution
The 8086 Book
PET and the IEEE 488 Bus (GPIB)
PET/CBM Personal Computer Guide, 2nd Edition
Business System Buyer's Guide
Osborne CP/M® User Guide
Apple II® User's Guide
Microprocessors for Measurement and Control
WordStar™ Made Easy
Some Common BASIC Programs
Some Common BASIC Programs — PET/CBM Edition
Some Common BASIC Programs — Atari® Edition
Some Common BASIC Programs — TRS-80™ Level II Edition
Some Common BASIC Programs — Apple II® Edition
Practical BASIC Programs
Practical BASIC Programs — TRS-80™ Level II Edition
Practical BASIC Programs — Apple II® Edition
Payroll with Cost Accounting
Accounts Payable and Accounts Receivable
General Ledger
8080 Programming for Logic Design
6800 Programming for Logic Design
Z80 Programming for Logic Design
CBASIC™ User Guide
Science & Engineering Programs — Apple II® Edition
Interfacing to S-100/IEEE 696 Microcomputers

Practical BASIC Programs Apple II® Edition

Income averaging, financial rate of return, net present value, statistical estimation theory, temperature conversion, home budgeting, and even checkbook reconciliation represent but a few of the 40 practical programs offered in **Practical BASIC Programs — Apple II® Edition.** This collection of programs can help solve the most difficult problems in the office, classroom, and home.

The programs can be keyed directly into your Apple II® computer as they are listed here. Sample runs, practical problems, and BASIC source listings make this an essential reference source for any Apple II® user. Each program includes an easy-to-understand discussion of subject matter, required inputs, and program results. Program notes offer adaptations to fit your individual needs.

Other OSBORNE/McGraw-Hill books of interest:

Some Common BASIC Programs: Apple II ® Edition
Science and Engineering Programs: Apple II® Edition
Apple II® User's Guide
6502 Assembly Language Subroutines
6502 Assembly Language Programming

® Apple II is a registered trademark of Apple Computer, Inc.

ISBN 0-931988-66-7